

Lambda Cases (lcases)

Dimitris Saridakis

Contents

1	Introduction	2
2	Language Description: General	2
2.1	Program Structure	2
2.2	Keywords	3
3	Language Description: Values	4
3.1	Basic Expressions	4
3.1.1	Literals and Identifiers	4
3.1.2	Parenthesis, Tuples and Lists	5
3.1.3	Parenthesis Function Application	7
3.1.4	Prefix and Postfix Functions	8
3.2	Operators	10
3.2.1	Function Application Operators	10
3.2.2	Function Composition Operators	11
3.2.3	Arithmetic Operators	11
3.2.4	Comparison and Boolean Operators	12
3.2.5	Environment Action Operators	13
3.2.6	Operator Expressions	15
3.2.7	Complete Table, Precedence and Associativity	16
3.3	Function Expressions	18
3.3.1	Regular Function Expressions	18
3.3.2	"cases" Function Expressions	18
3.4	Value Definitions and "where" Expressions	20
3.4.1	Value Definitions	20
3.4.2	"where" Expressions	21
3.5	Indentation and Complete Grammar for Values	22
3.5.1	Indentation System	22
3.5.2	Complete Grammar for Values	23
4	Language Description: Types and Type Logic	25
4.1	Types	25
4.1.1	Type Expressions	26
4.1.2	Type Definitions: Tuple Types	29
4.1.3	Type Definitions: Or Types	30
4.2	Type Logic	32
4.2.1	Proposition Definitions: Atomic Type Propositions	32
4.2.2	Proposition Definitions: Compound Type Propositions	33
4.2.3	Type Theorems	33
4.3	Complete Grammar for Types and Type Logic	33

5	Predefined	33
5.1	Constants	33
5.2	Functions	33
5.3	Types	33
5.4	Type Propositions	33
6	Parser implementation	33
6.1	AST Types	33
6.2	Parsers	33
7	Translation to Haskell	33
8	Running Examples	33
9	Conclusion	33

1 Introduction

Haskell is a delightful language. Yet, for some reason, it doesn't seem to have it's rightful place in terms of popularity in industry. Why is it so? Is it inherently hard to learn and therefore only the brave dare to use it, or could it be that the syntax is perplexing to the amateur eye? It is my belief that with some syntax changes that give a greater familiarity to the new user, there would be no language more compelling than (the new) Haskell. In an attempt to achieve that familiarity, I present some (hopefully useful) new syntax, of which some is closer to the imperative/OOP style (to attract more already experienced programmers from these languages), some is closer to mathematics (in which most programmers should be experienced) and some is closer to natural language (in which we are all already experienced).

2 Language Description: General

2.1 Program Structure

An lcases program consists of a set of definitions and theorems. Definitions are split into value definitions, type definitions and type proposition definitions. Theorems are proven type propositions. Functions as well as "Environment Actions" (see section 3.2.5) are also considered values. The definition of the "main" value determines the program's behaviour.

Program example: extended euclidean algorithm

```
// type definitions

tuple_type Coeffs
value (previous, current) : Int x Int

tuple_type GcdAndCoeffs
value (gcd, a, b) : Int x Int x Int

// algorithm

ext_euc
: (Int, Int) => GcdAndCoeffs
= ext_euc_rec(init_a_coeffs, init_b_coeffs)
  where
    init_a_coeffs, init_b_coeffs
```

```

    : all Coeffs
    = (1, 0), (0, 1)
ext_euc_rec
  : (Coeffs, Coeffs, Int, Int) => GcdAndCoeffs
  = (a_coeffs, b_coeffs, x, cases) =>
    0 => (x, a_coeffs.previous, b_coeffs.previous)
    y => ext_euc_rec(next <- a_coeffs, next <- b_coeffs, y, x -> mod <- y)
    where
      next
        : Coeffs => Coeffs
        = cs => (cs.current, cs.previous - x -> div <- y * cs.current)

// reading, printing and main

read_two_ints
  : (Int x Int)IOAction
  = print <- "Please give me 2 ints";
  get_line ;> split_words o> apply(from_string)to_all o> cases =>
    [x, y] => (x, y) -> io_action
    ... => io_error <- "You didn't give me 2 ints"

print_gcd_equation
  : (Int, Int, GcdAndCoeffs) => (EmptyVal)IOAction
  = (x, y, gcd_cs) => print("gcd = " + gcd_cs.gcd + " = " + linear_comb_string)
  where
    linear_comb_string
      : String
      = gcd_cs.a + " * " + x + " + " + gcd_cs.b + " * " + y

main
  : (EmptyVal)IOAction
  = read_two_ints ;> ints =>
    print_gcd_equation(ints.1st, ints.2nd, ext_euc(ints.1st, ints.2nd))

```

Program grammar

$\langle \text{program} \rangle ::= (\langle \text{value-def} \rangle \mid \langle \text{type-def} \rangle \mid \langle \text{type-prop-def} \rangle \mid \langle \text{type-theo} \rangle)^+$

2.2 Keywords

The `lcases` keywords are the following:

`cases` `where` `all` `tuple_type` `value` `or_type` `values`
`type_proposition` `equivalent` `type_theorem` `proof`

Each keyword's functionality is described in the respective section shown in the table below:

Keyword	Section
<code>cases</code>	3.3 Function Expressions
<code>where all</code>	3.4 Value Definitions
<code>tuple_type</code> <code>value</code> <code>or_type</code> <code>values</code>	4.1 Types
<code>type_proposition</code> <code>function</code> <code>equivalent</code> <code>type_theorem</code> <code>proof</code>	4.2 Type Logic

The "cases" and "where" keywords are also reserved words. Therefore, even though they can be generated by the "identifiers" grammar, they cannot be used as identifiers (see "Literals and Identifiers" section 3.1.1).

3 Language Description: Values

3.1 Basic Expressions

3.1.1 Literals and Identifiers

Literals

- *Examples*

```
1 2 17 42 -100
1.61 2.71 3.14 -1234.567
'a' 'b' 'c' 'x' 'y' 'z' ', ' ', ' '\n'
"Hello World!" "What's up, doc?" "Alrighty then!"
```

- *Description*

There are literals for the four basic types: Int, Real, Char, String. These are the usual integers, real numbers, characters and strings. The exact specification of literals is the same as in the Haskell report.

- *Grammar*

$\langle literal \rangle ::= \langle literal \rangle$

TODO add the grammar from the haskell report

Identifiers

- *Examples*

```
x y z
a1 a2 a3
funny_identifier
unnecessarily_long_identifier
apply()to_all
```

- *Description*

An identifier is the name of a value or a parameter. It is used in the definition of a value (see "value definition" section 3.4) and in expressions that use that value, or in the parameters of a function and in the body of that function.

An identifier starts with a lower case letter and is followed by lower case letters or underscores. It is also possible to have pairs of parentheses in the middle of an identifier (see "Parenthesis Function Application" section 3.1.3 for why this can be useful). Finally, an identifier can be ended with a digit.

- *Grammar*

$\langle identifier \rangle ::= [a-z] [a-z_]* ('()' [a-z_]+)^* [[0-9]]$

Even though the "cases" and "where" keywords can be generated by this grammar, they cannot be used as identifiers.

3.1.2 Parenthesis, Tuples and Lists

Parenthesis

- *Examples*

```
(1 + 2)
(((1 + 2) * 3) ^ 4)
(x => f(x) + 1)
(s => "f(val) + 1 is: " + s)
(val -> (x => f(x) + 1) -> to_string -> (s => "f(val) + 1 is: " + s))
("Line is: " + line)
(get_line ;> line => print("Line is: " + line))
(do(3)times <- (get_line ;> line => print("Line is: " + line)))
```

- *Description*

An expression is put in parenthesis to prioritize it or isolate it in a bigger (operator) expression. The expressions inside parenthesis are operator or function expressions.

- *Grammar*

$\langle \text{paren-expr} \rangle ::= '(\langle \text{op-or-func-expr} \rangle)'$

$\langle \text{op-or-func-expr} \rangle ::= \langle \text{simple-op-expr} \rangle \mid \langle \text{op-expr-func-end} \rangle \mid \langle \text{simple-func-expr} \rangle$

Tuples

- *Examples*

```
(1, "What's up, doc?")
(2, "Alrighty then!", 3.14)
(x, y, z, w)
(1, my_function, (x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1/2))
```

- *Description*

Tuples are used to group many values (of possibly different types) into one. The type of the tuple can be either the product of the types of the fields or a defined `tuple_type` which is equivalent to the aforementioned product type (see "tuple_type" section 4.1.2). For example, the type of the second tuple above could be:

```
Int x String x Real
```

or:

```
MyType
```

assuming "MyType" has been defined in a similar way to the following:

```
tuple_type MyType
value (my_int, my_string, my_real) : Int x String x Real
```

- *Big Tuples*

Example

```

my_big_tuple
: String x Int x Real x String x String x (String x Real x Real)
= ( "Hey, I'm the first field and I'm also a relatively big string."
  , 42, 3.14, "Hey, I'm the first small string", "Hey, I'm the second small string"
  , ("Hey, I'm a string inside the nested tuple", 2.71, 1.61)
  )

```

Description

It is possible to stretch a (big) tuple expression over multiple lines (only) in a separate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '(' is after the "=" part of the value definition and the first field must be in the same line.
- The tuple can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '(' character was in the first line.
- The tuple must be ended by a line that only contains the ')' character and is also indented so that the ')' is in same column where the '(' character was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

```

⟨tuple⟩ ::= '(' ⟨line-expr⟩ '⟦' ⟨comma-sep-line-exprs⟩ ')'
⟨comma-sep-line-exprs⟩ ::= ⟨line-expr⟩ ( '⟦' ⟨line-expr⟩ )*
⟨line-expr⟩ ::= ⟨no-paren-op-arg⟩ | ⟨op-or-func-expr⟩

⟨big-tuple⟩ ::=
  '(' ⟨line-expr⟩ [ '\n' ⟨indent⟩ ] '⟦' ⟨comma-sep-line-exprs⟩
  ( '\n' ⟨indent⟩ '⟦' ⟨comma-sep-line-exprs⟩ )*
  '\n' ⟨indent⟩ ')'

```

Lists

- *Examples*

```

[1, 2, 17, 42, -100]
[1.61, 2.71, 3.14, -1234.567]
["Hello World!", "What's up, doc?", "Alrighty then!"]
[x => x + 1, x => x + 2, x => x + 3]
[x, y, z, w]

```

- *Description*

Lists are used to group many values of the same type into one. The type of the list is `ListOf(A)s` where `A` is the type of every value inside. Therefore, the types of the first four examples are:

```

ListOf(Int)s
ListOf(Real)s
ListOf(String)s
(A)And(Int)AddTo(B) ==> ListOf(A => B)s

```

And the last list is only legal if `x`, `y`, `z` and `w` all have the same type. Assuming they do and it's the type `T`, the type of the list is:

ListOf(T)s

- *Big Lists*

It is possible to stretch a (big) list expression over multiple lines (only) in a separate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '[' is after the "=" part of the value definition and the first element must be in the same line.
- The list can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '[' character was in the first line.
- The tuple must be ended by a line that only contains the ']' character and is also indented so that the ']' is in same column where the '[' character was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

Example:

```
my_big_list
: ListOf(Int => (EmptyVal)IOAction)s
= [ x => print("I'm the first function and x + 1 is: " + (x + 1))
  , x => print("I'm the second function and x + 2 is: " + (x + 2))
  , x => print("I'm the third function and x + 3 is: " + (x + 3))
  ]
```

- *Grammar*

$\langle list \rangle ::= '[' [\langle comma-sep-line-exprs \rangle] ']'$

$\langle big-list \rangle ::= '[' \langle comma-sep-line-exprs \rangle ('\n' \langle indent \rangle ',' \langle comma-sep-line-exprs \rangle)^* '\n' \langle indent \rangle ']'$

3.1.3 Parenthesis Function Application

- *Examples*

```
f(x)
f(x, y, z)
(x)to_string
apply(f)to_all
apply(f)to_all(1)
```

- *Description*

Function application in lcases can be done in many different ways in an attempt to maximize readability. In this section, we discuss the ways function application can be done with parenthesis.

In the first two examples, we have the usual mathematical function application which is also used in most programming languages and should be familiar to the reader. That is, function application is done with the arguments of the function in parenthesis separated by commas and **appended** to the function identifier.

We extend this idea by allowing the arguments to be **prepended** to the function identifier (third example). Finally, it is also possible to have the arguments **inside** the function identifier provided the function has been **defined with parentheses inside the identifier**. For example, below is the definition of "apply()to_all":

```

apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
  = (f, cases) =>
    empty => empty
    non_empty:l => non_empty:(f <- l.head, apply(f)to_all <- l.tail)

```

The actual definition doesn't matter at this point, what matters is that the identifier is "apply()to_all" with the parentheses **included**. This is very useful for defining functions where the argument in the middle makes the function application look and sound more like natural language.

It is possible to have many parentheses pairs in a single function application (last example). The arguments are always inserted to the function from **left to right**. Therefore, when multiple parentheses pairs are present the arguments of the leftmost parentheses are inserted first then the next ones to the right and so on.

- *Grammar*

```

⟨paren-func-app⟩ ::=
  [ ⟨arguments⟩ ] ⟨identifier-with-arguments⟩ [ ⟨arguments⟩ ]
  |   ⟨arguments⟩ ⟨identifier⟩ [ ⟨arguments⟩ ]
  |   ⟨identifier⟩ ⟨arguments⟩

⟨arguments⟩ ::= ' ( ' ⟨comma-sep-line-exprs⟩ ' ) '

⟨identifier-with-arguments⟩ ::=
  [a-z] [a-z_]* ( ' ( ' [a-z_]+ ) * ⟨arguments⟩ [a-z_]+ ( ( ' ( ' | ⟨arguments⟩ ) [a-z_]+ ) * [ [0-9] ]

```

3.1.4 Prefix and Postfix Functions

Prefix Functions

- *Examples*

```

the_value:1
non_empty:1
error:e
result:r
apply(the_value:)to_all

```

- *Description*

Prefix functions are automatically generated from **or_type** definitions (see "Or Types" section 4.1.3). They are functions that convert a value of a particular type to a value that is a case of an **or_type** and has values of this type inside. For example in the first example above we have:

```

1
  : Int
the_value:1
  : Possibly(Int)

```

Where the function **thevalue:** is automatically generated from the definition of the **Possibly** type:

```

or_type Possibly(A)
values the_value:A | no_value

```


And it has the type `A => Possibly(A)`.

These functions are called prefix functions because they are prepended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function `the_value:` is an argument of the function `apply()to_all`. Prefix functions always end with a colon.

- *Grammar*

$\langle pre\text{-}func \rangle ::= \langle identifier \rangle \text{'\text{:}'}$

$\langle pre\text{-}func\text{-}app \rangle ::= \langle pre\text{-}func \rangle (\langle basic\text{-}expr \rangle | \langle paren\text{-}expr \rangle | \langle pre\text{-}func\text{-}app \rangle)$

$\langle basic\text{-}expr \rangle ::= \langle literal \rangle | \langle identifier \rangle | \langle tuple \rangle | \langle list \rangle | \langle paren\text{-}func\text{-}app \rangle | \langle post\text{-}func\text{-}app \rangle$

Postfix Functions

- *Examples*

```
name.first_name
list.head
date.year
tuple.1st
apply(.1st)to_all
```

- *Description*

Postfix functions are automatically generated from `tuple_type` definitions (see "Tuple Types" section 4.1.2). They are functions that take a `tuple_type` value and return a particular field (i.e. projection functions). For example in the first example above we have:

```
name
  : Name
name.first_name
  : String
```

Where the function `.first_name` is automatically generated from the definition of the `Name` type:

```
tuple_type Name
value (first_name, last_name) : String x String
```

And it has the type `Name => String`.

There are also the following special projection functions for **product types**: `.1st` `.2nd` `.3rd` `.4th` `.5th`. For the 4th example above, assuming:

```
tuple
  : Int x String
```

We have:

```
tuple.1st
  : Int
```

The general types of these functions are:

```
.1st
  : (A)Is(B)s1st ==> B => A
.2nd
  : (A)Is(B)s2nd ==> B => A
...
```

These functions are called postfix functions because they are appended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function `.1st` is an argument of the function `apply()to_all`. Postfix functions always begin with a dot.

- *Grammar*

$\langle post\text{-}func \rangle ::= \text{'.'} \langle identifier \rangle$

$\langle post\text{-}func\text{-}app \rangle ::= (\langle paren\text{-}expr \rangle | \langle basic\text{-}expr \rangle) \langle post\text{-}func \rangle$

3.2 Operators

3.2.1 Function Application Operators

Operator	Type
<code>-></code>	$(A, A \Rightarrow B) \Rightarrow B$
<code><-</code>	$(A \Rightarrow B, A) \Rightarrow B$

The function application operators `"->"` and `"<-"` are a different way to apply functions to arguments than the usual parenthesis function application. They are meant to look like arrows that point from the argument to the function. These operators are very useful for chaining many function applications without the clutter of having to open and close parentheses for each one of the functions. For example, assuming we have the following functions with the behaviour suggested by their names and types:

```
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
string_length
  : String => Int
filter_with
  : (A => Bool, ListOf(A)s) => ListOf(A)s
is_odd
  : Int => Bool
sum_ints
  : ListOf(Int)s => Int
```

And a list of strings:

```
strings
  : ListOf(String)s
```

Here is a simple way to get the total number of characters in all the strings that have odd length:

```
chars_in_odd_length_strings
  : Int
  = strings -> apply(string_length)to_all -> filter_with(is_odd) -> sum_ints
```

Ofcourse this can be done equivalently using the other operator:

```
chars_in_odd_length_strings
  : Int
  = sum_ints <- filter_with(is_odd) <- apply(string_length)to_all <- strings
```

These operators can also be used together to put a function between two arguments if that function is commonly used that way in math (or if it looks better for a certain function). For example the "mod" function can be used like so:

$$x \rightarrow \text{mod} \leftarrow y$$

Which is equivalent to:

$$\text{mod}(x, y)$$

3.2.2 Function Composition Operators

Operator	Type
$\circ>$	$(A \Rightarrow B, B \Rightarrow C) \Rightarrow (A \Rightarrow C)$
$<\circ$	$(B \Rightarrow C, A \Rightarrow B) \Rightarrow (A \Rightarrow C)$

The function composition operators " $\circ>$ " and " $<\circ$ " are used to compose functions, each one in the corresponding direction. The use of the letter 'o' is meant to be similar to the mathematical function composition symbol ' \circ ' and the symbols ' $>$ ', ' $<$ ' are used so that the operator points from the function which is applied first to the function which is applied second. A neat example using function composition is the following. Assuming we have the following functions with the behaviour suggested by their names and types:

```
split_words
  : String => ListOf(String)s
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
reverse_string
  : String => String
merge_words
  : ListOf(String)s => String
```

We can reverse the all the words in a string like so:

```
reverse_words
  : String => String
= split_words o> apply(reverse_string)to_all o> merge_words
```

Ofcourse this can be done equivalently using the other operator:

```
reverse_words
  : String => String
= merge_words <o apply(reverse_string)to_all <o split_words
```

3.2.3 Arithmetic Operators

Operator	Type
\wedge	$(A)\text{ToThe}(B)\text{HasType}(C) \Rightarrow (A, B) \Rightarrow C$
$*$	$(A)\text{And}(B)\text{MultiplyTo}(C) \Rightarrow (A, B) \Rightarrow C$
$/$	$(A)\text{Divides}(B)\text{To}(C) \Rightarrow (B, A) \Rightarrow C$
$+$	$(A)\text{And}(B)\text{AddTo}(C) \Rightarrow (A, B) \Rightarrow C$
$-$	$(A)\text{SubtractsFrom}(B)\text{To}(C) \Rightarrow (B, A) \Rightarrow C$

The usual arithmetic operators work as they are expected, similarly to mathematics and other programming languages for the usual types. However, they are generalized. The examples below show their generality:

```
>> 1 + 1
2
>> 1 + 3.14
4.14
>> 'a' + 'b'
"ab"
>> 'w' + "ord"
"word"
>> "Hello " + "World!"
"Hello World!"
>> 5 * 'a'
"aaaaa"
>> 5 * "hi"
"hihihihihi"
>> "1,2,3" - ',', '
"123"
```

Let's analyze further the example of addition. The type can be read as such: the '+' operator has the type $(A, B) \Rightarrow C$, provided that the type proposition $(A) \text{And} (B) \text{AddTo} (C)$ holds. This proposition being true, means that addition has been defined for these three types (see section "Type Logic" 4.2 for more on type propositions). For example, by the examples above we can deduce that the following propositions are true (in the order of the examples):

```
(Int)And(Int)AddTo(Int)
(Int)And(Real)AddTo(Real)
(Char)And(Char)AddTo(String)
(Char)And(String)AddTo(String)
(Int)And(Char)MultiplyTo(String)
(Int)And(String)MultiplyTo(String)
(Char)SubtractsFrom(String)To(String)
```

This allows us to use the familiar arithmetic operators in types that are not necessarily numbers but it is somewhat intuitively obvious what they should do in those other types. Furthermore, their behaviour can be defined by the user for new user defined types!

3.2.4 Comparison and Boolean Operators

Operator	Type
= /=	$(A) \text{HasEquality} \Rightarrow (A, A) \Rightarrow \text{Bool}$
>= <=	$(A) \text{HasOrder} \Rightarrow (A, A) \Rightarrow \text{Bool}$
> <	$(A) \text{And} (B) \text{AreComparable} \Rightarrow (A, B) \Rightarrow \text{Bool}$
&	$(\text{Bool}, \text{Bool}) \Rightarrow \text{Bool}$

The comparison and boolean operators behave the same as in Haskell and very similarly to most programming languages. The main difference is that in lcases the "equals", "and" and "or" operators have the symbol once (`= & |`) rather than twice (`== && ||`).

3.2.5 Environment Action Operators

Operator	Type
<code>; ></code>	$(E) \text{IsAnEnvAction} \implies (E(A), A \Rightarrow E(B)) \Rightarrow E(B)$
<code>;</code>	$(E) \text{IsAnEnvAction} \implies (E(A), E(B)) \Rightarrow E(B)$

Simple Example

```
print_string("I'll repeat the line.") ; get_line ;> print_string
```

The example above demonstrates the use of the environment action operators with the `IOAction` type, which is how IO is done in `lcases`. Some light can be shed on how this is done, if we take a look at the types (as always!):

```
print_string
  : String => (EmptyVal)IOAction
print_string("I'll repeat the line.")
  : (EmptyVal)IOAction
get_line
  : (String)IOAction
;
  : (E)IsAnEnvAction ==> (E(A), E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line
  : (String)IOAction
  where (IOAction)IsAnEnvAction is true, E = IOAction, A = EmptyVal, B = String
;>
  : (E)IsAnEnvAction ==> (E(A), A => E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line ;> print_string
  : (EmptyVal)IOAction
  where (IOAction)IsAnEnvAction is true, E = IOAction, A = String, B = EmptyVal
```

Example program

```
main
  : (EmptyVal)IOAction
= print_string <- "Hello! What's your name?" ; get_line ;> name =>
  print_string("Oh hi " + name + "! What's your age?") ; get_line ;> age =>
  print_string("Oh that's crazy " + name + "! I didn't expect you to be " + age + "!");
```

In this bigger but similar example the types are:

```
print_string
  : String => (EmptyVal)IOAction
get_line
  : (String)IOAction

print_string <- "Hello! ... "
  : (EmptyVal)IOAction
print_string("Oh hi...")
  : (EmptyVal)IOAction
print_string("Oh that's crazy...")
  : (EmptyVal)IOAction
;
  : (E)IsAnEnvAction ==> (E(A), E(B)) => E(B)
```

```

print_string("Oh hi...") ; get_line
  : (String)IOAction
  where (IOAction)IsAnEnvAction is true, E = IOAction, A = EmptyVal, B = String

age => print_string("Oh that's crazy...")
  : String => (EmptyVal)IOAction

;>
  : (E)IsAnEnvAction ==> (E(A), A => E(B)) => E(B)

print_string("Oh hi...") ; get_line ;> age =>
print_string("Oh that's crazy...")
  : (EmptyVal)IOAction
  where (IOAction)IsAnEnvAction is true, E = IOAction, A = String, B = EmptyVal

print_string <- "Hello..." ; get_line
  : (String)IOAction

name => print_string("Oh hi ... (till the end)")
  : String => (EmptyVal)IOAction

print_string <- "Hello..." ; get_line ;> name =>
print_string("Oh hi ... (till the end)")
  : (EmptyVal)IOAction

```

Therefore, "main : (EmptyVal)IOAction" checks out. The key here is to remember that function expressions extend to the end of the whole expression. Therefore, we have "name => ... (till the end)" and "age => ... (till the end)" as the second arguments of the two occurrences of the ";>" operator.

Description

The environment action operators are used to combine values that do environment actions into values that do more complicated environment actions. Environment actions are also represented by types. More accurately, type constructors that take a type as an argument and produce a new type (just like ListOf(s)). A value of the type $E(A)$ where $(E)IsAnEnvAction$ does an environment action of type E that produces value of type A .

The effect of the ";" operator described in words is the following: given a value of type $E(A)$ and a value of type $E(B)$ (which do environment actions that produce values of type A and B respectively), create a new value that does both actions (provided the first did not result in an error). The overall effect is a value that does an environment action of type E (the combination of the "smaller" actions) which produces a value of type B (the one produced by the second action) and therefore it is of type $E(B)$.

Note that the value of type A produced by the first action is not used anywhere. This happens mostly when $A = EmptyVal$ and it is because values of type $E(EmptyVal)$ are used for their environment action only (e.g. `print_string(...) : (EmptyVal)IOAction`).

How the two environment actions of the $E(A)$ and $E(B)$ values are combined to produce the new environment action is specific to the environment action type E .

The effect of the ";>" operator described in words is the following: given a value of type $E(A)$ (which does an environment action of type E that produces a value of type A) and a value of type $A \Rightarrow E(B)$ (which is a function that takes a value of type A and returns an environment action of type E that produces a value of type B), combine those two values by creating a value that does the following:

- Performs the first action that produces a value of type A
- Takes the value of type A produced (provided there was no error) and passes it to the function of type A => E(B) that then returns an action
- Performs the resulting action

The overall effect is an environment action of type E that at the end produces a value is of type B and therefore the new value is of type E(B).

3.2.6 Operator Expressions

- *Examples*

```
1 + 2
1 + x * 3 ^ y
"Hello " + "World!"
x -> f -> g
f o> g o> h
x = y
x >= y - z & x < 2 * y
get_line ; get_line ;> line => print("Second line: " + line)
```

- *Description*

Operator expressions are expressions that use operators. Operators act like two-argument-functions that are placed in between their arguments. Therefore, they have function types and they act as it is described in their respective sections above this one.

An operator expression might have multiple operators. The order of operations is explained in the next section ("Complete Table, Precedence and Associativity") in Table 2.

Just like functions, the sub-expressions that act as arguments to an operator, must have types that match the types expected by the operator.

It is possible to end an operator expression with a function. This is mostly useful with the ";>" operator (see previous section: "Environment Operators"), but it is also possible with the following operators: "->", "o>", "<o".

- *Big Operator Expressions*

It is possible to stretch a (big) operator expression over multiple lines. In that case:

- The operator expression must split in a new line after an operator (not an argument).
- Every line after the first must be indented so that it begins at the column where the first character of the operator expression was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

$$\langle op\text{-}expr \rangle ::= \langle simple\text{-}op\text{-}expr \rangle \mid \langle op\text{-}expr\text{-}func\text{-}end \rangle \mid \langle big\text{-}op\text{-}expr \rangle \mid \langle cases\text{-}op\text{-}expr \rangle$$

$$\langle simple\text{-}op\text{-}expr \rangle ::= \langle op\text{-}arg \rangle (\text{'\square'} \langle op \rangle \text{'\square'} \langle op\text{-}arg \rangle) +$$

$$\langle op\text{-}expr\text{-}func\text{-}end \rangle ::= \langle simple\text{-}op\text{-}expr \rangle \text{'\square'} \langle op \rangle \text{'\square'} \langle simple\text{-}func\text{-}expr \rangle$$

$\langle \text{big-op-expr} \rangle ::=$
 $\langle \text{op-expr-line} \rangle (\text{'\n'} \langle \text{indent} \rangle \langle \text{op-expr-line} \rangle)^*$
 $\text{'\n'} \langle \text{indent} \rangle (\langle \text{op-arg} \rangle | \langle \text{simple-op-expr} \rangle | [\langle \text{op-expr-line} \rangle \text{'\u'}] (\langle \text{simple-func-expr} \rangle | \langle \text{big-func-expr} \rangle))$
 $\langle \text{cases-op-expr} \rangle ::= \langle \text{op-expr-line} \rangle (\text{'\n'} \langle \text{indent} \rangle \langle \text{op-expr-line} \rangle)^* (\text{'\n'} \langle \text{indent} \rangle | \text{'\u'}) \langle \text{cases-func-expr} \rangle$
 $\langle \text{op-expr-line} \rangle ::= (\langle \text{op-arg} \rangle | \langle \text{simple-op-expr} \rangle) \text{'\u'} \langle \text{op} \rangle$
 $\langle \text{op-arg} \rangle ::= \langle \text{no-paren-op-arg} \rangle | \langle \text{paren-expr} \rangle$
 $\langle \text{no-paren-op-arg} \rangle ::= \langle \text{basic-expr} \rangle | \langle \text{pre-func} \rangle | \langle \text{post-func} \rangle | \langle \text{pre-func-app} \rangle$
 $\langle \text{op} \rangle ::= \text{'->'} | \text{'<-'} | \text{'o>'} | \text{'<o'} | \text{'^'} | \text{'*'} | \text{'/'} | \text{'+'} | \text{'-'} | \text{'='} | \text{'/='} | \text{'>'} | \text{'<'} | \text{'>='} | \text{'<='} | \text{'\&'} | \text{'|'} | \text{'>'} | \text{'>'}$

3.2.7 Complete Table, Precedence and Associativity

Table 1: The complete table of lcases operators along with their types and their short descriptions.

Operator	Type	Description
->	(A, A => B) => B	Right function application
<-	(A => B, A) => B	Left function application
o>	(A => B, B => C) => (A => C)	Right function composition
<o	(B => C, A => B) => (A => C)	Left function composition
^	(A)ToThe(B)HasType(C) ==> (A, B) => C	General exponentiation
*	(A)And(B)MultiplyTo(C) ==> (A, B) => C	General multiplication
/	(A)Divides(B)To(C) ==> (B, A) => C	General division
+	(A)And(B)AddTo(C) ==> (A, B) => C	General addition
-	(A)SubtractsFrom(B)To(C) ==> (B, A) => C	General subtraction
=	(A)HasEquality ==> (A, A) => Bool	Equality
/=	(A)HasInequality ==> (A, A) => Bool	Inequality
>=	(A)HasGreaterOrEqual ==> (A, A) => Bool	Greater than or equal to
<=	(A)HasLessOrEqual ==> (A, A) => Bool	Less than or equal to
>	(A)CanBeGreaterThan(B) ==> (A, B) => Bool	Greater than
<	(A)CanBeLessThan(B) ==> (A, B) => Bool	Less than
&	(Bool, Bool) => Bool	Boolean operators
; >	(E)IsAnEnvAction ==> (E(A), A => E(B)) => E(B)	Monad bind
;	(E)IsAnEnvAction ==> (E(A), E(B)) => E(B)	Monad then

The order of operations is done from highest to lowest precedence. In the same level of precedence the order is done

from left to right if the associativity is "Left" and from right to left if the associativity is "Right". For the operators that have associativity "None" it is not allowed to place them in the same operator expression. The precedence and associativity of the operators is shown in the table below.

Table 2: The table of precedence and associativity of the lcases operators.

Operator	Precedence	Associativity
->	10 (highest)	Left
<-	9	Right
o> <o	8	Left
^	7	Right
* /	6	Left
+ -	5	Left
= /= > < >= <=	4	None
&	3	Left
	2	Left
; > ;	1	Left

3.3 Function Expressions

$\langle \text{func-expr} \rangle ::= \langle \text{simple-func-expr} \rangle \mid \langle \text{big-func-expr} \rangle \mid \langle \text{cases-func-expr} \rangle$

3.3.1 Regular Function Expressions

- *Examples*

```
a => 17 * a + 42
```

```
(x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1 / 2)
```

- *Description*

Regular function expressions are used to define functions or be part of bigger expressions as anonymous functions. They are comprised by their parameters and their body. Parameter are specified by identifiers. The parameters are either only one, in which case there is no parenthesis, or they are many, in which case they are in parenthesis, separated by commas. The parameters and the body are separated by an arrow ("=>"). The body is an operator expression.

- *Big Function Expressions*

It is possible to stretch a (big) function expression over multiple lines. In that case:

- The function expression must split in a new line after the arrow ("=>") following the parameters.
- Every line after the first must be indented so that it begins at the column where the first character of the parameters was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

$\langle \text{simple-func-expr} \rangle ::= \langle \text{parameters} \rangle \text{ '=>' } \langle \text{simple-func-body} \rangle$

$\langle \text{big-func-expr} \rangle ::= \langle \text{parameters} \rangle \text{ '=>\n' } \langle \text{indent} \rangle (\langle \text{simple-func-body} \rangle \mid \langle \text{big-op-expr} \rangle)$

$\langle \text{parameters} \rangle ::= \langle \text{identifier} \rangle \mid \langle \text{'(' } \langle \text{identifier} \rangle \text{ (' , ' } \langle \text{identifier} \rangle \text{) + '}' } \rangle$

$\langle \text{simple-func-body} \rangle ::= \langle \text{no-paren-op-arg} \rangle \mid \langle \text{simple-op-expr} \rangle \mid \langle \text{op-expr-func-end} \rangle$

3.3.2 "cases" Function Expressions

- *Examples*

```
print_sentimental_bool
: Bool => (EmptyVal)IOAction
= cases =>
  true => print <- "It's true!! :)"
  false => print <- "It's false... :("
```

```
or_type TrafficLight
values green | amber | red
```

```
print_sentimental_traffic_light
: TrafficLight => (EmptyVal)IOAction
= cases =>
```

```

green => print <- "It's green! Let's go!!! :)"
amber => print <- "Go go go, fast!"
red => print <- "Stop right now! You're going to kill us!!"

is_not_red
: TrafficLight => Bool
= cases =>
  green => true
  amber => true
  red => false

is_seventeen_or_forty_two
: Int => Bool
= cases =>
  17 => true
  42 => true
  ... => false

traffic_lights_match
: (TrafficLight, TrafficLight) => Bool
= (cases, cases) =>
  (green, green) => true
  (amber, amber) => true
  (red, red) => true
  ... => false

gcd
: (Int, Int) => Int
= (x, cases) =>
  0 => x
  y => gcd(y, x -> mod <- y)

is_empty
: ListOf(A)s => Bool
= cases =>
  empty => true
  non_empty:anything => false

apply()to_all
: (A => B, ListOf(A)s) => ListOf(B)s
= (f, cases) =>
  empty => empty
  non_empty:list => non_empty:(f <- list.head, apply(f)to_all <- list.tail)

```

- *Description*

"cases" is a keyword that works as a special parameter. The difference is that instead of giving the name "cases" to that parameter, it allows the programmer to pattern match on the possible values of that parameter and return a different result for each particular case.

The "cases" keyword can only be used on parameters that have either one of the basic types (Int, Real, Char, String) or an `or_type` (e.g. Bool, ListOf(A)s).

The last case can be "... => (body of default case)" to capture all remaining cases while dismissing the value (e.g. `is_seventeen_or_forty_two` example), or it can be "some_id => (body of default case)" to capture all remaining cases while being able to use the value with the name "some_id" (e.g. "y" in `gcd` example).

It is possible to use the "cases" keyword in multiple parameters to match on all of them. By doing that, each case represents a particular combination of values for the parameters involved (e.g. `traffic_lights_match` example).

It is also possible to use a "where" expression below a particular case. The "where" expression must be indented two spaces more than the line where that particular case begins.

A function expression that uses the "cases" syntax must contain the "cases" keyword in at least one parameter. The number of matching expressions in all cases must be the same as the number of parameters with the "cases" keyword.

- *Grammar*

$$\begin{aligned} \langle \text{cases-func-expr} \rangle &::= \langle \text{cases-parameters} \rangle \text{ '␣=>' } \langle \text{case} \rangle + \langle \text{end-case} \rangle \\ \langle \text{cases-parameters} \rangle &::= \langle \text{cases-parameter} \rangle \mid \text{ '(' } \langle \text{cases-parameter} \rangle \text{ ('␣' } \langle \text{cases-parameter} \rangle \text{) } + \text{ ')' } \\ \langle \text{cases-parameter} \rangle &::= \langle \text{parameter} \rangle \mid \text{ 'cases' } \\ \\ \langle \text{case} \rangle &::= \text{ '\n' } \langle \text{indent} \rangle \langle \text{matching} \rangle \text{ '␣=>' } \langle \text{case-body} \rangle \\ \langle \text{end-case} \rangle &::= \text{ '\n' } \langle \text{indent} \rangle \text{ ('...' } \mid \langle \text{matching} \rangle \text{) '␣=>' } \langle \text{case-body} \rangle \\ \\ \langle \text{matching} \rangle &::= \langle \text{literal} \rangle \mid \langle \text{identifier} \rangle \mid \langle \text{pre-func} \rangle \langle \text{matching} \rangle \mid \langle \text{tuple-matching} \rangle \mid \langle \text{list-matching} \rangle \\ \langle \text{tuple-matching} \rangle &::= \text{ '(' } \langle \text{matching} \rangle \text{ ('␣' } \langle \text{matching} \rangle \text{) } + \text{ ')' } \\ \langle \text{list-matching} \rangle &::= \text{ '[' } [\langle \text{matching} \rangle \text{ ('␣' } \langle \text{matching} \rangle \text{) }^* [\text{ '␣' , ... }]] \text{ ']' } \\ \\ \langle \text{case-body} \rangle &::= \text{ (} \langle \text{simple-func-body} \rangle \mid \langle \text{big-op-expr} \rangle \text{) } [\langle \text{where-expr} \rangle] \end{aligned}$$

3.4 Value Definitions and "where" Expressions

3.4.1 Value Definitions

- *Examples*

```
foo
  : Int
  = 42

val1, val2, val3
  : Int, Bool, Char
  = 42, true, 'a'

int1, int2, int3
  : all Int
  = 1, 2, 3
```

```
f
  : (Int, Int, Int) => Int
  = (a, b, c) => a + b * c
```

- *Description*

Value definitions are the main building block of lcases programs. To define a new value you give it a name, a type and an expression. The name is in the first line. The second line is indented two spaces more and begins by " : " and continues with the type expression. The third line is indented as the second, begins by " = " and continues with the value expression (which extends to as many lines as needed).

A value definition is either in the first column, where it can be "seen" by all other value definitions, or it is in a "where" expression (see section below), where it can be "seen" by the expression above the "where" and all the other definitions in the same "where" expression.

A value definition can be followed by a "where" expression where intermediate values used in the value expression are defined. In that case, the "where" expression must be indented two spaces more than the "=" line of the value definition.

It is possible to group value definitions together by separating the names, the types and the expressions with commas. This is very useful for not cluttering the program with many definitions for values with small expressions (e.g. constants). When grouping definitions together it is also possible to use the keyword "all" to give the same type to all the values.

- *Grammar*

$\langle \text{value-def} \rangle ::= \langle \text{indent} \rangle \langle \text{identifier} \rangle \backslash \mathbf{n} \langle \text{indent} \rangle \text{ : } \langle \text{type} \rangle \backslash \mathbf{n} \langle \text{indent} \rangle \text{ = } \langle \text{value-expr} \rangle [\langle \text{where-expr} \rangle]$

$\langle \text{value-expr} \rangle ::= \langle \text{no-paren-op-arg} \rangle \mid \langle \text{op-expr} \rangle \mid \langle \text{func-expr} \rangle \mid \langle \text{big-tuple} \rangle \mid \langle \text{big-list} \rangle$

$\langle \text{grouped-value-defs} \rangle ::=$
 $\langle \text{indent} \rangle \langle \text{identifier} \rangle (\text{ ' , ' } \langle \text{identifier} \rangle) +$
 $\text{ ' n ' } \langle \text{indent} \rangle \text{ : } \langle \text{type} \rangle (\text{ ' , ' } \langle \text{type} \rangle) + \mid \text{ ' all ' } \langle \text{type} \rangle)$
 $\text{ ' n ' } \langle \text{indent} \rangle \text{ = } \langle \text{comma-sep-line-exprs} \rangle (\text{ ' n ' } \langle \text{indent} \rangle \text{ ' , ' } \langle \text{comma-sep-line-exprs} \rangle)^*$

3.4.2 "where" Expressions

- *Examples*

```
sort :
  (A)HasOrder ==> ListOf(A)s => ListOf(A)s
  = cases =>
    empty => empty
    non_empty:l => sort(less_l) + l.head + sort(greater_l)
    where
      less_l, greater_l
        : all ListOf(A)s
        = filter_with(x => x < l.head, l.tail)
          , filter_with(x => x >= l.head, l.tail)

tuple_type Coeffs
value (previous, current) : Int x Int

tuple_type GcdAndCoeffs
value (gcd, a, b) : Int x Int x Int
```

```

ext_euc
  : (Int, Int) => GcdAndCoeffs
  = ext_euc_rec((1, 0), (0, 1))

ext_euc_rec
  : (Coeffs, Coeffs, Int, Int) => GcdAndCoeffs
  = (a_coeffs, b_coeffs, x, cases) =>
    0 => (x, a_coeffs.previous, b_coeffs.previous)
    y => ext_euc_rec(next <- a_coeffs, next <- b_coeffs, y, x -> mod <- y)
      where
        next: Coeffs => Coeffs
          = cs => (cs.current, cs.previous - x / y * cs.current)

big_string
  : String
  = s1 + s2 + s3 + s4
  where
    s1, s2, s3, s4 : all String
      = "Hello, my name is Struggling Programmer."
      , "I have tried way too many times to fit a big chunk of text"
      , "inside my program, without it hitting the half-screen mark!"
      , "I am so glad I finally discovered lcases!!!"

```

- *Description*

"where" expressions allow the programmer to use values inside an expression and define them below it. They are very useful for reusing or abbreviating expressions that are specific to a particular definition or case.

A "where" expression begins by a line that only has word "where" in it. It is indented as described in the "Value Definitions" (3.4.1) or "'cases' Function Expressions" (3.3.2) section according to which of the two is above it. The definitions are placed below the "where" line and must have the same indentation.

- *Grammar*

$\langle where\text{-}expr \rangle ::= \backslash n \langle indent \rangle \text{'where'} \backslash n (\langle value\text{-}def \rangle \mid \langle grouped\text{-}value\text{-}defs \rangle) +$

3.5 Indentation and Complete Grammar for Values

3.5.1 Indentation System

The $\langle indent \rangle$ nonterminal is not a normal BNF nonterminal. It is a context sensitive construct that enforces the indentation rules of lcases. It depends on a integer value that shall be named "indentation level" (il). The $\langle indent \rangle$ nonterminal corresponds to $2 * il$ space characters. The indentation level follows the rules below:

Indentation Rules

- At the beginnng: $il = 0$.
- At the end of the first line of a value definition: $il \leftarrow il + 1$. This also applies to grouped value definitions.
- At the end of the third line ("=" line) of a (single) value definition: $il \leftarrow il + 1$.
- At the end of a (single) value definition: $il \leftarrow il - 2$.
- At the end of grouped value definitions: $il \leftarrow il - 1$.

- After the "=>" arrow in a case: $il \leftarrow il + 1$.
- At the end of a case body: $il \leftarrow il - 1$.
- In a cases function expression which does not begin in the "=" line of a value definition:
 - After the arrow "=>" at the end of the paremeters: $il \leftarrow il + 1$.
 - At the end of the whole cases function expression: $il \leftarrow il - 1$.

3.5.2 Complete Grammar for Values

$\langle literal \rangle ::= \langle literal \rangle$

$\langle identifier \rangle ::= [a-z] [a-z_]* ('(' [a-z_]+)^* [[0-9]]$

$\langle paren-expr \rangle ::= '(' \langle op-or-func-expr \rangle ')'$

$\langle op-or-func-expr \rangle ::= \langle simple-op-expr \rangle \mid \langle op-expr-func-end \rangle \mid \langle simple-func-expr \rangle$

$\langle tuple \rangle ::= '(' \langle line-expr \rangle ', _ ' \langle comma-sep-line-exprs \rangle ')'$

$\langle comma-sep-line-exprs \rangle ::= \langle line-expr \rangle (', _ ' \langle line-expr \rangle)^*$

$\langle line-expr \rangle ::= \langle no-paren-op-arg \rangle \mid \langle op-or-func-expr \rangle$

$\langle big-tuple \rangle ::=$
 $\quad '(' \langle line-expr \rangle ['\backslash n' \langle indent \rangle] ', _ ' \langle comma-sep-line-exprs \rangle$
 $\quad ('\backslash n' \langle indent \rangle ', ' \langle comma-sep-line-exprs \rangle)^*$
 $\quad '\backslash n' \langle indent \rangle ')'$

$\langle list \rangle ::= '[' [\langle comma-sep-line-exprs \rangle] ']'$

$\langle big-list \rangle ::= '[' \langle comma-sep-line-exprs \rangle ('\backslash n' \langle indent \rangle ', ' \langle comma-sep-line-exprs \rangle)^* '\backslash n' \langle indent \rangle ']'$

$\langle paren-func-app \rangle ::=$
 $\quad [\langle arguments \rangle] \langle identifier-with-arguments \rangle [\langle arguments \rangle]$
 $\quad \mid \langle arguments \rangle \langle identifier \rangle [\langle arguments \rangle]$
 $\quad \mid \langle identifier \rangle \langle arguments \rangle$

$\langle arguments \rangle ::= '(' \langle comma-sep-line-exprs \rangle ')'$

$\langle identifier-with-arguments \rangle ::=$
 $\quad [a-z] [a-z_]* ('(' [a-z_]+)^* \langle arguments \rangle [a-z_]+ (('(' \mid \langle arguments \rangle) [a-z_]+)^* [[0-9]]$

$\langle pre-func \rangle ::= \langle identifier \rangle ':'$

$\langle pre-func-app \rangle ::= \langle pre-func \rangle (\langle basic-expr \rangle \mid \langle paren-expr \rangle \mid \langle pre-func-app \rangle)$

$\langle \text{basic-expr} \rangle ::= \langle \text{literal} \rangle \mid \langle \text{identifier} \rangle \mid \langle \text{tuple} \rangle \mid \langle \text{list} \rangle \mid \langle \text{paren-func-app} \rangle \mid \langle \text{post-func-app} \rangle$

$\langle \text{post-func} \rangle ::= \text{'.'} \langle \text{identifier} \rangle$

$\langle \text{post-func-app} \rangle ::= (\langle \text{paren-expr} \rangle \mid \langle \text{basic-expr} \rangle) \langle \text{post-func} \rangle$

$\langle \text{op-expr} \rangle ::= \langle \text{simple-op-expr} \rangle \mid \langle \text{op-expr-func-end} \rangle \mid \langle \text{big-op-expr} \rangle \mid \langle \text{cases-op-expr} \rangle$

$\langle \text{simple-op-expr} \rangle ::= \langle \text{op-arg} \rangle (\text{'\u2192'} \langle \text{op} \rangle \text{'\u2192'} \langle \text{op-arg} \rangle) +$

$\langle \text{op-expr-func-end} \rangle ::= \langle \text{simple-op-expr} \rangle \text{'\u2192'} \langle \text{op} \rangle \text{'\u2192'} \langle \text{simple-func-expr} \rangle$

$\langle \text{big-op-expr} \rangle ::=$
 $\langle \text{op-expr-line} \rangle (\text{'\n'} \langle \text{indent} \rangle \langle \text{op-expr-line} \rangle)^*$
 $\text{'\n'} \langle \text{indent} \rangle (\langle \text{op-arg} \rangle \mid \langle \text{simple-op-expr} \rangle \mid [\langle \text{op-expr-line} \rangle \text{'\u2192'}] (\langle \text{simple-func-expr} \rangle \mid \langle \text{big-func-expr} \rangle))$

$\langle \text{cases-op-expr} \rangle ::= \langle \text{op-expr-line} \rangle (\text{'\n'} \langle \text{indent} \rangle \langle \text{op-expr-line} \rangle)^* (\text{'\n'} \langle \text{indent} \rangle \mid \text{'\u2192'}) \langle \text{cases-func-expr} \rangle$

$\langle \text{op-expr-line} \rangle ::= (\langle \text{op-arg} \rangle \mid \langle \text{simple-op-expr} \rangle) \text{'\u2192'} \langle \text{op} \rangle$

$\langle \text{op-arg} \rangle ::= \langle \text{no-paren-op-arg} \rangle \mid \langle \text{paren-expr} \rangle$

$\langle \text{no-paren-op-arg} \rangle ::= \langle \text{basic-expr} \rangle \mid \langle \text{pre-func} \rangle \mid \langle \text{post-func} \rangle \mid \langle \text{pre-func-app} \rangle$

$\langle \text{op} \rangle ::= \text{'->'} \mid \text{'<-'} \mid \text{'o>'} \mid \text{'<o'} \mid \text{'^'} \mid \text{'*'} \mid \text{'/'} \mid \text{'+'} \mid \text{'-'} \mid \text{'='} \mid \text{'/='} \mid \text{'>'} \mid \text{'<'} \mid \text{'>='} \mid \text{'<='} \mid \text{'\&'} \mid \text{'|'} \mid \text{'>;'} \mid \text{'<;'}$

$\langle \text{func-expr} \rangle ::= \langle \text{simple-func-expr} \rangle \mid \langle \text{big-func-expr} \rangle \mid \langle \text{cases-func-expr} \rangle$

$\langle \text{simple-func-expr} \rangle ::= \langle \text{parameters} \rangle \text{'\u2192\u2192'} \langle \text{simple-func-body} \rangle$

$\langle \text{big-func-expr} \rangle ::= \langle \text{parameters} \rangle \text{'\u2192\u2192'} \text{'\n'} \langle \text{indent} \rangle (\langle \text{simple-func-body} \rangle \mid \langle \text{big-op-expr} \rangle)$

$\langle \text{parameters} \rangle ::= \langle \text{identifier} \rangle \mid \text{'('} \langle \text{identifier} \rangle (\text{'\u2192\u2192'} \langle \text{identifier} \rangle) + \text{'\u2192\u2192'}$

$\langle \text{simple-func-body} \rangle ::= \langle \text{no-paren-op-arg} \rangle \mid \langle \text{simple-op-expr} \rangle \mid \langle \text{op-expr-func-end} \rangle$

$\langle \text{cases-func-expr} \rangle ::= \langle \text{cases-parameters} \rangle \text{'\u2192\u2192'} \langle \text{case} \rangle + \langle \text{end-case} \rangle$

$\langle \text{cases-parameters} \rangle ::= \langle \text{cases-parameter} \rangle \mid \text{'('} \langle \text{cases-parameter} \rangle (\text{'\u2192\u2192'} \langle \text{cases-parameter} \rangle) + \text{'\u2192\u2192'}$

$\langle \text{cases-parameter} \rangle ::= \langle \text{parameter} \rangle \mid \text{'cases'}$

$\langle \text{case} \rangle ::= \text{'\n'} \langle \text{indent} \rangle \langle \text{matching} \rangle \text{'\u2192\u2192'} \langle \text{case-body} \rangle$

$\langle \text{end-case} \rangle ::= \text{'\n'} \langle \text{indent} \rangle (\text{'...'} \mid \langle \text{matching} \rangle) \text{'\u2192\u2192'} \langle \text{case-body} \rangle$

$\langle matching \rangle ::= \langle literal \rangle \mid \langle identifier \rangle \mid \langle pre-func \rangle \langle matching \rangle \mid \langle tuple-matching \rangle \mid \langle list-matching \rangle$
 $\langle tuple-matching \rangle ::= \text{'('} \langle matching \rangle \text{'('} \langle matching \rangle \text{)+ '}'$
 $\langle list-matching \rangle ::= \text{'['} [\langle matching \rangle \text{'('} \langle matching \rangle \text{)* [' , ... ']] '}'$
 $\langle case-body \rangle ::= (\text{'\u'} \mid \text{'\n'} \langle indent \rangle) (\langle simple-func-body \rangle \mid \langle big-op-expr \rangle) [\langle where-expr \rangle]$
 $\langle value-def \rangle ::= \langle indent \rangle \langle identifier \rangle \text{'\n'} \langle indent \rangle \text{'::\u'} \langle type \rangle \text{'\n'} \langle indent \rangle \text{'=\u'} \langle value-expr \rangle [\langle where-expr \rangle]$
 $\langle value-expr \rangle ::= \langle no-paren-op-arg \rangle \mid \langle op-expr \rangle \mid \langle func-expr \rangle \mid \langle big-tuple \rangle \mid \langle big-list \rangle$
 $\langle grouped-value-defs \rangle ::=$
 $\quad \langle indent \rangle \langle identifier \rangle (\text{' ,\u'} \langle identifier \rangle) +$
 $\quad \text{'\n'} \langle indent \rangle \text{'::\u'} (\langle type \rangle (\text{' ,\u'} \langle type \rangle) + \mid \text{'all'} \langle type \rangle)$
 $\quad \text{'\n'} \langle indent \rangle \text{'=\u'} \langle comma-sep-line-exprs \rangle (\text{'\n'} \langle indent \rangle \text{' , ' } \langle comma-sep-line-exprs \rangle) *$
 $\langle where-expr \rangle ::= \text{'\n'} \langle indent \rangle \text{'where\n'} (\langle value-def \rangle \mid \langle grouped-value-defs \rangle) +$

4 Language Description: Types and Type Logic

4.1 Types

The constructs regarding types are two: **type expressions** and **type definitions**.

Type expressions are divided into the following categories:

- Type Identifiers
- Type Variables
- Function Types
- Product Types
- Type Application Types
- Conditional Types

which are described in the following section.

The grammar of a type expression is:

$\langle type \rangle ::= [\langle condition \rangle] \langle simple-type \rangle$
 $\langle simple-type \rangle ::= \langle type-id \rangle \mid \langle type-var \rangle \mid \langle func-type \rangle \mid \langle prod-type \rangle \mid \langle type-app \rangle$

Type definitions are divided into **tuple_type** definitions and **or_type** definitions which are described in the sections 4.1.2 and 4.1.3 respectively.

The grammar of a type definition is:

$\langle type-def \rangle ::= \langle tuple-type-def \rangle \mid \langle or-type-def \rangle$

4.1.1 Type Expressions

Type Identifiers

- *Examples*

```
Int
Real
Char
String
FunnyType
MyDefinedType
```

- *Description*

A type identifier is either the name of a basic type (Int, Real, Char, String) or the name of some defined type that has no type parameters. It begins with a capital letter and is followed by one or more capital or lowercase letters.

- *Grammar*

$\langle \text{type-id} \rangle ::= [\text{A-Z}] [\text{A-Za-z}]^+$

Type Variables

- *Examples*

```
A
B
C
X
Y
Z
```

- *Examples of type variables inside bigger type expressions*

```
A => A
(A => B, B => C) => (A => C)
((A, A) => A, A, ListOf(A)s) => A
```

- *Description*

Type Variables are used inside larger type expressions of polymorphic types. A polymorphic type is a type where any function of that type can be used as a function of any type that corresponds to substituting every type variable of the polymorphic type with a particular type. The easiest example of a polymorphic type is the type of the identity function where we have:

```
id
  : A => A
  = x => x
```

```
id(1)
  : Int
  where A is substituted by Int and id gets the type Int => Int
```

```
id("Hello")
  : String
  where A is substituted by String and id gets the type String => String
```

A type variable is a single capital letter.

- *Grammar*

$\langle \text{type-var} \rangle ::= [A-Z]$

Function Types

- *Examples*

```
String => String
Real => Int
A => A
Int x Int => Int
(Real, Real, Real) => Real
(A => B, B => C) => (A => C)
(Int => Int) => Int
```

- *Description*

A function type expression is comprised of the expressions of the types of the parameters and the expression of the type of the result, separated by the arrow "=>". If there are more than one parameters, the expressions of their types are inside parentheses and separated by commas. If there is only one parameter, the expression of its type is put in parentheses only if it is a function type. The same applies to the type of the result.

- *Grammar*

$\langle \text{func-type} \rangle ::= \langle \text{param-types-expr} \rangle \text{ '=>' } \langle \text{one-type} \rangle$
 $\langle \text{param-types-expr} \rangle ::= \langle \text{one-type} \rangle \mid \langle \text{'('} \langle \text{simple-type} \rangle \langle \text{' , ' } \langle \text{simple-type} \rangle \rangle \text{'})' \rangle$
 $\langle \text{one-type} \rangle ::= \langle \text{type-id} \rangle \mid \langle \text{type-var} \rangle \mid \langle \text{prod-type} \rangle \mid \langle \text{type-app} \rangle \mid \langle \text{'('} \langle \text{func-type} \rangle \langle \text{')' } \rangle$

Product Types

- *Examples*

```
Int x Int
Real x Real x Real
Int x Real x String
ListOf(Int)s x (Int x ListOf(String)s)
(Int => Int) x (Int x Real) x (Real => String)
```

- *Description*

Product types are the types of tuples. They are comprised of the expressions of the types of the fields separated by the string " x " (space 'x' space) because 'x' is very similar the symbol used in the cartesian product. If any of the fields has a product or a function type then the corresponding type expression must be inside parentheses.

- *Grammar*

$\langle \text{prod-type} \rangle ::= \langle \text{field-type} \rangle \langle \text{' x ' } \langle \text{field-type} \rangle \rangle \text{'+'}$
 $\langle \text{field-type} \rangle ::= \langle \text{type-id} \rangle \mid \langle \text{type-var} \rangle \mid \langle \text{type-app} \rangle \mid \langle \text{'('} \langle \text{func-type} \rangle \mid \langle \text{prod-type} \rangle \rangle \langle \text{')' } \rangle$

Type Application Types

- *Examples*

```
Possibly(Int)
ListOf(Real)s
TreeOf(String)s
Error(String)OrResult(Int)
ListOf(Int => Int)s
ListOf(A)s
```

- *Description*

Type application types are types that are produced by passing arguments to a type function generated by a `tuple_type` or an `or_type` definition. For example, given the definition of `ListOf(A)s`:

```
or_type ListOf(A)s
values non_empty:NonEmptyListOf(A)s | empty
```

We have that `ListOf()`s is a type function that receives one type parameter and returns a resulting type. For example `ListOf(Int)s` is the result of passing the type argument `Int` to `ListOf()`s.

Type application types have the same form as the name in the `tuple_type` or `or_type` definition, with the difference that any of the type parameters can be substituted with an argument type expression.

- *Grammar*

```
⟨type-app⟩ ::=
  [ ⟨types-in-paren⟩ ] ⟨type-id-with-args⟩ [ ⟨types-in-paren⟩ ]
  |   ⟨types-in-paren⟩ ⟨type-id⟩ [ ⟨types-in-paren⟩ ]
  |   ⟨type-id⟩ ⟨types-in-paren⟩

⟨type-id-with-args⟩ ::= ⟨type-id⟩ ( ⟨types-in-paren⟩ [A-Za-z]+ )+

⟨types-in-paren⟩ ::= ‘(’ ⟨simple-type⟩ ( ‘,’ ‘⟨simple-type⟩ ’)* ‘)’
```

Conditional Types

- *Examples*

```
(A)HasEquality ==> (A, A) => Bool
(A)And(B)AddTo(C) ==> (A, B) => C
(A)Is(B)sFirst ==> B => A
(T)HasStringRepr ==> T => String
(E)IsAnEnvAction ==> (E(A), A => E(B)) => E(B)
```

- *Description*

Conditional types are the types of values that are polymorphic not because of their structure but because they have been defined (seperately) for many different combinations of types (a.k.a. ad hoc polymorphism). They are comprised of a condition and a "simple" type (i.e. a type without a condition) which are seperated by the arrow "`==>`". The condition is a type proposition which refers to type variables inside the "simple" type and it must hold whenever the polymorphic value of that type is used. For example:

```
get_first
  : (A)Is(B)sFirst => B => A
```

can be used as follows:

```
pair, triple, list
  : Int x String, Real x Char x Int, ListOf(String)s
  = (42, "The answer to everything"), (3.14, 'a', 1), ["Hi!", "Hello", Heeey"]

>> pair -> get_first
  : Int
  = 42
>> triple -> get_first
  : Real
  = 3.14
>> list -> get_first
  : String
  = "Hi!"
```

and that is because the following propositions hold:

```
(Int)Is(Int x String)sFirst
(Real)Is(Real x Char x Int)sFirst
(String)Is(ListOf(String)s)sFirst
```

which it turn means that the function `get_first` has been defined for these combinations of types. For more on how conditions, propositions and ad hoc polymorphism works, see the "Type Logic" section (4.2).

- *Grammar*

As described in the beginning of this "Types" section the grammar of a type is:

```
⟨type⟩ ::= [ ⟨condition⟩ ] ⟨simple-type⟩
⟨simple-type⟩ ::= ⟨type-id⟩ | ⟨type-var⟩ | ⟨func-type⟩ | ⟨prod-type⟩ | ⟨type-app⟩
```

And therefore here only the grammar of the condition must be written:

```
⟨condition⟩ ::= ⟨prop-app⟩ '⊢==>⊢'
```

4.1.2 Type Definitions: Tuple Types

Example Definitions

```
tuple_type Name
value (first_name, last_name) : String x String

tuple_type Date
value (day, month, year) : Int x Int x Int

tuple_type MathematicianInfo
value (name, nationality, date_of_birth) : Name x String x Date

tuple_type TreeOf(A)s
values (root, subtrees) : A x ListOf(TreeOf(A)s)s

tuple_type Indexed(T)
value (index, val) : Int x T
```

Usage Examples

```
euler_info
  : MathematicianInfo
  = (("Leonhard", "Euler"), "Swiss", (15, 4, 1707))

name_to_string
  : Name => String
  = n => "\nFirst Name: " + n.first_name + "\nLast Name: " + n.last_name

print_name_and_nationality
  : ClientInfo => (EmptyVal)IOAction
  = ci => print(ci.name -> name_to_string + "\nNationality: " + ci.nationality)

sum_nodes
  : TreeOf(Int)s => Int
  = tree => tree.root + tree.subtrees -> apply(sum_nodes)to_all -> sum_list
```

Description

A tuple type is equivalent to a product type with a new name and names for the fields for convinience. A tuple type generates postfix functions for all of the fields by using a '.' before the name of the field. For example the MathematicianInfo type above generates the following functions:

```
.name
  : MathematicianInfo => Name
.nationality
  : MathematicianInfo => String
.date_of_birth
  : MathematicianInfo => Date
```

These functions are named "postfix functions" because they can be appended to their argument.

Definition Grammar

```
 $\langle tuple\_type\_def \rangle ::= 'tuple\_type_{\square}' \langle type\_name \rangle '\backslash nvalue_{\square}' '(' \langle identifier \rangle (' ,_{\square} \langle identifier \rangle )^* ') ' '\square :_{\square}' \langle prod\_type \rangle$ 
 $\langle type\_name \rangle ::= [ \langle params\_in\_paren \rangle ] ( \langle type\_id \rangle | \langle type\_id\_with\_params \rangle ) [ \langle params\_in\_paren \rangle ]$ 
 $\langle type\_id\_with\_params \rangle ::= \langle type\_id \rangle ( \langle params\_in\_paren \rangle [A-Za-z]^+ ) +$ 
 $\langle params\_in\_paren \rangle ::= '(' \langle type\_var \rangle (' , ' \langle type\_var \rangle )^* ')'$ 
```

4.1.3 Type Definitions: Or Types

Examples

```
or_type Bool
values true | false

or_type Possibly(A)
values the_value:A | no_value

// needed tuple_type for ListOf(A)s
tuple_type NonEmptyListOf(A)s
value (head, tail) : A x ListOf(A)s
```

```
or_type ListOf(A)s
values non_empty:NonEmptyListOf(A)s | empty
```

```
or_type Error(A)OrResult(B)
values error:A | result:B
```

Usage

```
is_empty
  : ListOf(A)s => Bool
  = cases =>
    empty => true
    non_empty:anything => false
```

```
get_head
  : ListOf(A)s => Possibly(A)
  = cases =>
    empty => no_value
    non_empty:list => the_value:list.head
```

```
sum_list
  : ListOf(Int)s => Int
  = cases
    empty => 0
    non_empty:l => l.head + sum_list(l.tail)
```

```
print_err_or_res
  : Error(A)OrResult(B) => (EmptyVal)IOAction
  = cases =>
    error:e => print("Error occurred: " + e -> to_string)
    result:r => print("All good! The result is: " + r -> to_string)
```

Description

Values of an `or_type` are one of many cases. Some cases have other values inside. The cases which have other values inside are followed by a semicolon and the type of the internal value. The same syntax can be used for matching that particular case in a function using the "cases" syntax, with the difference that after the colon, we write the name given to the value inside. An `or_type` definition automatically creates prefix functions for each case with an internal value. For example, for the case "`non_empty`" of a list, the function "`non_empty:`" is automatically created from the definition for which we can say:

```
non_empty:
  : NonEmptyListOf(A)s => ListOf(A)s
```

Similarly:

```
the_value:
  : A => Possibly(A)
```

These functions are called "prefix functions" because they are prepended to their argument. For example:

```
non_empty_1
  : NonEmptyListOf(Int)s
  = (1, [2, 3, 4])
```

```

1
: ListOf(Int)s
= non_empty:non_empty_1

```

These functions can be used like any other function as arguments to other functions. For example:

```

non_empty_ls_to_ls
: ListOf(NonEmptyListOf(A)s)s => ListOf(ListOf(A)s)s
= apply(non_empty:)to_each

```

Definition Grammar

```

⟨or-type-definition⟩ ::=
  'or_type' ⟨type-name⟩
  'nvalues' ⟨identifier⟩ [ ':' ⟨type⟩ ] ( 'l' ⟨identifier⟩ [ ':' ⟨type⟩ ] ) *

```

4.2 Type Logic

The constructs regarding type logic are two: **type propositions** and **type theorems**. Type propositions are divided into two categories: **atomic type propositions** and **compound type propositions** which are described in the paragraphs below. The grammar of a type proposition definition is:

```

⟨type-prop-def⟩ ::= ⟨atom-prop-def⟩ | ⟨comp-prop-def⟩

```

Type theorems are described in section 4.2.3.

4.2.1 Proposition Definitions: Atomic Type Propositions

- *Examples*

```

type_proposition (T)Has_String_Represantion
value
  to_string : T => String

type_proposition (A)Is(B)s_First
value
  first : B => A

type_proposition (T)Has_A_Wrapper
value
  wrapper : A => T(A)

type_proposition (T)Has_Internal_Application
value
  apply_internally : (A => B, T(A)) => T(B)

type_proposition (T)Has_Wrapped_Internal_Application
value
  app_wrapped_f_internally : (T(A => B), T(A)) => T(B)

```

- *Description*

Atomic type propositions are propositions that linked to a ad hoc polymorphic function.

- *Grammar*

```

⟨atom-prop-def⟩ ::= 'type_proposition' ⟨prop-name-expr⟩ 'nvalue' n' ⟨identifier⟩ ' : ' ⟨simple-type⟩
⟨prop-name-expr⟩ ::= ⟨prop-name-expr⟩

```


4.2.2 Proposition Definitions: Compound Type Propositions

- *Examples*
- *Description*

- *Grammar*

$\langle grammar \rangle ::= \langle grammar \rangle$

4.2.3 Type Theorems

4.3 Complete Grammar for Types and Type Logic

5 Predefined

5.1 Constants

5.2 Functions

5.3 Types

5.4 Type Propositions

6 Parser implimentation

The parser was implemented using the parsec library.

6.1 AST Types

6.2 Parsers

7 Translation to Haskell

8 Running Examples

9 Conclusion