

Lambda Cases (lcases)

Dimitris Saridakis

Contents

1	Introduction	3
2	Language Description: General	3
2.1	Program Structure	3
2.2	Keywords	4
3	Language Description: Values	5
3.1	Basic Expressions	5
3.1.1	Literals and Identifiers	5
3.1.2	Parenthesis, Tuples and Lists	6
3.1.3	Parenthesis Function Application	8
3.1.4	Prefix and Postfix Functions	9
3.2	Operators	12
3.2.1	Function Application and Function Composition Operators	12
3.2.2	Arithmetic, Comparison and Boolean Operators	13
3.2.3	Environment Action Operators	15
3.2.4	Operator Expressions	17
3.2.5	Complete Table, Precedence and Associativity	19
3.3	Function Expressions	21
3.3.1	Regular Function Expressions	21
3.3.2	"cases" Function Expressions	22
3.4	Value Definitions and "where" Expressions	23
3.4.1	Value Definitions	23
3.4.2	"where" Expressions	24
3.5	Indentation and Complete Grammar for Values	25
3.5.1	Indentation System	25
3.5.2	Complete Grammar for Values	26
4	Language Description: Types and Type Logic	28
4.1	Types	28
4.1.1	Type Expressions	28
4.1.2	Type Definitions	32
4.1.3	Type Nicknames	35
4.2	Type Logic	35
4.2.1	Proposition Definitions	35
4.2.2	Theorems	38
4.3	Complete Grammar for Types and Type Logic	40
5	Language Description: Predefined	42
5.1	Values	42
5.2	Types	43
5.3	Type Propositions	43

6	Parser implementation	44
6.1	AST Types	45
6.2	Parsers	45
7	Translation to Haskell	45
8	Running Examples	45
9	Conclusion	45

1 Introduction

Haskell is a delightful language. Yet, for some reason, it doesn't seem to have its rightful place in terms of popularity in industry. Why is it so? Is it inherently hard to learn and therefore only the brave dare to use it, or could it be that the syntax is perplexing to the amateur eye? It is my belief that with some syntax changes that give a greater familiarity to the new user, there would be no language more compelling than (the new) Haskell. In an attempt to achieve that familiarity, I present some (hopefully useful) new syntax, of which some is closer to the imperative/OOP style (to attract more already experienced programmers from these languages), some is closer to mathematics (in which most programmers should be experienced) and some is closer to natural language (in which we are all already experienced).

2 Language Description: General

2.1 Program Structure

An lcases program consists of a set of definitions, type nicknames and theorems. Definitions are split into value definitions, type definitions and type proposition definitions. Theorems are proven type propositions. Functions as well as "Environment Actions" (see section 3.2.3) are also considered values. The definition of the "main" value determines the program's behaviour.

Program example: Euclidean Algorithm

```
gcd : Int^2 => Int
  = (x, cases)
    0 => x
    y => gcd(y, x -> mod <- y)

read_two_ints : (Int^2)FromIO
  = print <- "Please give me 2 ints";
    get_line ;> split_to_words o> cases
      [x, y] => wrap_with_io(from_string(x), from_string(y))
      ... => show_err("You didn't give me 2 ints")

tuple_type NumsAndGcd
value
  (x, y, gcd) : Int^3

print_gcd_message : NumsAndGcd => IO
  = nums_gcd => print(message)
  where
    message : String
      = "The GCD of " + nums_gcd.x + " and " + nums_gcd.y + " is = " + nums_gcd.gcd

main : IO
  = read_two_ints ;> (i1, i2) => print_gcd_message(i1, i2, gcd(i1, i1))
```

Program grammar

```
<program> ::= <nl>* <program-part> ( <nl> <nl> <program-part> )* <spaces>

<program-part> ::= <value-def> | <grouped-value-defs> | <type-def> | <t-nickname> | <type-prop-def> | <type-theo>

<nl> :: ( '␣' | '\t' ) * '\n'
```

2.2 Keywords

The lcases keywords are the following:

`cases` `where` `all` `tuple_type` `value` `or_type` `values`
`type_proposition` `equivalent` `type_theorem` `proof`

Each keyword's functionality is described in the respective section shown in the table below:

Keyword	Section
<code>cases</code>	3.3 Function Expressions
<code>where</code> <code>all</code>	3.4 Value Definitions
<code>tuple_type</code> <code>value</code> <code>or_type</code> <code>values</code> <code>type_nickname</code>	4.1 Type Definitions
<code>type_proposition</code> <code>value</code> <code>equivalent</code> <code>type_theorem</code> <code>proof</code>	4.2 Type Logic

The "cases" and "where" keywords are also reserved words. Therefore, even though they can be generated by the "identifiers" grammar, they cannot be used as identifiers (see "Literals and Identifiers" section 3.1.1).

3 Language Description: Values

3.1 Basic Expressions

3.1.1 Literals and Identifiers

Literals

- *Examples*

```
1 2 17 42 -100
1.61 2.71 3.14 -1234.567
'a' 'b' 'c' 'x' 'y' 'z' '.' ',' '\n'
"Hello World!" "What's up, doc?" "Alrighty then!"
```

- *Description*

There are literals for the four basic types: Int, Real, Char, String. These are the usual integers, real numbers, characters and strings.

- *Grammar*

$\langle literal \rangle ::= \langle int-lit \rangle \mid \langle real-lit \rangle \mid \langle char-lit \rangle \mid \langle string-lit \rangle$

Identifiers

- *Examples*

```
x y z
a1 a2 a3
unnecessarily_long_identifier
self_referencing_identifier
apply()to_all
```

- *Description*

An identifier is the name of a value or a parameter. It is used in the definition of a value and in expressions that use that value, or in the parameters of a function and in the body of that function.

An identifier starts with a lower case letter and is followed by lower case letters or underscores. It is also possible to have pairs of parentheses in the middle of an identifier (see "Parenthesis Function Application" section 3.1.3 for why this can be useful). Finally, an identifier can be ended with a digit.

- *Grammar*

$\langle identifier \rangle ::= [a-z] [a-z_]* ('(' [a-z_]+ ')')^* [[0-9]]$

Even though the "cases" and "where" keywords can be generated by this grammar, they cannot be used as identifiers.

3.1.2 Parenthesis, Tuples and Lists

Parenthesis

- *Examples*

```
(1 + 2)
(((1 + 2) * 3) ^ 4)
(val -> (x => f(x) + 1) -> to_string -> (s => "f(val) + 1 is: " + s))
(do(3)times <- (get_line ;> line => print("Line is: " + line)))
```

- *Description*

An expression is put in parenthesis to prioritize it or isolate it in a bigger (operator) expression. The expressions inside parenthesis are operator or function expressions.

Parenthesis expressions cannot extend over multiple lines. For expressions that extend over multiple lines new values must be defined.

- *Grammar*

$\langle \text{paren-expr} \rangle ::= '(\langle \text{line-op-expr} \rangle \mid \langle \text{line-func-expr} \rangle)'$

Tuples

- *Examples*

```
(1, "What's up, doc?")
(2, "Alrighty then!", 3.14)
(x, y, z, w)
(1, my_function, (x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1/2))
```

- *Description*

Tuples are used to group many values (of possibly different types) into one. The type of a tuple can be either the product of the types of the fields or a defined `tuple_type` which is equivalent to the aforementioned product type (see "Tuple Types" in section 4.1.2 for details). For example, the type of the second tuple above could be:

```
Int x String x Real
```

or:

```
MyType
```

assuming "MyType" has been defined in a similar way to the following:

```
tuple_type MyType
value
  (my_int, my_string, my_real) : Int x String x Real
```

- *Big Tuples*

Example

```

my_big_tuple
  : String x Int x Real x String x String x (String x Real x Real)
  = ( "Hey, I'm the first field and I'm also a relatively big string."
    , 42, 3.14, "Hey, I'm the first small string", "Hey, I'm the second small string"
    , ("Hey, I'm a string inside the nested tuple", 2.71, 1.61)
    )

```

Description

It is possible to stretch a (big) tuple expression over multiple lines (only) in a separate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '(' is after the "=" part of the value definition and the first field must be in the same line.
- The tuple can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '(' character was in the first line.
- The tuple must be ended by a line that only contains the ')' character and is also indented so that the ')' is in same column where the '(' character was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

```

⟨tuple⟩ ::= '(' ⟨line-expr⟩ ⟨comma⟩ ⟨line-exprs⟩ ')'
⟨line-exprs⟩ ::= ⟨line-expr⟩ ( ⟨comma⟩ ⟨line-expr⟩ )*
⟨line-expr⟩ ::= ⟨no-paren-op-arg⟩ | ⟨line-op-expr⟩ | ⟨line-func-expr⟩
⟨comma⟩ ::= ',' [ '␣' ]

⟨big-tuple⟩ ::=
  '(' [ '␣' ] ⟨line-expr⟩ [ ⟨nl⟩ ⟨indent⟩ ] ⟨comma⟩ ⟨line-exprs⟩
  ( ⟨nl⟩ ⟨indent⟩ ⟨comma⟩ ⟨line-exprs⟩ )* ⟨nl⟩ ⟨indent⟩ ')'

```

Lists

- *Examples*

```

[1, 2, 17, 42, -100]
[1.61, 2.71, 3.14, -1234.567]
["Hello World!", "What's up, doc?", "Alrighty then!"]
[x => x + 1, x => x + 2, x => x + 3]
[x, y, z, w]

```

- *Description*

Lists are used to group many values of the same type into one. The type of the list is ListOf(A)s where A is the type of every value inside. Therefore, the types of the first four examples are:

```

ListOf(Int)s
ListOf(Real)s
ListOf(String)s
(A)And(Int)Add_To(B) ==> ListOf(A => B)s

```

And the last list is only legal if x, y, z and w all have the same type. Assuming they do and it's the type T, the type of the list is:

ListOf(T)s

- *Big Lists*

Example

```
my_big_list : ListOf(Int => IO)s
= [ x => print("I'm the first function and x + 1 is: " + (x + 1))
  , x => print("I'm the second function and x + 2 is: " + (x + 2))
  , x => print("I'm the third function and x + 3 is: " + (x + 3))
  ]
```

Description

It is possible to stretch a (big) list expression over multiple lines (only) in a separate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '[' is after the "=" part of the value definition and the first element must be in the same line.
 - The list can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '[' character was in the first line.
 - The tuple must be ended by a line that only contains the ']' character and is also indented so that the ']' is in same column where the '[' character was in the first line.
 - The precise indentation rules are described in the section "Indentation System" 3.5.1.
- *Grammar*

$\langle list \rangle ::= '[' [\langle line\text{-}exprs \rangle] ']'$

$\langle big\text{-}list \rangle ::= '[' ['\lceil'] \langle line\text{-}exprs \rangle (\langle nl \rangle \langle indent \rangle \langle comma \rangle \langle line\text{-}exprs \rangle)^* \langle nl \rangle \langle indent \rangle ']'$

3.1.3 Parenthesis Function Application

- *Examples*

```
f(x)
f(x, y, z)
(x)to_string
apply(f)to_all
apply(f)to_all(1)
```

- *Description*

Function application in lcases can be done in many different ways in an attempt to maximize readability. In this section, we discuss the ways function application can be done with parenthesis.

In the first two examples, we have the usual mathematical function application which is also used in most programming languages and should be familiar to the reader, i.e. function application is done with the arguments of the function in parenthesis separated by commas and **appended** to the function identifier.

We extend this idea by allowing the arguments to be **prepended** to the function identifier (third example). Finally, it is also possible to have the arguments **inside** the function identifier provided the function has been **defined with parentheses inside the identifier**. For example, below is the definition of "apply()to_all":


```

apply()to_all : (A => B) x ListOf(A)s => ListOf(B)s
= (f, cases)
  empty_l => empty_l
  non_empty_l:l => non_empty_l:(f <- l.head, apply(f)to_all <- l.tail)

```

The actual definition doesn't matter at this point, what matters is that the identifier is "apply()to_all" with the parentheses **included**. This is very useful for defining functions where the argument in the middle makes the function application look and sound more like natural language.

It is possible to have many parentheses pairs in a single function application (last example). The arguments are always inserted to the function from **left to right**. Therefore, when multiple parentheses pairs are present the arguments of the leftmost parentheses are inserted first then the next ones to the right and so on.

- *Grammar*

```

⟨paren-func-app⟩ ::=
  [ ⟨arguments⟩ ] ⟨identifier-with-arguments⟩ [ ⟨arguments⟩ ]
  |   ⟨arguments⟩ ⟨identifier⟩ [ ⟨arguments⟩ ]
  |   ⟨identifier⟩ ⟨arguments⟩

⟨arguments⟩ ::= '(' ⟨line-exprs⟩ ')'

⟨identifier-with-arguments⟩ ::=
  [a-z] [a-z_]* ( '(' [a-z_]+ ) * ⟨arguments⟩ [a-z_]+ ( ( '(' | ⟨arguments⟩ ) [a-z_]+ ) * [ [0-9] ]

```

3.1.4 Prefix and Postfix Functions

Prefix Functions

- *Examples*

```

the_value:1
non_empty_l:l
error:e
result:r
apply(the_value:)to_all

```

- *Description*

Prefix functions are automatically generated from `or_type` definitions (see "Or Types" in section 4.1.2). They are functions that convert a value of a particular type to a value that is a case of an `or_type` and has values of this type inside. For example in the first example above we have:

```

1 : Int
the_value:1 : Possibly(Int)

```

Where the function `thevalue:` is automatically generated from the definition of the `Possibly` type:

```

or_type Possibly(A)
values
  the_value:A | no_value

```

And it has the type `A => Possibly(A)`.

These functions are called prefix functions because they are prepended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function `the_value:` is an argument of the function `apply()to_all`. Prefix functions always end with a colon.

- *Grammar*

$\langle pre\text{-}func \rangle ::= \langle identifier \rangle \text{'.'}$
 $\langle pre\text{-}func\text{-}app \rangle ::= \langle pre\text{-}func \rangle (\langle basic\text{-}expr \rangle | \langle paren\text{-}expr \rangle | \langle pre\text{-}func\text{-}app \rangle | \langle post\text{-}func\text{-}app \rangle)$
 $\langle basic\text{-}expr \rangle ::= \langle literal \rangle | \langle identifier \rangle | \langle special\text{-}id \rangle | \langle tuple \rangle | \langle list \rangle | \langle paren\text{-}func\text{-}app \rangle$
 $\langle special\text{-}id \rangle ::= \text{'1st'} | \text{'2nd'} | \text{'3rd'} | \text{'4th'} | \text{'5th'}$

Postfix Functions

- *Examples*

```

name.first_name
list.head
date.year
tuple.1st
apply(.1st)to_all

```

- *Description*

Postfix functions are automatically generated from `tuple_type` definitions (see "Tuple Types" in section 4.1.2). They are functions that take a `tuple_type` value and return a particular field (i.e. projection functions). For example in the first example above we have:

```

name : Name
name.first_name : String

```

Where the function `.first_name` is automatically generated from the definition of the `Name` type:

```

tuple_type Name
value
  (first_name, last_name) : String^2

```

And it has the type `Name => String`.

There are also the following special projection functions that work on tuples (i.e. product type or tuple type values): `.1st` `.2nd` `.3rd` `.4th` `.5th`. For the 4th example above, assuming:

```
tuple : Int x String
```

We have:

```
tuple.1st : Int
```

The general types of these functions are:

```

.1st : (A)Is(B)s_1st ==> B => A
.2nd : (A)Is(B)s_2nd ==> B => A
...

```

These functions are called postfix functions because they are appended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function `.1st` is an argument of the function `apply()to_all`. Postfix functions always begin with a dot.

There are a special postfix function called `".change"` which is described in the following paragraph.

- *Grammar*

$\langle post\text{-}func \rangle ::= \text{'.'} (\langle identifier \rangle | \langle special\text{-}id \rangle | \langle change \rangle)$
 $\langle post\text{-}func\text{-}app \rangle ::= (\langle paren\text{-}expr \rangle | \langle basic\text{-}expr \rangle) \langle post\text{-}func \rangle +$

The ".change" Function

- *Examples*

```
tuple.change{1st = 42, 3rd = 17}
state.change{counter = counter + 1}
name.change{last_name = "Gauss"}
point.change{x = 1.61, y = 2.71, z = 3.14}
apply(.change{1st = 1st + 1})to_all
```

- *Description*

The ".change" function is a special postfix function that works on all tuples (i.e. product type or tuple type values). It returns a new tuple that is the input tuple with some fields changed. Which fields are going to change and to what new value is specified inside curly brackets after the ".change". The following special identifiers can be used for referring to the fields: 1st 2nd 3rd 4th 5th (1st and 5th example). If the tuple is of a tuple type, the identifiers of the fields specified in the type definition can be used (examples 2, 3 and 4). Therefore, we are assuming the following (or similar) if the examples are to type check:

```
tuple : Int x ... x Int (optionally more: x ... ) or some equivalent tuple type
```

```
tuple_type MyStateType
value
  (... , counter, ...) : ... x Int x ...
```

```
state : MyStateType
```

```
tuple_type MyNameType
value
  (... , last_name, ...) : ... x String x ...
```

```
name : MyNameType
```

```
tuple_type Point
value
  (x, y, z) : Real^3
```

```
point : Point
```

```
apply(.change{1st = 1st + 1})to_all
  : ListOf(A)s => ListOf(A)s
  where A must be some product or tuple type where the first field can be added
  with an integer
```

The changes of the fields have the following structure: "field = expression of new value" and they are separated by commas. The input tuple's fields (i.e. the "old" values) can be used inside the expression of a new value and they are referred to by the field identifier.

- *Grammar*

$\langle change \rangle ::= \text{'change\{' } \langle field-change \rangle (\langle comma \rangle \langle field-change \rangle)^* \text{'\}'}$

$\langle field-change \rangle ::= (\langle identifier \rangle | \langle special-id \rangle) \text{'_=' } \langle line-expr \rangle$

3.2 Operators

3.2.1 Function Application and Function Composition Operators

Function Application Operators

Operator	Type
<code>-></code>	$A \times (A \Rightarrow B) \Rightarrow B$
<code><-</code>	$(A \Rightarrow B) \times A \Rightarrow B$

The function application operators "`->`" and "`<-`" are a different way to apply functions to arguments than the usual parenthesis function application. They are meant to look like arrows that point from the argument to the function. These operators are very useful for chaining many function applications without the clutter of having to open and close parentheses for each one of the functions. For example, assuming we have the following functions with the behaviour suggested by their names and types:

```
apply()to_all : (A => B) x ListOf(A)s => ListOf(B)s
string_length : String => Int
filter_with : (A => Bool) x ListOf(A)s => ListOf(A)s
is_odd : Int => Bool
sum_ints : ListOf(Int)s => Int
```

And a list of strings:

```
strings : ListOf(String)s
```

Here is a simple way to get the total number of characters in all the strings that have odd length:

```
chars_in_odd_length_strings : Int
= strings -> apply(string_length)to_all -> filter_with(is_odd) -> sum_ints
```

This can be done equivalently using the other operator:

```
chars_in_odd_length_strings : Int
= sum_ints <- filter_with(is_odd) <- apply(string_length)to_all <- strings
```

These operators can also be used together to put a function between two arguments if that function is commonly used that way in math (or if it looks better for a certain function). For example the "`mod`" function can be used like so:

```
x -> mod <- y
```

Which is equivalent to:

```
mod(x, y)
```

Function Composition Operators

Operator	Type
<code>o></code>	$(A \Rightarrow B) \times (B \Rightarrow C) \Rightarrow (A \Rightarrow C)$
<code><o</code>	$(B \Rightarrow C) \times (A \Rightarrow B) \Rightarrow (A \Rightarrow C)$

The function composition operators "`o>`" and "`<o`" are used to compose functions, each one in the corresponding direction. The use of the letter 'o' is meant to be similar to the mathematical function composition symbol ' \circ ' and the symbols '`>`', '`<`' are used so that the operator points from the function which is applied first to the function which is applied second. A neat example using function composition is the following. Assuming we have the following functions with the behaviour suggested by their names and types:

```

split_to_words : String => ListOf(String)s
apply()to_all : (A => B) x ListOf(A)s => ListOf(B)s
reverse_string : String => String
merge_words : ListOf(String)s => String

```

We can reverse the all the words in a string like so:

```

reverse_words : String => String
  = split_to_words o> apply(reverse_string)to_all o> merge_words

```

This can be done equivalently using the other operator:

```

reverse_words : String => String
  = merge_words <o apply(reverse_string)to_all <o split_to_words

```

3.2.2 Arithmetic, Comparison and Boolean Operators

Arithmetic Operators

Operator	Type
\wedge	$(A)\text{To_The}(B)\text{Has_Type}(C) \implies A \times B \Rightarrow C$
$*$	$(A)\text{And}(B)\text{Multiply_To}(C) \implies A \times B \Rightarrow C$
$/$	$(A)\text{Divided_By}(B)\text{Has_Type}(C) \implies A \times B \Rightarrow C$
$+$	$(A)\text{And}(B)\text{Add_To}(C) \implies A \times B \Rightarrow C$
$-$	$(A)\text{Minus}(B)\text{Has_Type}(C) \implies A \times B \Rightarrow C$

The usual arithmetic operators work as they are expected, similarly to mathematics and other programming languages for the usual types. However, they are generalized. The examples below show their generality:

```

>> 1 + 1
  : Int
  = 2
>> 1 + 3.14
  : Real
  = 4.14
>> 'a' + 'b'
  : String
  = "ab"
>> 'w' + "ord"
  : String
  = "word"
>> "Hello " + "World!"
  : String
  = "Hello World!"
>> 5 * 'a'
  : String
  = "aaaaa"
>> 5 * "hi"
  : String
  = "hihihihihi"
>> "1,2,3" - ','
  : String
  = "123"

```

Let's analyze further the example of addition. The type can be read as such: the '+' operator has the type $A \times B \Rightarrow C$, provided that the type proposition $(A) \text{And} (B) \text{Add_To}(C)$ holds. This proposition being true, means that addition has been defined for these three types (see section "Type Logic" 4.2 for more on type propositions). For example, by the examples above we can deduce that the following propositions are true (in the order of the examples):

```
(Int)And(Int)Add_To(Int)
(Int)And(Real)Add_To(Real)
(Char)And(Char)Add_To(String)
(Char)And(String)Add_To(String)
(Int)And(Char)Multiply_To(String)
(Int)And(String)Multiply_To(String)
(String)Minus(Char)Has_Type(String)
```

This allows us to use the familiar arithmetic operators in types that are not necessarily numbers but it is somewhat intuitively obvious what they should do in those other types. Furthermore, their behaviour can be defined by the user for new user defined types!

Comparison and Boolean Operators

Operator	Type
=	$(A) \text{And} (B) \text{Can_Be_Equal} \Rightarrow A \times B \Rightarrow \text{Bool}$
!=	$(A) \text{And} (B) \text{Can_Be_Unequal} \Rightarrow A \times B \Rightarrow \text{Bool}$
>=	$(A) \text{Can_Be_Gr_Or_Eq_To}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$
<=	$(A) \text{Can_Be_Le_Or_Eq_To}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$
>	$(A) \text{Can_Be_Greater_Than}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$
<	$(A) \text{Can_Be_Less_Than}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$
&	$\text{Bool}^2 \Rightarrow \text{Bool}$

Comparison operators are also generalized. The main reason for the generalization is to be able to compare numbers of different types. Consider the following example:

```
>> 1
   : Int
   = 1
>> 1.1
   : Real
   = 1.1
>> 1.1 = 1
   : Bool
   = false
```

In order for the example to work we need to be able to compare integers and reals. Similarly, all the comparison operators need to be able to work on arguments of different types. Other than that, the comparison and boolean operators behave similarly to most programming languages. It's also worth noting that the "equals", "and" and "or" operators have the symbol once ($=$ & |) instead of twice ($=$ && ||).

3.2.3 Environment Action Operators

Operator	Type
<code>; ></code>	$(E)\text{Has_Use} \implies E(A) \times (A \Rightarrow E(B)) \Rightarrow E(B)$
<code>;</code>	$(E)\text{Has_Then} \implies E(A) \times E(B) \Rightarrow E(B)$

Simple Example

```
print_string("I'll repeat the line.") ; get_line ;> print_string
```

The example above demonstrates the use of the environment action operators with the `FromIO` type, which is how IO is done in `lcases`. Some light can be shed on how this is done, if we take a look at the types (as always!):

```
print_string : String => (EmptyVal)FromIO
print_string("I'll repeat the line.") : (EmptyVal)FromIO
get_line : (String)FromIO

;
  : (E)Has_Then ==> E(A) x E(B) => E(B)
;>
  : (E)Has_Use ==> E(A) x (A => E(B)) => E(B)

print_string("I'll repeat the line.") ; get_line
  : (String)FromIO
  where (FromIO)Has_Then is true, E = FromIO, A = EmptyVal, B = String

print_string("I'll repeat the line.") ; get_line ;> print_string
  : (EmptyVal)FromIO
  where (FromIO)Has_Use is true, E = FromIO, A = String, B = EmptyVal
```

Example program

```
main : (EmptyVal)FromIO
  = print_string <- "Hello! What's your name?" ; get_line ;> name =>
    print_string("Oh hi " + name + "! What's your age?") ; get_line ;> age =>
    print_string("Oh that's crazy " + name + "! I didn't expect you to be " + age + "!");
```

In this bigger but similar example the types are:

```
print_string : String => (EmptyVal)FromIO
get_line : (String)FromIO

print_string <- "Hello! ... " : (EmptyVal)FromIO
print_string("Oh hi...) : (EmptyVal)FromIO
print_string("Oh that's crazy...) : (EmptyVal)FromIO

;
  : (E)Has_Then ==> E(A) x E(B) => E(B)

print_string("Oh hi...) ; get_line
  : (String)FromIO
  where (FromIO)Has_Then is true, E = FromIO, A = EmptyVal, B = String

age => print_string("Oh that's crazy...)
```

```

: String => (EmptyVal)FromIO

;>
: (E)Has_Use ==> E(A) x (A => E(B)) => E(B)

print_string("Oh hi...") ; get_line ;> age =>
print_string("Oh that's crazy...")
: (EmptyVal)FromIO
where (FromIO)Has_Use is true, E = FromIO, A = String, B = EmptyVal

print_string <- "Hello..." ; get_line
: (String)FromIO

name => print_string("Oh hi ... (till the end)
: String => (EmptyVal)FromIO

print_string <- "Hello..." ; get_line ;> name =>
print_string("Oh hi ... (till the end)
: (EmptyVal)FromIO

```

Therefore, "main : (EmptyVal)FromIO" checks out. The key here is to remember that function expressions extend to the end of the whole expression. Therefore, we have "name => ... (till the end)" and "age => ... (till the end)" as the second arguments of the two occurrences of the ">" operator.

Description

The environment action operators are used to combine values that do environment actions into values that do more complicated environment actions. Environment actions are type functions that take a type argument and produce a type (just like ListOf(s)). These type functions have the "then" operator (;) and the "use" operator (;>) defined for them. A value of the type E(A) where (E)Has_Then does an environment action of type E that produces a value of type A which can then be combined with another one with the "then" operator. Similarly, with the "use" operator the produced value of an action can be used by a function that returns another action.

The effect of the ";" operator described in words is the following: given a value of type E(A) and a value of type E(B) (which do environment actions that produce values of type A and B respectively), create a new value that does both actions (provided the first did not result in an error). The overall effect is a value that does an environment action of type E (the combination of the "smaller" actions) which produces a value of type B (the one produced by the second action) and therefore it is of type E(B).

Note that the value of type A produced by the first action is not used anywhere. This happens mostly when A = EmptyVal and it is because values of type E(EmptyVal) are used for their environment action only (e.g. print_string(...) : (EmptyVal)FromIO).

How the two environment actions of the E(A) and E(B) values are combined to produce the new environment action is specific to the environment action type E.

The effect of the ">" operator described in words is the following: given a value of type E(A) (which does an environment action of type E that produces a value of type A) and a value of type A => E(B) (which is a function that takes a value of type A and returns an environment action of type E that produces a value of type B), combine those two values by creating a value that does the following:

- Performs the first action that produces a value of type A
- Takes the value of type A produced (provided there was no error) and passes it to the function of type A => E(B) that then returns an action

- Performs the resulting action

The overall effect is an environment action of type E that in the end produces a value of type B and therefore the new value is of type E(B).

3.2.4 Operator Expressions

- *Examples*

```
1 + 2
1 + x * 3 ^ y
"Hello " + "World!"
x -> f -> g
f o> g o> h
x = y
x >= y - z & x < 2 * y
get_line ; get_line ;> line => print("Second line: " + line)
```

- *Description*

Operator expressions are expressions that use operators. Operators act like two-argument-functions that are placed in between their arguments. Therefore, they have function types and they act as it is described in their respective sections above this one.

An operator expression might have multiple operators. The order of operations is explained in the next section ("Complete Table, Precedence and Associativity") in Table 2.

Just like functions, the sub-expressions that act as arguments to an operator, must have types that match the types expected by the operator.

It is possible for the second argument of an operator to be a function expression. This is mostly useful with the ";>" operator (see previous section: "Environment Operators"), but it is also possible with the following operators: "->", "o>", "<o".

- *Big Operator Expressions*

Example

```
"Hello, I'm a big string that's going to contain multiple values from " +
"inside the imaginary program that I'm a part of. Here they are:\n" +
"value1 = " + value1 + ", value2 = " + value2 + ", value3 = " + value3 +
", value4 = " + value4 + ", value5 = " + value5
```

Description

It is possible to stretch a (big) operator expression over multiple lines. In that case:

- The operator expression must split in a new line after an operator (not an argument).
- Every line after the first must be indented so that it begins at the column where the first line of the operator expression begun.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

$\langle op\text{-}expr \rangle ::= \langle line\text{-}op\text{-}expr \rangle \mid \langle big\text{-}op\text{-}expr \rangle$

3.2.5 Complete Table, Precedence and Associativity

Table 1: The complete table of lcases operators along with their types and their short descriptions.

Operator	Type	Description
->	$A \times (A \Rightarrow B) \Rightarrow B$	Right function application
<-	$(A \Rightarrow B) \times A \Rightarrow B$	Left function application
o>	$(A \Rightarrow B) \times (B \Rightarrow C) \Rightarrow (A \Rightarrow C)$	Right function composition
<o	$(B \Rightarrow C) \times (A \Rightarrow B) \Rightarrow (A \Rightarrow C)$	Left function composition
^	$(A)\text{To_The}(B)\text{Has_Type}(C) \Rightarrow A \times B \Rightarrow C$	General exponentiation
*	$(A)\text{And}(B)\text{Multiply_To}(C) \Rightarrow A \times B \Rightarrow C$	General multiplication
/	$(A)\text{Divided_By}(B)\text{Has_Type}(C) \Rightarrow A \times B \Rightarrow C$	General division
+	$(A)\text{And}(B)\text{Add_To}(C) \Rightarrow A \times B \Rightarrow C$	General addition
-	$(A)\text{Minus}(B)\text{Has_Type}(C) \Rightarrow A \times B \Rightarrow C$	General subtraction
=	$(A)\text{And}(B)\text{Can_Be_Equal} \Rightarrow A \times B \Rightarrow \text{Bool}$	General Equality
!=	$(A)\text{And}(B)\text{Can_Be_Unequal} \Rightarrow A \times B \Rightarrow \text{Bool}$	General Inequality
>=	$(A)\text{Can_Be_Gr_Or_Eq_To}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$	General greater than or equal to
<=	$(A)\text{Can_Be_Le_Or_Eq_To}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$	General less than or equal to
>	$(A)\text{Can_Be_Greater_Than}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$	General greater than
<	$(A)\text{Can_Be_Less_Than}(B) \Rightarrow A \times B \Rightarrow \text{Bool}$	General less than
&	$\text{Bool}^2 \Rightarrow \text{Bool}$	Boolean operators
; >	$(E)\text{Has_Use} \Rightarrow E(A) \times (A \Rightarrow E(B)) \Rightarrow E(B)$	Do, use, do
;	$(E)\text{Has_Then} \Rightarrow E(A) \times E(B) \Rightarrow E(B)$	Do then do

The order of operations is done from highest to lowest precedence. In the same level of precedence the order is done from left to right if the associativity is "Left" and from right to left if the associativity is "Right". For the operators that have associativity "None" it is not allowed to place them in the same operator expression. The precedence and associativity of the operators is shown in the table below.

Table 2: The table of precedence and associativity of the lcases operators.

Operator	Precedence	Associativity
->	10 (highest)	Left
<-	9	Right
o> <o	8	Left
^	7	Right
* /	6	Left
+ -	5	Left
= != > < >= <=	4	None
&	3	Left
	2	Left
; > ;	1	Left

3.3 Function Expressions

Function expressions are divided into **regular function expressions** and **"cases" function expressions** which are described in the following sections.

The grammar for a function expression is:

$\langle func\text{-}expr \rangle ::= \langle line\text{-}func\text{-}expr \rangle \mid \langle big\text{-}func\text{-}expr \rangle \mid \langle cases\text{-}func\text{-}expr \rangle$

3.3.1 Regular Function Expressions

- *Examples*

```
a => 17 * a + 42
(x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1 / 2)
```

- *Description*

Regular function expressions are used to define functions or be part of bigger expressions as anonymous functions. They are comprised by their parameters and their body. A parameter has an identifier. The parameters are either only one, in which case there is no parenthesis, or they are many, in which case they are in parenthesis, separated by commas. The parameters and the body are separated by an arrow ("=>"). The body is an operator expression or an operator argument expression.

- *Big Function Expressions*

Example

```
(value1, value2, value3, value4, value5, value6, value7) =>
print_line("value1 = " + value1 + ", value2 = " + value2 + ", value3 = " + value3) ;
print_line("value4 = " + value4 + ", value5 = " + value5 + ", value6 = " + value6) ;
print_line("value7 = " + value7)
```

Description

It is possible to stretch a (big) function expression over multiple lines. In that case:

- The function expression must split in a new line after the "=>" arrow.
- Every line after the first must be indented so that it begins at the column where the first character of the parameters was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

$\langle line\text{-}func\text{-}expr \rangle ::= \langle parameters \rangle \text{ '=>' } \langle line\text{-}func\text{-}body \rangle$

$\langle big\text{-}func\text{-}expr \rangle ::= \langle parameters \rangle \text{ '=>' } \langle big\text{-}func\text{-}body \rangle$

$\langle parameters \rangle ::= \langle identifier \rangle \mid \langle paren\text{-}comma\text{-}sep\text{-}ids \rangle$

$\langle paren\text{-}comma\text{-}sep\text{-}ids \rangle ::= \text{'(' } \langle identifier \rangle (\langle comma \rangle \langle identifier \rangle) + \text{'}'$

$\langle line\text{-}func\text{-}body \rangle ::= \text{'\textasciitilde'} (\langle no\text{-}paren\text{-}op\text{-}arg \rangle \mid \langle line\text{-}op\text{-}expr \rangle)$

$\langle big\text{-}func\text{-}body \rangle ::= \langle nl \rangle \langle indent \rangle (\langle no\text{-}paren\text{-}op\text{-}arg \rangle \mid \langle op\text{-}expr \rangle)$

3.3.2 "cases" Function Expressions

- *Examples*

```
print_sentimental_bool : Bool => IO
= cases
  true => print <- "It's true!! :)"
  false => print <- "It's false... :("

or_type TrafficLight
values
  green | amber | red

print_sentimental_traffic_light : TrafficLight => IO
= cases
  green => print <- "It's green! Let's go!!! :)"
  amber => print <- "Go go go, fast!"
  red => print <- "Stop right now! You're going to kill us!!"

is_not_red : TrafficLight => Bool
= cases
  green => true
  amber => true
  red => false

is_seventeen_or_forty_two : Int => Bool
= cases
  17 => true
  42 => true
  ... => false

traffic_lights_match : TrafficLight^2 => Bool
= (cases, cases)
  (green, green) => true
  (amber, amber) => true
  (red, red) => true
  ... => false

gcd : Int^2 => Int
= (x, cases)
  0 => x
  y => gcd(y, x -> mod <- y)

is_empty : ListOf(A)s => Bool
= cases
  empty_l => true
  non_empty_l:anything => false

apply()to_all : (A => B) x ListOf(A)s => ListOf(B)s
= (f, cases)
  empty_l => empty_l
  non_empty_l:list => non_empty_l:(f <- list.head, apply(f)to_all <- list.tail)
```

- *Description*

"cases" is a keyword that works as a special parameter. The difference is that instead of giving the name "cases" to that parameter, it allows the programmer to pattern match on the possible values of that parameter and return a different result for each particular case.

The last case can be "... => (body of default case)" to capture all remaining cases while dismissing the value (e.g. `is_seventeen_or_forty_two` example), or it can be "some_id => (body of default case)" to capture all remaining cases while being able to use the value with the name "some_id" (e.g. "y" in `gcd` example).

It is possible to use the "cases" keyword in multiple parameters to match on all of them. By doing that, each case represents a particular combination of values for the parameters involved (e.g. `traffic_lights_match` example).

It is also possible to use a "where" expression below a particular case. The "where" expression must be indented two spaces more than the line where that particular case begins.

A function expression that uses the "cases" syntax must contain the "cases" keyword in at least one parameter. The number of matching expressions in all cases must be the same as the number of parameters with the "cases" keyword.

- *Grammar*

$$\begin{aligned} \langle \text{cases-func-expr} \rangle &::= \langle \text{cases-parameters} \rangle \langle \text{case} \rangle + [\langle \text{end-case} \rangle] \\ \langle \text{cases-parameters} \rangle &::= \langle \text{cases-parameter} \rangle | ' (\langle \text{cases-parameter} \rangle (\langle \text{comma} \rangle \langle \text{cases-parameter} \rangle) + ') ' \\ \langle \text{cases-parameter} \rangle &::= \langle \text{identifier} \rangle | \text{'cases'} \\ \langle \text{case} \rangle &::= \langle \text{nl} \rangle \langle \text{indent} \rangle \langle \text{matching} \rangle \text{'_=>'} \langle \text{case-body} \rangle \\ \langle \text{end-case} \rangle &::= \langle \text{nl} \rangle \langle \text{indent} \rangle \text{'..._=>'} \langle \text{case-body} \rangle \\ \langle \text{matching} \rangle &::= \langle \text{literal} \rangle | \langle \text{identifier} \rangle | \langle \text{pre-func} \rangle \langle \text{matching} \rangle | \langle \text{tuple-matching} \rangle | \langle \text{list-matching} \rangle \\ \langle \text{tuple-matching} \rangle &::= ' (\langle \text{matching} \rangle (\langle \text{comma} \rangle \langle \text{matching} \rangle) + ') ' \\ \langle \text{list-matching} \rangle &::= ' [\langle \text{matching} \rangle (\langle \text{comma} \rangle \langle \text{matching} \rangle) *] ' \\ \langle \text{case-body} \rangle &::= (\langle \text{line-func-body} \rangle | \langle \text{big-func-body} \rangle) [\langle \text{where-expr} \rangle] \end{aligned}$$

3.4 Value Definitions and "where" Expressions

3.4.1 Value Definitions

- *Examples*

```
foo : Int
    = 42

f : Int^3 => Int
  = (a, b, c) => a + b * c
```

```

val1, val2, val3 : Int, Bool, Char
  = 42, true, 'a'

int1, int2, int3 : all Int
  = 1, 2, 3

```

- *Description*

Value definitions are the main building block of lcases programs. To define a new value you give it a name, a type and an expression. The name is an identifier in the first line. The second line is indented two spaces more and begins by " : " and continues with the type expression. The third line is indented as the second, begins by " = " and continues with the value expression (which extends to as many lines as needed).

A value definition is either in the first column, where it can be "seen" by all other value definitions, or it is in a "where" expression (see section below), where it can be "seen" by the expression above the "where" and all the other definitions in the same "where" expression.

A value definition can be followed by a "where" expression where intermediate values used in the value expression are defined. In that case, the "where" expression must be indented two spaces more than the "=" line of the value definition.

It is possible to group value definitions together by separating the names, the types and the expressions with commas. This is very useful for not cluttering the program with many definitions for values with small expressions (e.g. constants). When grouping definitions together it is also possible to use the keyword "all" to give the same type to all the values.

- *Grammar*

```

⟨value-def⟩ ::=
  ⟨indent⟩ ⟨identifier⟩ ( ' ' | ⟨nl⟩ ⟨indent⟩ ) ' : ' ⟨type⟩ ⟨nl⟩ ⟨indent⟩ ' = ' ⟨value-expr⟩ [ ⟨where-expr⟩ ]

⟨value-expr⟩ ::= ⟨no-paren-op-arg⟩ | ⟨op-expr⟩ | ⟨func-expr⟩ | ⟨big-tuple⟩ | ⟨big-list⟩

⟨grouped-value-defs⟩ ::=
  ⟨indent⟩ ⟨identifier⟩ ( ⟨comma⟩ ⟨identifier⟩ )+
  ( ' ' | ⟨nl⟩ ⟨indent⟩ ) ' : ' ( ⟨type⟩ ( ⟨comma⟩ ⟨type⟩ )+ | 'all' ⟨type⟩ )
  ⟨nl⟩ ⟨indent⟩ ' = ' ⟨line-exprs⟩ ( ⟨nl⟩ ⟨indent⟩ ⟨comma⟩ ⟨line-exprs⟩ ) *

```

3.4.2 "where" Expressions

- *Examples*

```

sort : ListOf(Int)s => ListOf(Int)s
  = cases
    empty_l => empty_l
    non_empty_l:l => sort(less_l) + l.head + sort(greater_l)
    where
      less_l, greater_l : all ListOf(Int)s
        = filter_with(x => x < l.head, l.tail)
        , filter_with(x => x >= l.head, l.tail)

sum_nodes : TreeOf(Int)s => Int
  = tree => tree.root + tree.subtrees -> apply(sum_nodes)to_all -> sum_list
  where
    sum_list : ListOf(Int)s => Int
      = cases

```



```

empty_l => 0
non_empty_l:1 => l.head + sum_list(l.tail)

big_string : String
= s1 + s2 + s3 + s4
  where
    s1, s2, s3, s4 : all String
    = "Hello, my name is Struggling Programmer."
    , "I have tried way too many times to fit a big chunk of text"
    , "inside my program, without it hitting the half-screen mark!"
    , "I am so glad I finally discovered lcases!"

```

- *Description*

"where" expressions allow the programmer to use values inside an expression and define them below it. They are very useful for reusing or abbreviating expressions that are specific to a particular definition or case.

A "where" expression begins by a line that only has the word "where" in it. It is indented as described in the "Value Definitions" (3.4.1) or "'cases' Function Expressions" (3.3.2) sections. The definitions are placed below the "where" line and must have the same indentation.

- *Grammar*

$$\langle where\text{-}expr \rangle ::= \langle nl \rangle \langle indent \rangle \text{'where'} \langle nl \rangle \langle value\text{-}def\text{-}or\text{-}defs \rangle (\langle nl \rangle \langle nl \rangle \langle value\text{-}def\text{-}or\text{-}defs \rangle)^*$$

$$\langle value\text{-}def\text{-}or\text{-}defs \rangle ::= \langle value\text{-}def \rangle \mid \langle grouped\text{-}value\text{-}defs \rangle$$

3.5 Indentation and Complete Grammar for Values

3.5.1 Indentation System

The $\langle indent \rangle$ nonterminal is not a normal BNF nonterminal. It is a context sensitive construct that enforces the indentation rules of lcases. It depends on a integer value called the "indentation level" (il). The $\langle indent \rangle$ nonterminal corresponds to $2 * il$ space characters. The indentation level follows the rules below:

Indentation Rules

- At the beginning: $il = 0$.
- At the end of the first line of a value definition: $il \leftarrow il + 1$. This also applies to grouped value definitions.
- At the end of the third line ("=" line) of a (single) value definition: $il \leftarrow il + 1$.
- At the end of a (single) value definition: $il \leftarrow il - 2$.
- At the end of grouped value definitions: $il \leftarrow il - 1$.
- After the "=>" arrow in a case: $il \leftarrow il + 1$.
- At the end of a case body: $il \leftarrow il - 1$.
- In a cases function expression which does not begin at the "=" line of a value definition:
 - After the arrow "=>" at the end of the parameters: $il \leftarrow il + 1$.
 - At the end of the cases function expression: $il \leftarrow il - 1$.
- In a type theorem (section 4.2.2), if the value expression is a big or cases expression:
 - After "=" line: $il \leftarrow il + 2$.
 - After the value expression: $il \leftarrow il - 2$.

3.5.2 Complete Grammar for Values

$\langle literal \rangle ::= \langle int-lit \rangle \mid \langle real-lit \rangle \mid \langle char-lit \rangle \mid \langle string-lit \rangle$

$\langle identifier \rangle ::= [a-z] [a-z_]* ('(' [a-z_]+)^* [[0-9]]$

$\langle paren-expr \rangle ::= '(' \langle line-op-expr \rangle \mid \langle line-func-expr \rangle ')'$

$\langle tuple \rangle ::= '(' \langle line-expr \rangle \langle comma \rangle \langle line-exprs \rangle ')'$

$\langle line-exprs \rangle ::= \langle line-expr \rangle (\langle comma \rangle \langle line-expr \rangle)^*$

$\langle line-expr \rangle ::= \langle no-paren-op-arg \rangle \mid \langle line-op-expr \rangle \mid \langle line-func-expr \rangle$

$\langle comma \rangle ::= ',' ['_ ']$

$\langle big-tuple \rangle ::=$
 $\quad '(' ['_ '] \langle line-expr \rangle [\langle nl \rangle \langle indent \rangle] \langle comma \rangle \langle line-exprs \rangle$
 $\quad (\langle nl \rangle \langle indent \rangle \langle comma \rangle \langle line-exprs \rangle)^* \langle nl \rangle \langle indent \rangle ')'$

$\langle list \rangle ::= '[' [\langle line-exprs \rangle] ']'$

$\langle big-list \rangle ::= '[' ['_ '] \langle line-exprs \rangle (\langle nl \rangle \langle indent \rangle \langle comma \rangle \langle line-exprs \rangle)^* \langle nl \rangle \langle indent \rangle ']'$

$\langle paren-func-app \rangle ::=$
 $\quad [\langle arguments \rangle] \langle identifier-with-arguments \rangle [\langle arguments \rangle]$
 $\quad \mid \langle arguments \rangle \langle identifier \rangle [\langle arguments \rangle]$
 $\quad \mid \langle identifier \rangle \langle arguments \rangle$

$\langle arguments \rangle ::= '(' \langle line-exprs \rangle ')'$

$\langle identifier-with-arguments \rangle ::=$
 $\quad [a-z] [a-z_]* ('(' [a-z_]+)^* \langle arguments \rangle [a-z_]+ (('(' \mid \langle arguments \rangle) [a-z_]+)^* [[0-9]]$

$\langle pre-func \rangle ::= \langle identifier \rangle ':'$

$\langle pre-func-app \rangle ::= \langle pre-func \rangle (\langle basic-expr \rangle \mid \langle paren-expr \rangle \mid \langle pre-func-app \rangle \mid \langle post-func-app \rangle)$

$\langle basic-expr \rangle ::= \langle literal \rangle \mid \langle identifier \rangle \mid \langle tuple \rangle \mid \langle list \rangle \mid \langle paren-func-app \rangle \mid \langle special-id \rangle$

$\langle post-func \rangle ::= '.' (\langle identifier \rangle \mid \langle special-id \rangle \mid \langle change \rangle)$

$\langle special-id \rangle ::= '1st' \mid '2nd' \mid '3rd' \mid '4th' \mid '5th'$

$\langle post-func-app \rangle ::= (\langle paren-expr \rangle \mid \langle basic-expr \rangle) \langle post-func \rangle +$

$\langle change \rangle ::= 'change\{ ' \langle field-change \rangle (\langle comma \rangle \langle field-change \rangle)^* ' \}'$

$\langle \text{field-change} \rangle ::= (\langle \text{identifier} \rangle \mid \langle \text{special-id} \rangle) \text{ '}_\sqcup=\sqcup\text{' } \langle \text{line-expr} \rangle$

$\langle \text{op-expr} \rangle ::= \langle \text{line-op-expr} \rangle \mid \langle \text{big-op-expr} \rangle$

$\langle \text{op-expr-start} \rangle ::= (\langle \text{op-arg} \rangle \langle \text{op} \rangle) +$

$\langle \text{line-op-expr} \rangle ::= \langle \text{op-expr-start} \rangle (\langle \text{op-arg} \rangle \mid \langle \text{line-func-expr} \rangle)$

$\langle \text{big-op-expr} \rangle ::= \langle \text{big-op-expr-op-split} \rangle \mid \langle \text{big-op-expr-func-split} \rangle$

$\langle \text{big-op-expr-op-split} \rangle ::= \langle \text{op-split-line} \rangle + [\langle \text{op-expr-start} \rangle] (\langle \text{op-arg} \rangle \mid \langle \text{func-expr} \rangle)$

$\langle \text{op-split-line} \rangle ::= \langle \text{op-expr-start} \rangle (\langle \text{nl} \rangle \mid \langle \text{op-arg} \rangle \langle \text{comp-op-nl} \rangle) \langle \text{indent} \rangle$

$\langle \text{big-op-expr-func-split} \rangle ::= \langle \text{op-expr-start} \rangle (\langle \text{big-func-expr} \rangle \mid \langle \text{cases-func-expr} \rangle)$

$\langle \text{op-arg} \rangle ::= \langle \text{no-paren-op-arg} \rangle \mid \langle \text{paren-expr} \rangle$

$\langle \text{no-paren-op-arg} \rangle ::= \langle \text{basic-expr} \rangle \mid \langle \text{pre-func} \rangle \mid \langle \text{post-func} \rangle \mid \langle \text{pre-func-app} \rangle \mid \langle \text{post-func-app} \rangle$

$\langle \text{op} \rangle ::= \text{'}_\sqcup\text{' } (\text{'o>' } \mid \text{'<o' }) \text{'}_\sqcup\text{' } [\text{'}_\sqcup\text{' }] \langle \text{optional-spaces-op} \rangle [\text{'}_\sqcup\text{' }]$

$\langle \text{optional-spaces-op} \rangle ::= \text{'->' } \mid \text{'<-'} \mid \text{'^'} \mid \text{'*'} \mid \text{'/' } \mid \text{'+' } \mid \text{'-'} \mid \text{'=' } \mid \text{'!='} \mid \text{'>' } \mid \text{'<' } \mid \text{'>=' } \mid \text{'<=' } \mid \text{'&' } \mid \text{'|'} \mid \text{'>;'} \mid \text{'<;'}$

$\langle \text{comp-op-nl} \rangle ::= \text{'}_\sqcup\text{' } (\text{'o>' } \mid \text{'<o' }) \text{'\n'}$

$\langle \text{func-expr} \rangle ::= \langle \text{line-func-expr} \rangle \mid \langle \text{big-func-expr} \rangle \mid \langle \text{cases-func-expr} \rangle$

$\langle \text{line-func-expr} \rangle ::= \langle \text{parameters} \rangle \text{'}_\sqcup=>\text{' } \langle \text{line-func-body} \rangle$

$\langle \text{big-func-expr} \rangle ::= \langle \text{parameters} \rangle \text{'}_\sqcup=>\text{' } \langle \text{big-func-body} \rangle$

$\langle \text{parameters} \rangle ::= \langle \text{identifier} \rangle \mid \langle \text{paren-comma-sep-ids} \rangle$

$\langle \text{paren-comma-sep-ids} \rangle ::= \text{'(' } \langle \text{identifier} \rangle (\langle \text{comma} \rangle \langle \text{identifier} \rangle) + \text{'}'$

$\langle \text{line-func-body} \rangle ::= \text{'}_\sqcup\text{' } (\langle \text{no-paren-op-arg} \rangle \mid \langle \text{line-op-expr} \rangle)$

$\langle \text{big-func-body} \rangle ::= \langle \text{nl} \rangle \langle \text{indent} \rangle (\langle \text{no-paren-op-arg} \rangle \mid \langle \text{op-expr} \rangle)$

$\langle \text{cases-func-expr} \rangle ::= \langle \text{cases-parameters} \rangle \langle \text{case} \rangle + [\langle \text{end-case} \rangle]$

$\langle \text{cases-parameters} \rangle ::= \langle \text{cases-parameter} \rangle \mid \text{'(' } \langle \text{cases-parameter} \rangle (\langle \text{comma} \rangle \langle \text{cases-parameter} \rangle) + \text{'}'$

$\langle \text{cases-parameter} \rangle ::= \langle \text{identifier} \rangle \mid \text{'cases'}$

$\langle \text{case} \rangle ::= \langle \text{nl} \rangle \langle \text{indent} \rangle \langle \text{matching} \rangle \text{'_=>'} \langle \text{case-body} \rangle$

$\langle \text{end-case} \rangle ::= \langle \text{nl} \rangle \langle \text{indent} \rangle \text{'\dots_=>'} \langle \text{case-body} \rangle$

$\langle \text{matching} \rangle ::= \langle \text{literal} \rangle \mid \langle \text{identifier} \rangle \mid \langle \text{pre-func} \rangle \langle \text{matching} \rangle \mid \langle \text{tuple-matching} \rangle \mid \langle \text{list-matching} \rangle$

$\langle \text{tuple-matching} \rangle ::= \text{'('} \langle \text{matching} \rangle (\langle \text{comma} \rangle \langle \text{matching} \rangle)+ \text{'}'$

$\langle \text{list-matching} \rangle ::= \text{'['} [\langle \text{matching} \rangle (\langle \text{comma} \rangle \langle \text{matching} \rangle)^*] \text{'}'$

$\langle \text{case-body} \rangle ::= (\langle \text{line-func-body} \rangle \mid \langle \text{big-func-body} \rangle) [\langle \text{where-expr} \rangle]$

$\langle \text{value-def} \rangle ::= \langle \text{indent} \rangle \langle \text{identifier} \rangle (\text{' '} \mid \langle \text{nl} \rangle \langle \text{indent} \rangle) \text{' :_ ' } \langle \text{type} \rangle \langle \text{nl} \rangle \langle \text{indent} \rangle \text{' =_ ' } \langle \text{value-expr} \rangle [\langle \text{where-expr} \rangle]$

$\langle \text{value-expr} \rangle ::= \langle \text{no-paren-op-arg} \rangle \mid \langle \text{op-expr} \rangle \mid \langle \text{func-expr} \rangle \mid \langle \text{big-tuple} \rangle \mid \langle \text{big-list} \rangle$

$\langle \text{grouped-value-defs} \rangle ::=$
 $\langle \text{indent} \rangle \langle \text{identifier} \rangle (\langle \text{comma} \rangle \langle \text{identifier} \rangle)+$
 $(\text{' '} \mid \langle \text{nl} \rangle \langle \text{indent} \rangle) \text{' :_ ' } (\langle \text{type} \rangle (\langle \text{comma} \rangle \langle \text{type} \rangle)+ \mid \text{'all_ ' } \langle \text{type} \rangle)$
 $\langle \text{nl} \rangle \langle \text{indent} \rangle \text{' =_ ' } \langle \text{line-exprs} \rangle (\langle \text{nl} \rangle \langle \text{indent} \rangle \langle \text{comma} \rangle \langle \text{line-exprs} \rangle)^*$

$\langle \text{where-expr} \rangle ::= \langle \text{nl} \rangle \langle \text{indent} \rangle \text{'where'} \langle \text{nl} \rangle \langle \text{value-def-or-defs} \rangle (\langle \text{nl} \rangle \langle \text{nl} \rangle \langle \text{value-def-or-defs} \rangle)^*$

$\langle \text{value-def-or-defs} \rangle ::= \langle \text{value-def} \rangle \mid \langle \text{grouped-value-defs} \rangle$

4 Language Description: Types and Type Logic

4.1 Types

The constructs regarding types are **type expressions**, **type definitions** and **type nicknames** and they are described in the following sections.

4.1.1 Type Expressions

Type expressions are divided into the following categories:

- Type Identifiers
- Type Variables
- Function Types
- Product Types
- Type Application Types
- Conditional Types

which are described in the following paragraphs.

The grammar of a type expression is:

$\langle type \rangle ::= [\langle condition \rangle] \langle simple-type \rangle$

$\langle simple-type \rangle ::= \langle type-id \rangle \mid \langle type-var \rangle \mid \langle func-type \rangle \mid \langle prod-type \rangle \mid \langle power-type \rangle \mid \langle type-app \rangle$

Type Identifiers

- *Examples*

```
Int
Real
Char
String
SelfReferencingType
MyDefinedType
```

- *Description*

A type identifier is either the name of a basic type (Int, Real, Char, String) or the name of some defined type that has no type parameters. It begins with a capital letter and is followed by one or more capital or lowercase letters.

- *Grammar*

$\langle type-id \rangle ::= [A-Z] [A-Za-z]^+$

Type Variables

- *Examples*

```
A B C
X Y Z
```

- *Examples of type variables inside bigger type expressions*

```
A => A
(A => B) x (B => C) => (A => C)
(A^2 => A) x A x ListOf(A)s => A
```

- *Description*

Type Variables are used inside larger type expressions of polymorphic types. A polymorphic type is a type where any function of that type can be used as a function of any type that corresponds to substituting every type variable of the polymorphic type with a particular type. The easiest example of a polymorphic type is the type of the identity function where we have:

```
id : A => A
    = x => x
```

```
id(1) : Int
    where A is substituted by Int and id gets the type Int => Int
```

```
id("Hello") : String
    where A is substituted by String and id gets the type String => String
```

A type variable is a single capital letter.

- *Grammar*

$\langle type-var \rangle ::= [A-Z]$

Function Types

- *Examples*

```
String => String
Real => Int
A => A
Int^2 => Int
Real^3 => Real
(A => B) x (B => C) => (A => C)
(Int => Int) => (Int => Int)
```

- *Description*

A function type expression is comprised of the input type expression and the output type expression separated by the arrow "=>". The input and output type expressions are type expressions which are put in parentheses only if they are function type expressions.

- *Grammar*

$$\langle \text{func-type} \rangle ::= \langle \text{in-or-out-type} \rangle \text{ '=>' } \langle \text{in-or-out-type} \rangle$$
$$\langle \text{in-or-out-type} \rangle ::= \langle \text{type-id} \rangle \mid \langle \text{type-var} \rangle \mid \langle \text{prod-type} \rangle \mid \langle \text{power-type} \rangle \mid \langle \text{type-app} \rangle \mid \text{'(' } \langle \text{func-type} \rangle \text{'}'$$

Product Types

- *Examples*

```
Int x Real x String
ListOf(Int)s x Int x ListOf(String)s
(Int => Int) x (Int x Real) x (Real => String)
Int^2
Real^3
```

- *Description*

Product types are the types of tuples. They are comprised of the expressions of the types of the fields separated by the string " x " (space 'x' space) because 'x' is very similar the symbol used in the cartesian product. If any of the fields is of a product or a function type then the corresponding type expression must be inside parentheses. A product type where all the fields are of the same type can be abbreviated with a power type expression which is comprised of the type, the power symbol '^' and the number of times the type is repeated.

- *Grammar*

$$\langle \text{prod-type} \rangle ::= \langle \text{field-type} \rangle \text{ ('x' } \langle \text{field-type} \rangle \text{)}^+$$
$$\langle \text{power-type} \rangle ::= \langle \text{field-type} \rangle \text{ ('^' } \langle \text{int-greater-than-one} \rangle \text{)}^+$$
$$\langle \text{field-type} \rangle ::= \langle \text{type-id} \rangle \mid \langle \text{type-var} \rangle \mid \langle \text{type-app} \rangle \mid \text{'(' } (\langle \text{func-type} \rangle \mid \langle \text{prod-type} \rangle) \text{'}'$$

Type Application Types

- *Examples*

```
Possibly(Int)
ListOf(Real)s
TreeOf(String)s
Error(String)OrResult(Int)
ListOf(Int => Int)s
ListOf(A)s
```

- *Description*

Type application types are types that are produced by passing arguments to a type function generated by a `tuple_type` or an `or_type` definition. For example, given the definition of `ListOf(A)s`:

```
or_type ListOf(A)s
values
  non_empty_1:NonEmptyListOf(A)s | empty_1
```

We have that `ListOf()`s is a type function that receives one type parameter and returns a resulting type. For example `ListOf(Int)s` is the result of passing the type argument `Int` to `ListOf()`s.

Type application types have the same form as the name in the `tuple_type` or `or_type` definition, with the difference that type parameters are substituted by the type expressions of the arguments.

- *Grammar*

```
⟨type-app⟩ ::=
  [ ⟨types-in-paren⟩ ] ⟨type-id-with-args⟩ [ ⟨types-in-paren⟩ ]
  | ⟨types-in-paren⟩ ⟨type-id-or-var⟩ [ ⟨types-in-paren⟩ ]
  | ⟨type-id-or-var⟩ ⟨types-in-paren⟩

⟨type-id-with-args⟩ ::= ⟨type-id⟩ ( ⟨types-in-paren⟩ [A-Za-z]+ )+

⟨type-id-or-var⟩ ::= ⟨type-id⟩ | ⟨type-var⟩

⟨types-in-paren⟩ ::= ‘(’ ⟨simple-type⟩ ( ⟨comma⟩ ⟨simple-type⟩ )* ‘)’
```

Conditional Types

- *Examples*

```
(A)And(B)Can_Be_Equal ==> A x B => Bool
(A)And(B)Add_To(C) ==> A x B => C
(A)Is(B)s_First ==> B => A
(T)Has_String_Repr ==> T => String
(E)Has_Use ==> E(A) x (A => E(B)) => E(B)
```

- *Description*

Conditional types are the types of values that are polymorphic not because of their structure but because they have been defined (seperately) for many different combinations of types (i.e. they are ad hoc polymorphic). They are comprised of a condition and a "simple" type (i.e. a type without a condition) which are seperated by the arrow "`==>`". The condition is a type proposition which refers to type variables inside the "simple" type and it must hold whenever the polymorphic value of that type is used. For example:

```
first : (A)Is(B)s_First => B => A
```

can be used as follows:

```
pair, triple, list
  : Int x String, Real x Char x Int, ListOf(String)s
  = (42, "The answer to everything"), (3.14, 'a', 1), ["Hi!", "Hello", Heeey"]

>> pair -> first
  : Int
  = 42
```

```
>> triple -> first
      : Real
      = 3.14
>> list -> first
      : String
      = "Hi!"
```

and that is because the following propositions hold:

```
(Int)Is(Int x String)s_First
(Real)Is(Real x Char x Int)s_First
(String)Is(ListOf(String)s)s_First
```

which it turn means that the function `first` has been defined for these combinations of types. For more on how conditions, propositions and ad hoc polymorphism works, see the "Type Logic" section (4.2).

- *Grammar*

$\langle condition \rangle ::= \langle prop-name \rangle ' \sqsubseteq == \sqsupset '$

4.1.2 Type Definitions

Type definitions are divided into `tuple_type` definitions and `or_type` definitions which are described in the following paragraphs.

The grammar of a type definition is:

$\langle type-def \rangle ::= \langle tuple-type-def \rangle \mid \langle or-type-def \rangle$

Tuple Types

- *Definition Examples*

```
tuple_type Name
value
  (first_name, last_name) : String^2

tuple_type Date
value
  (day, month, year) : Int^3

tuple_type MathematicianInfo
value
  (name, nationality, date_of_birth) : Name x String x Date

tuple_type TreeOf(A)s
value
  (root, subtrees) : A x ListOf(TreeOf(A)s)s

tuple_type Indexed(T)
value
  (index, val) : Int x T
```

- *Usage Examples*


```

euler_info : MathematicianInfo
  = (("Leonhard", "Euler"), "Swiss", (15, 4, 1707))

name_to_string : Name => String
  = n => "\nFirst Name: " + n.first_name + "\nLast Name: " + n.last_name

print_name_and_nationality : MathematicianInfo => IO
  = ci => print(ci.name -> name_to_string + "\nNationality: " + ci.nationality)

sum_nodes : TreeOf(Int)s => Int
  = tree => tree.root + tree.subtrees -> apply(sum_nodes)to_all -> sum_list

```

- *Description*

A tuple type is equivalent to a product type with a new name and names for the fields for convinience. A tuple type generates postfix functions for all of the fields by using a '.' before the name of the field. For example the MathematicianInfo type above generates the following functions:

```

.name : MathematicianInfo => Name
.nationality : MathematicianInfo => String
.date_of_birth : MathematicianInfo => Date

```

These functions are named "postfix functions" because they can be appended to their argument.

- *Grammar*

```

⟨tuple-type-def⟩ ::=
  'tuple_type' ⟨type-name⟩ ⟨nl⟩
  'value' ⟨nl⟩ '⊔' ⟨paren-comma-sep-ids⟩ '⊔:⊔' ( ⟨prod-type⟩ | ⟨power-type⟩ )

⟨type-name⟩ ::= [ ⟨params-in-paren⟩ ] ⟨type-id⟩ ( ⟨params-in-paren⟩ [A-Za-z]+ ) * [ ⟨params-in-paren⟩ ]

⟨params-in-paren⟩ ::= '(' ⟨type-var⟩ ( ⟨comma⟩ ⟨type-var⟩ ) * ')'

```

Or Types

- *Definition Examples*

```

or_type Bool
values
  true | false

or_type Possibly(A)
values
  the_value:A | no_value

// needed tuple_type for ListOf(A)s
tuple_type NonEmptyListOf(A)s
value
  (head, tail) : A x ListOf(A)s

or_type ListOf(A)s
values
  non_empty_l:NonEmptyListOf(A)s | empty_l

```

```

or_type Error(A)OrResult(B)
values
  error:A | result:B

```

- *Usage Examples*

```

is_empty : ListOf(A)s => Bool
= cases
  empty_l => true
  non_empty_l:anything => false

get_head : ListOf(A)s => Possibly(A)
= cases
  empty_l => no_value
  non_empty_l:list => the_value:list.head

sum_list : ListOf(Int)s => Int
= cases
  empty_l => 0
  non_empty_l:l => l.head + sum_list(l.tail)

print_err_or_res : Error(A)OrResult(B) => IO
= cases
  error:e => print("Error occurred: " + e -> to_string)
  result:r => print("All good! The result is: " + r -> to_string)

```

- *Description*

The values of an `or_type` are split into cases. Some cases have other values inside. The cases which have other values inside are followed by a semicolon and the type of the internal value. The same syntax can be used for matching that particular case in a function using the "cases" syntax. An `or_type` definition automatically creates prefix functions for each case with an internal value (which are simply conversions from the type of the internal value to the `or_type`). For example, for the case "`non_empty_l`" of a list, the function "`non_empty_l:`" is automatically created from the definition for which we can say:

```

non_empty_l:
  : NonEmptyListOf(A)s => ListOf(A)s

```

Similarly:

```

the_value:
  : A => Possibly(A)

```

These functions are called "prefix functions" because they are prepended to their argument. For example:

```

non_e_l : NonEmptyListOf(Int)s
= (1, [2, 3, 4])
>> non_empty_l:non_e_l
  : ListOf(Int)s
= [1, 2, 3, 4]

```

These functions can be used like any other function as arguments to other functions. For example:

```

non_empty_ls_to_ls : ListOf(NonEmptyListOf(A)s)s => ListOf(ListOf(A)s)s
= apply(non_empty_l:)to_each

```

- *Grammar*

```

⟨or-type-def⟩ ::=
  'or_type' ⟨type-name⟩ ⟨nl⟩
  'values' ⟨nl⟩ '␣' ⟨identifier⟩ [ ':' ⟨simple-type⟩ ] ( '␣|' ⟨identifier⟩ [ ':' ⟨simple-type⟩ ] ) *

```

4.1.3 Type Nicknames

- *Examples*

```

type_nickname Ints = ListOf(Int)s
type_nickname IntStringPairs = ListOf(Int x String)s
type_nickname IO = (EmptyVal)FromIO
type_nickname ErrOrRes(A) = Error(String)OrResult(A)
type_nickname Parse(A)FuncT = String => A x String

```

- *Description*

Type nicknames are used to abbreviate or give a more descriptive name to a type. They start with the keyword "type_nickname", followed by the nickname, then an equal sign and they end with the type to be nicknamed.

- *Grammar*

```

⟨t-nickname⟩ ::= 'type_nickname' ⟨type-name⟩ '␣=' ⟨simple-type⟩

```

4.2 Type Logic

Type logic is the mechanism for ad hoc polymorphism in lcases. The central notion of **type logic** is the **type proposition**. A type proposition is a proposition that has types as parameters and is true or false for particular type arguments.

Type propositions can either be defined or proven (for certain type arguments). Therefore, the following constructs exist and accomplish the aforementioned respectively: **type proposition definitions** and **type theorems**. These constructs are described in detail in the following sections. From this point onwards the "type" part will be omitted, i.e. propositions are always type propositions and theorems are always type theorems.

4.2.1 Proposition Definitions

Proposition definitions are split into definitions of **atomic propositions** and definitions of **renaming propositions** which are described in the following paragraphs.

Atomic Propositions

- *Examples*

```

type_proposition (A)Is(B)s_First
value
  first : B => A

type_proposition (T)Has_String_Repr
value
  to_string : T => String

type_proposition (T)Has_A_Wrapper

```

```

value
  wrapper : A => T(A)

type_proposition (T)Has_Internal_App
value
  apply()internally : (A => B) x T(A) => T(B)

```

The examples above define the following (ad hoc) polymorphic functions which have the respective (conditional) types:

```

first : (A)Is(B)s_First ==> B => A

to_string : (T)Has_String_Represantion ==> T => String

wrapper : (T)Has_A_Wrapper ==> A => T(A)

apply()internally : (T)Has_Internal_App ==> (A => B) x T(A) => T(B)

```

- *Description*

An atomic proposition definition defines simultaneously the **atomic proposition** itself and a **polymorphic value** (usually, but not necessarily, a function), by defining the form of the type of the value given the type parameters of the proposition. The proposition is true or false when the type parameters are substituted by specific type arguments depending on whether the implementation of the value has been defined for these type arguments. The aforementioned truthvalue determines whether the value is used correctly inside the program and therefore whether the program will typecheck. In order to add more types for which the function works, i.e. define the function for these types, i.e. make the proposition true for these types, one must prove a theorem. The specifics of theorems are described in the next section. For now, we'll show the example for everything mentioned in this paragraph for the proposition `(A)Is(B)s_First`:

- Proposition Definition:

```

type_proposition (A)Is(B)s_First
value
  first : B => A

```

- Function defined and its type:

```

first : (A)Is(B)s_First ==> B => A

```

- Theorems for specific types:

```

type_theorem (A)Is(A x B)s_First
proof
  first = .1st

type_theorem (A)Is(ListOf(A)s)s_First
proof
  first =
    cases
      empty_l => show_err("Tried to take the first element of an empty_l list")
      non_empty_l:l => l.head

```

- Usage of the function

```

pair, list
  : Int x String, ListOf(String)s
  = (42, "The answer to everything"), ["Hi!", "Hello", Heeey"]

>> pair -> first
  : Int
  = 42
>> list -> first
  : String
  = "Hi!"

```

An atomic proposition definition begins with the keyword `type_proposition` followed by the name of the proposition (including the type parameters) in the first line. The second line is the keyword `value`. The third line is indented once and has the identifier and the type expression of the value separated by the string `" : "`.

Renaming Propositions

- *Examples*

```

type_proposition (T)Has_Equality
equivalent
  (T)And(T)Can_Be_Equal

type_proposition (A)And(B)Are_Comparable
equivalent
  (A)Can_Be_Less_Than(B), (A)And(B)Can_Be_Equal, (A)Can_Be_Greater_Than(B)

type_proposition (T)Has_Comparison
equivalent
  (T)And(T)Are_Comparable

```

- *Description*

A renaming proposition definition is used to abbreviate one or the conjunction of many propositions (i.e. AND of all of them) into one new proposition.

A renaming proposition definition begins with the keyword `type_proposition` followed by the name of the proposition (including the type parameters) in the first line. The second line is the keyword `equivalent`. The third line is indented once and has either one proposition or (if it is a conjunction) many propositions separated by commas (where the commas essentially mean "and").

Grammar for Proposition Definitions

$\langle \text{type-prop-def} \rangle ::= \langle \text{atom-prop-def} \rangle \mid \langle \text{renaming-prop-def} \rangle$

$\langle \text{atom-prop-def} \rangle ::= \langle \text{prop-name-line} \rangle \langle \text{nl} \rangle \text{'value'} \langle \text{nl} \rangle \text{'_'} \langle \text{identifier} \rangle \text{'_:'} \langle \text{simple-type} \rangle$

$\langle \text{renaming-prop-def} \rangle ::= \langle \text{prop-name-line} \rangle \langle \text{nl} \rangle \text{'equivalent'} \langle \text{nl} \rangle \text{'_'} \langle \text{prop-name} \rangle (\langle \text{comma} \rangle \langle \text{prop-name} \rangle)^*$

$\langle \text{prop-name-line} \rangle ::= \text{'type_proposition_'} \langle \text{prop-name} \rangle$

$\langle \text{prop-name} \rangle ::=$
 $[A-Z] (\langle \text{name-part} \rangle \langle \text{params-in-paren} \rangle) + [\langle \text{name-part} \rangle]$
 $\mid (\langle \text{params-in-paren} \rangle \langle \text{name-part} \rangle) + [\langle \text{params-in-paren} \rangle]$

$\langle \text{name-part} \rangle ::= ([A-Za-z] \mid \text{'_'} [A-Z]) +$

4.2.2 Theorems

Theorems are split into theorems of **atomic propositions** and theorems of **implication propositions** which are described in the following paragraphs.

Atomic Propositions

- *Examples*

```
type_theorem (Possibly())Has_A_Wrapper
proof
  wrapper = the_value:

type_theorem (ListOf()s)Has_A_Wrapper
proof
  wrapper = x => [x]

type_theorem (Possibly())Has_Internal_App
proof
  apply()internally =
    (f, cases)
      no_value => no_value
      the_value:x => the_value:f(x)

type_theorem (ListOf()s)Has_Internal_App
proof
  apply()internally =
    (f, cases)
      empty_l => empty_l
      non_empty_l:l => non_empty_l:(f(l.head), l.tail -> apply(f)internally)
```

- *Usage*

```
a, b : all Possibly(Int)
      = wrapper(1), no_value

l1, l2, l3 : all ListOf(Int)s
            = wrapper(1), empty_l, [1, 2, 3]

>> a
   : Possibly(Int)
   = the_value:1
>> b
   : Possibly(Int)
   = no_value
>> l1
   : ListOf(Int)s
   = [1]
>> l2
   : ListOf(Int)s
   = []

>> a -> apply(x => x + 1)internally
   : Possibly(Int)
```

```

    = the_value:2
>> b -> apply(x => x + 1)internally
    : Possibly(Int)
    = no_value
>> l1 -> apply(x => x + 1)internally
    : ListOf(Int)s
    = [2]
>> l2 -> apply(x => x + 1)internally
    : ListOf(Int)s
    = []
>> l3 -> apply(x => x + 1)internally
    : ListOf(Int)s
    = [2, 3, 4]

```

- *Description*

A theorem of an atomic proposition proves the proposition for specific type arguments, by implementing the value associated to the proposition for these type arguments. Therefore, the value associated with the proposition can be used with all the combinations of type arguments for which the proposition is true, i.e. the combinations of type arguments for which the value has been implemented.

A proof of a theorem of an atomic proposition is correct when the implementation of the value associated with the proposition follows the form of the type given to the value by the definition of the proposition, i.e. the only difference between the type of the value in the theorem and the type of the value in the definition is that the type parameters of the proposition are substituted by the type arguments of the theorem.

A theorem of an atomic proposition begins with the keyword **type_theorem** followed by the name of the proposition with the type parameters substituted by the specific types for which the proposition will be proven. The second line is the keyword **proof**. The third line is indented once and it is the line in which the proof begins. The proof begins with the identifier of the value associated with the proposition and is followed by the string " = " and the value expression which implements the value.

Implication Propositions

- *Examples*

```

type_theorem (A)And(B)Can_Be_Equal => (A)And(B)Can_Be_Unequal
proof
  a \= b = not(a = b)

type_theorem (A)Can_Be_Greater_Than(B) => (A)Can_Be_Le_Or_Eq_To(B)
proof
  a <= b = not(a > b)

type_propositon (A)And(B)Have_Eq_And_Gr
equivalent
  (A)And(B)Can_Be_Equal, (A)Can_Be_Greater_Than(B)

type_theorem (A)And(B)Have_Eq_And_Gr => (A)Can_Be_Gr_Or_Eq_To(B)
proof
  a >= b = a = b | a > b

```

- *Description*

A theorem of an implication proposition is very similar to a theorem of an atomic proposition in the sense that it also implements a value in the proof. The difference is that the implementation uses another ad hoc polymorphic value (or many). Therefore, the implementation does not prove the proposition associated to the value it implements, because it assumes that the polymorphic value(s) used in the implementation is(are) already defined. In other words it proves the following: "if this(these) ad hoc polymorphic value(s) is(are) defined then we can also define this other one". This can be translated into the following implication proposition: "if the proposition associated to the value(s) we are using is true then the proposition associated to the value we are defining is true", which can be condensed to the notation with the " \Rightarrow " arrow used in the examples.

The proof of an implication proposition allows the compiler to automatically create the definition for an ad hoc polymorphic value for a particular combination of types given the definitions of the ad hoc polymorphic values used in the implementation for this same combination of types. This mechanism essentially gives definitions for free, that is in the sense that when you define a set of ad hoc polymorphic values for a particular set of types you get for free all the ad hoc polymorphic values that can be defined using a subset of the defined ones.

A theorem of an implication proposition is grammatically the same as a theorem of an atomic proposition with the only difference being that an implication proposition is comprised by two atomic propositions separated by the " \Rightarrow " arrow.

Grammar for Theorems

$\langle type-theo \rangle ::=$
 $\quad \text{'type_theorem' } \langle prop\text{-}name\text{-}sub \rangle [\text{' } \Rightarrow \text{' } \langle prop\text{-}name\text{-}sub \rangle] \langle nl \rangle$
 $\quad \text{'proof' } \langle nl \rangle \text{' } \langle identifier \rangle [\langle op \rangle \langle identifier \rangle] \text{' } \langle tt\text{-}value\text{-}expr \rangle$

$\langle prop\text{-}name\text{-}sub \rangle ::=$
 $\quad [A-Z] (\langle name\text{-}part \rangle \langle param\text{-}subs\text{-}in\text{-}paren \rangle) + [\langle name\text{-}part \rangle]$
 $\quad | (\langle param\text{-}subs\text{-}in\text{-}paren \rangle \langle name\text{-}part \rangle) + [\langle param\text{-}subs\text{-}in\text{-}paren \rangle]$

$\langle param\text{-}subs\text{-}in\text{-}paren \rangle ::= \text{'(' } \langle param\text{-}sub \rangle (\langle comma \rangle \langle param\text{-}sub \rangle)^* \text{'}'$

$\langle param\text{-}sub \rangle ::= \langle simple\text{-}type \rangle | \langle type\text{-}func \rangle$

$\langle type\text{-}func \rangle ::= [\text{'('}] \langle type\text{-}id \rangle (\text{'('} [A-Za-z] +) + [\text{'('}] | \text{'('} \langle type\text{-}id \rangle [\text{'('}] | \langle type\text{-}id \rangle \text{'('}$

$\langle tt\text{-}value\text{-}expr \rangle ::= \text{' } \langle line\text{-}expr \rangle | \langle nl \rangle \langle indent \rangle \langle big\text{-}or\text{-}cases\text{-}expr \rangle$

$\langle big\text{-}or\text{-}cases\text{-}expr \rangle ::= \langle big\text{-}op\text{-}expr \rangle | \langle big\text{-}func\text{-}expr \rangle | \langle cases\text{-}func\text{-}expr \rangle | \langle big\text{-}tuple \rangle | \langle big\text{-}list \rangle$

4.3 Complete Grammar for Types and Type Logic

$\langle type \rangle ::= [\langle condition \rangle] \langle simple\text{-}type \rangle$

$\langle simple\text{-}type \rangle ::= \langle type\text{-}id \rangle | \langle type\text{-}var \rangle | \langle func\text{-}type \rangle | \langle prod\text{-}type \rangle | \langle power\text{-}type \rangle | \langle type\text{-}app \rangle$

$\langle type\text{-}id \rangle ::= [A-Z] [A-Za-z]^+$

$\langle type\text{-}var \rangle ::= [A-Z]$

$\langle func\text{-}type \rangle ::= \langle in\text{-}or\text{-}out\text{-}type \rangle \text{' } \Rightarrow \text{' } \langle in\text{-}or\text{-}out\text{-}type \rangle$

$\langle in-or-out-type \rangle ::= \langle type-id \rangle \mid \langle type-var \rangle \mid \langle prod-type \rangle \mid \langle power-type \rangle \mid \langle type-app \rangle \mid \text{'('} \langle func-type \rangle \text{'}'$

$\langle prod-type \rangle ::= \langle field-type \rangle (\text{'x'} \langle field-type \rangle) +$

$\langle power-type \rangle ::= \langle field-type \rangle (\text{'^'} \langle int-greater-than-one \rangle) +$

$\langle field-type \rangle ::= \langle type-id \rangle \mid \langle type-var \rangle \mid \langle type-app \rangle \mid \text{'('} (\langle func-type \rangle \mid \langle prod-type \rangle) \text{'}'$

$\langle type-app \rangle ::=$
 $\quad [\langle types-in-paren \rangle] \langle type-id-with-args \rangle [\langle types-in-paren \rangle]$
 $\quad \mid \langle types-in-paren \rangle \langle type-id-or-var \rangle [\langle types-in-paren \rangle]$
 $\quad \mid \langle type-id-or-var \rangle \langle types-in-paren \rangle$

$\langle type-id-with-args \rangle ::= \langle type-id \rangle (\langle types-in-paren \rangle [A-Za-z]^+) +$

$\langle type-id-or-var \rangle ::= \langle type-id \rangle \mid \langle type-var \rangle$

$\langle types-in-paren \rangle ::= \text{'('} \langle simple-type \rangle (\langle comma \rangle \langle simple-type \rangle)^* \text{'}'$

$\langle condition \rangle ::= \langle prop-name \rangle \text{'_=>'}$

$\langle type-def \rangle ::= \langle tuple-type-def \rangle \mid \langle or-type-def \rangle$

$\langle tuple-type-def \rangle ::=$
 $\quad \text{'tuple_type_'} \langle type-name \rangle \langle nl \rangle$
 $\quad \text{'value' } \langle nl \rangle \text{'_'} \langle paren-comma-sep-ids \rangle \text{'_:'} (\langle prod-type \rangle \mid \langle power-type \rangle)$

$\langle type-name \rangle ::= [\langle params-in-paren \rangle] \langle type-id \rangle (\langle params-in-paren \rangle [A-Za-z]^+)^* [\langle params-in-paren \rangle]$

$\langle params-in-paren \rangle ::= \text{'('} \langle type-var \rangle (\langle comma \rangle \langle type-var \rangle)^* \text{'}'$

$\langle or-type-def \rangle ::=$
 $\quad \text{'or_type_'} \langle type-name \rangle \langle nl \rangle$
 $\quad \text{'values' } \langle nl \rangle \text{'_'} \langle identifier \rangle [\text{'_:'} \langle simple-type \rangle] (\text{'_'} \mid \text{'_'} \langle identifier \rangle [\text{'_:'} \langle simple-type \rangle])^*$

$\langle t-nickname \rangle ::= \text{'type_nickname_'} \langle type-name \rangle \text{'_='} \langle simple-type \rangle$

$\langle type-prop-def \rangle ::= \langle atom-prop-def \rangle \mid \langle renaming-prop-def \rangle$

$\langle atom-prop-def \rangle ::= \langle prop-name-line \rangle \langle nl \rangle \text{'value' } \langle nl \rangle \text{'_'} \langle identifier \rangle \text{'_:'} \langle simple-type \rangle$

$\langle renaming-prop-def \rangle ::= \langle prop-name-line \rangle \langle nl \rangle \text{'equivalent' } \langle nl \rangle \text{'_'} \langle prop-name \rangle (\langle comma \rangle \langle prop-name \rangle)^*$

$\langle prop-name-line \rangle ::= \text{'type_proposition_'} \langle prop-name \rangle$

$\langle \text{prop-name} \rangle ::=$
 $[A-Z] (\langle \text{name-part} \rangle \langle \text{params-in-paren} \rangle) + [\langle \text{name-part} \rangle]$
 $| (\langle \text{params-in-paren} \rangle \langle \text{name-part} \rangle) + [\langle \text{params-in-paren} \rangle]$
 $\langle \text{name-part} \rangle ::= ([A-Za-z] | \text{'_'}[A-Z]) +$
 $\langle \text{type-theo} \rangle ::=$
 $\text{'type_theorem' } \langle \text{prop-name-sub} \rangle [\text{'_=>'} \langle \text{prop-name-sub} \rangle] \langle \text{nl} \rangle$
 $\text{'proof' } \langle \text{nl} \rangle \text{'_'} \langle \text{identifier} \rangle [\langle \text{op} \rangle \langle \text{identifier} \rangle] \text{'_=' } \langle \text{tt-value-expr} \rangle$
 $\langle \text{prop-name-sub} \rangle ::=$
 $[A-Z] (\langle \text{name-part} \rangle \langle \text{param-subs-in-paren} \rangle) + [\langle \text{name-part} \rangle]$
 $| (\langle \text{param-subs-in-paren} \rangle \langle \text{name-part} \rangle) + [\langle \text{param-subs-in-paren} \rangle]$
 $\langle \text{param-subs-in-paren} \rangle ::= \text{'(' } \langle \text{param-sub} \rangle (\langle \text{comma} \rangle \langle \text{param-sub} \rangle)^* \text{'}'$
 $\langle \text{param-sub} \rangle ::= \langle \text{simple-type} \rangle | \langle \text{type-func} \rangle$
 $\langle \text{type-func} \rangle ::= [\text{'('}] \langle \text{type-id} \rangle (\text{'('} [A-Za-z] +) + [\text{'('}] | \text{'('} \langle \text{type-id} \rangle [\text{'('}] | \langle \text{type-id} \rangle \text{'('})$
 $\langle \text{tt-value-expr} \rangle ::= \text{'_'} \langle \text{line-expr} \rangle | \langle \text{nl} \rangle \langle \text{indent} \rangle \langle \text{big-or-cases-expr} \rangle$
 $\langle \text{big-or-cases-expr} \rangle ::= \langle \text{big-op-expr} \rangle | \langle \text{big-func-expr} \rangle | \langle \text{cases-func-expr} \rangle | \langle \text{big-tuple} \rangle | \langle \text{big-list} \rangle$

5 Language Description: Predefined

5.1 Values

- Constants: `undefined`, `pi`
- Functions
 - Miscellaneous: `not`, `id`, `show_error`
 - Numerical
 - * Miscellaneous: `sqrt`, `abs`, `max`, `min`
 - * Trigonometric: `sin`, `cos`, `tan`, `asin`, `acos`, `atan`
 - * Division related: `div`, `mod`, `gcd`, `lcm`, `even`, `odd`
 - * Rounding: `truncate`, `round`, `floor`, `ceiling`
 - * e and log: `exp`, `ln`, `log`
 - List:
 - `length`, `is_in`, `apply()``to_all`, `filter_with`, `take()``from`, `leave()``from`, `zip`, `unzip`, `zip_with`
 - IO
 - * Input: `get_char`, `get_line`, `get_input`, `read_file`
 - * Output: `print`, `print_string`, `print_line`, `write_file`
 - Ad Hoc Polymorphic:
 - `first`, `second`, `third`, `fourth`, `fifth`, `wrap`, `to_string`, `from_string`, `apply()``internally`, `wrapd_apply()``intern`

5.2 Types

- Basic: Int, Real, Char, String TODO? array: eg 10 * Int?
- Or Types: EmptyVal, Bool, Possibly(), ListOf()s, Result()OrError()
- Tuple Types: NonEmptyListOf()s
- Type Nicknames: IO

5.3 Type Propositions

- Operator Propositions:
 - (A)To_The(B)Has_Type(C)
 - (A)And(B)Multiply_To(C)
 - (A)Divided_By(B)Has_Type(C)
 - (A)And(B)Add_To(C)
 - (A)Minus(B)Has_Type(C)
 - (A)And(B)Can_Be_Equal
 - (A)And(B)Can_Be_Unequal
 - (A)Can_Be_Gr_Or_Eq_To(B)
 - (A)Can_Be_Le_Or_Eq_To(B)
 - (A)Can_Be_Greater_Than(B)
 - (A)Can_Be_Less_Than(B)
 - (T)Has_Use
 - (T)Has_Then
- Function Propositions:
 - (A)Is(B)s_First
 - (A)Is(B)s_Second
 - (A)Is(B)s_Third
 - (A)Is(B)s_Fourth
 - (A)Is(B)s_Fifth
 - (T)Has_A_Wrapper
 - (T)Has_String_Repr
 - (T)Can_Be_Parsed
 - (T)Has_Internal_App
 - (T)Has_Wrapd_Intern_App
- Renaming Propositions:
 - (T)Has_Exponentiation
 - (T)Has_Multiplication
 - (T)Has_Division
 - (T)Has_Addition
 - (T)Has_Subtraction

- (T)Has_Equality
- (T)Has_Non_Equality
- (T)Has_Greater_Or_Equal
- (T)Has_Less_Or_Equal
- (T)Has_Greater
- (T)Has_Less
- Theorems:
 - (A)Is(A x B)s_First
 - (A)Is(A x B x C)s_First
 - (A)Is(A x B x C x D)s_First
 - (A)Is(A x B x C x D x E)s_First
 - (A)Is(ListOf(A)s)s_First
 - (B)Is(A x B)s_Second
 - (B)Is(A x B x C)s_Second
 - (B)Is(A x B x C x D)s_Second
 - (B)Is(A x B x C x D x E)s_Second
 - (A)Is(ListOf(A)s)s_Second
 - (C)Is(A x B x C)s_Third
 - (C)Is(A x B x C x D)s_Third
 - (C)Is(A x B x C x D x E)s_Third
 - (A)Is(ListOf(A)s)s_Third
 - (D)Is(A x B x C x D)s_Fourth
 - (D)Is(A x B x C x D x E)s_Fourth
 - (A)Is(ListOf(A)s)s_Fourth
 - (E)Is(A x B x C x D x E)s_Fifth
 - (A)Is(ListOf(A)s)s_Fifth
 - TODO wrapper
 - (Int)Has_String_Repr
 - (Char)Has_String_Repr
 - (Real)Has_String_Repr
 - (A)Has_String_Repr => (ListOf(A)s)Has_String_Repr
 - TODO apply()internally
 - TODO wrapd_apply()intern
 - TODO ;>
 - TODO ;

6 Parser implementation

The parser was implemented using the parsec library.

- 6.1 AST Types
- 6.2 Parsers
- 7 Translation to Haskell
- 8 Running Examples
- 9 Conclusion