# Lambda Cases (lcases)

# Dimitris Saridakis

# Contents

1	Intr	roduction	3				
2	<b>Lan</b> 2.1	Language Description: General         2.1 Program Structure					
	$\frac{2.1}{2.2}$	Keywords	4				
	2.2	Ticy words	7				
3	Lan	guage Description: Values	5				
	3.1	Basic Expressions	5				
		3.1.1 Literals and Identifiers	5				
		3.1.2 Parenthesis, Tuples and Lists	6				
		3.1.3 Parenthesis Function Application	8				
		3.1.4 Prefix and Postfix Functions	9				
	3.2	Operators	11				
		3.2.1 Function Application and Function Composition Operators	11				
		3.2.2 Arithmetic, Comparison and Boolean Operators	12				
		3.2.3 Environment Action Operators	14				
		3.2.4 Operator Expressions	16				
		3.2.5 Complete Table, Precedence and Associativity	18				
	3.3	Function Expressions	20				
		3.3.1 Regular Function Expressions	20				
		3.3.2 "cases" Function Expressions	20				
	3.4	Value Definitions and "where" Expressions	22				
		3.4.1 Value Definitions	22				
		3.4.2 "where" Expressions	23				
	3.5	Indentation and Complete Grammar for Values	24				
		3.5.1 Indentation System	24				
		3.5.2 Complete Grammar for Values	25				
1	T	omega Description. Types and Type Lorie	27				
4	4.1	aguage Description: Types and Type Logic Types	27				
	4.1	4.1.1 Type Expressions	$\frac{27}{27}$				
		4.1.1 Type Expressions	31				
	4.2	Type Logic	$\frac{31}{34}$				
	4.2	4.2.1 Proposition Definitions	$\frac{34}{34}$				
		4.2.1 Theorems	36				
	4.3	Complete Grammar for Types and Type Logic	39				
	4.0	Complete Grammar for Types and Type Logic	00				
5	Lan	guage Description: Predefined	40				
	5.1	Values	40				
	5.2	Types					
	5.3	Type Propositions	41				

6	Parser implementation	<b>4</b> 1
	6.1 AST Types	41
	6.2 Parsers	41
7	Translation to Haskell	<b>4</b> 1
8	Running Examples	<b>4</b> 1
9	Conclusion	<b>4</b> 1

#### Introduction 1

Haskell is a delightful language. Yet, for some reason, it doesn't seem to have its rightful place in terms of popularity in industry. Why is it so? Is it inherently hard to learn and therefore only the brave dare to use it, or could it be that the syntax is perplexing to the amateur eye? It is my belief that with some syntax changes that give a greater familiarity to the new user, there would be no language more compelling than (the new) Haskell. In an attempt to achieve that familiarity, I present some (hopefully useful) new syntax, of which some is closer to the imperative/OOP style (to attract more already experienced programmers from these languages), some is closer to mathematics (in which most programmers should be experienced) and some is closer to natural language (in which we are all already experienced).

#### 2 Language Description: General

#### 2.1 **Program Structure**

An leases program consists of a set of definitions and theorems. Definitions are split into value definitions, type definitions and type proposition definitions. Theorems are proven type propositions. Functions as well as "Environment Actions" (see section 3.2.3) are also considered values. The definition of the "main" value determines the program's behaviour.

## Program example: Euclidean Algorithm

```
gcd
  : (Int, Int) => Int
  = (x, cases) =>
     0 \Rightarrow x
     y \Rightarrow gcd(y, x \rightarrow mod \leftarrow y)
read_two_ints
  : (Int x Int) IOAction
  = print <- "Please give me 2 ints";</pre>
     get_line ;> split_to_words o> cases =>
       [x, y] => io_wrapper([from_string(x), from_string(y)])
       ... => show_err("You didn't give me 2 ints")
tuple_type NumsAndGcd
value (x, y, gcd) : Int x Int x Int
print_gcd_message
  : NumsAndGcd => (EmptyVal)IOAction
  = nums_gcd => print(message)
     where
    message
       : String
       = "The GCD of " + nums_gcd.x + " and " + nums_gcd.y + " is = " + nums_gcd.gcd
main
  : (EmptyVal)IOAction
  = read_two_ints ;> ints =>
     print_gcd_message((ints.1st, ints.2nd, gcd(ints.1st, ints.2nd))
Program grammar
\langle program \rangle ::= (\langle value-def \rangle \mid \langle type-def \rangle \mid \langle type-prop-def \rangle \mid \langle type-theo \rangle) +
```

## 2.2 Keywords

The lcases keywords are the following:

cases where all tuple\_type value or\_type values type\_proposition equivalent type\_theorem proof

Each keyword's functionality is described in the respective section shown in the table below:

Keyword	Section
cases	3.3 Function Expressions
where all	3.4 Value Definitions
tuple_type value or_type values	4.1.2 Type Definitions
type_proposition value equivalent type_theorem proof	4.2 Type Logic

The "cases" and "where" keywords are also reserved words. Therefore, even though they can be generated by the "identifiers" grammar, they cannot be used as identifiers (see "Literals and Identifiers" section 3.1.1).

# 3 Language Description: Values

## 3.1 Basic Expressions

#### 3.1.1 Literals and Identifiers

#### Literals

• Examples

```
1 2 17 42 -100

1.61 2.71 3.14 -1234.567

'a' 'b' 'c' 'x' 'y' 'z' '.' ',' '\n'

"Hello World!" "What's up, doc?" "Alrighty then!"
```

 $\bullet$  Description

There are literals for the four basic types: Int, Real, Char, String. These are the usual integers, real numbes, characters and strings. The exact specification of literals is the same as in the Haskell report.

• Grammar

```
\langle literal \rangle ::= \langle literal \rangle
```

TODO add the grammar from the haskell report

#### **Identifiers**

• Examples

```
x y z
a1 a2 a3
funny_identifier
unnecessarily_long_identifier
apply()to_all
```

• Description

An identifier is the name of a value or a parameter. It is used in the definition of a value and in expressions that use that value, or in the parameters of a function and in the body of that function.

An identifier starts with a lower case letter and is followed by lower case letters or underscores. It is also possible to have pairs of parentheses in the middle of an identifier (see "Parenthesis Function Application" section 3.1.3 for why this can be useful). Finally, an identifier can be ended with a digit.

• Grammar

```
\langle identifier \rangle ::= [a-z] [a-z_]^* ( '() ' [a-z_] + )^* [ [0-9] ]
```

Even though the "cases" and "where" keywords can be generated by this grammar, they cannot be used as identifiers.

#### 3.1.2 Parenthesis, Tuples and Lists

#### **Parenthesis**

• Examples

```
(1 + 2)

(((1 + 2) * 3) ^ 4)

(val \rightarrow (x \Rightarrow f(x) + 1) \rightarrow to\_string \rightarrow (s \Rightarrow "f(val) + 1 is: " + s))

(do(3)times \leftarrow (get\_line ;> line \Rightarrow print("Line is: " + line)))
```

• Description

An expression is put in parenthesis to prioritize it or isolate it in a bigger (operator) expression. The expressions inside parenthesis are operator or function expressions.

• Grammar

```
\langle paren-expr \rangle ::= '(' \langle op-or-func-expr \rangle ')'
\langle op-or-func-expr \rangle ::= \langle simple-op-expr \rangle \mid \langle op-expr-func-end \rangle \mid \langle simple-func-expr \rangle
```

## **Tuples**

• Examples

```
(1, "What's up, doc?")
(2, "Alrighty then!", 3.14)
(x, y, z, w)
(1, my_function, (x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1/2))
```

• Description

Tuples are used to group many values (of possibly different types) into one. The type of a tuple can be either the product of the types of the fields or a defined tuple\_type which is equivalent to the afformentioned product type (see "Tuple Types" in section 4.1.2 for details). For example, the type of the second tuple above could be:

```
Int x String x Real
or:
MyType
assuming "MyType" has been defined in a similar way to the following:
tuple_type MyType
value (my_int, my_string, my_real) : Int x String x Real
• Big Tuples
```

## Example

```
my_big_tuple
  : String x Int x Real x String x String x (String x Real x Real)
  = ( "Hey, I'm the first field and I'm also a relatively big string."
   , 42, 3.14, "Hey, I'm the first small string", "Hey, I'm the second small string"
   , ("Hey, I'm a string inside the nested tuple", 2.71, 1.61)
   )
```

It is possible to stretch a (big) tuple expression over multiple lines (only) in a seperate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '(' is after the "=" part of the value definition and the first field must be in the same line.
- The tuple can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '(' character was in the first line.
- The tuple must be ended by a line that only contains the ')' character and is also indented so that the ')' is in same column where the '(' character was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.
- Grammar

```
 \langle tuple \rangle ::= `(` \langle line-expr \rangle `, \_' \langle comma-sep-line-exprs \rangle `)` \\ \langle comma-sep-line-exprs \rangle ::= \langle line-expr \rangle ( `, \_' \langle line-expr \rangle )* \\ \langle line-expr \rangle ::= \langle no-paren-op-arg \rangle \mid \langle op-or-func-expr \rangle \\ \langle big-tuple \rangle ::= \\ `(` \langle line-expr \rangle [ ``n' \langle indent \rangle ] `, \_' \langle comma-sep-line-exprs \rangle \\ (``n' \langle indent \rangle `, ` \langle comma-sep-line-exprs \rangle )* \\ ``n' \langle indent \rangle `)`
```

#### Lists

• Examples

```
[1, 2, 17, 42, -100]
[1.61, 2.71, 3.14, -1234.567]
["Hello World!", "What's up, doc?", "Alrighty then!"]
[x => x + 1, x => x + 2, x => x + 3]
[x, y, z, w]
```

• Description

Lists are used to group many values of the same type into one. The type of the list is ListOf(A)s where A is the type of every value inside. Therefore, the types of the first four examples are:

```
ListOf(Int)s
ListOf(Real)s
ListOf(String)s
(A)And(Int)Add_To(B) ==> ListOf(A => B)s
```

And the last list is only legal if x, y, z and w all have the same type. Assuming they do and it's the type T, the type of the list is:

ListOf(T)s

• Big Lists

#### Example

```
my_big_list
  : ListOf(Int => (EmptyVal)IOAction)s
  = [ x => print("I'm the first function and x + 1 is: " + (x + 1))
    , x => print("I'm the second function and x + 2 is: " + (x + 2))
    , x => print("I'm the third function and x + 3 is: " + (x + 3))
    ]
```

It is possible to stretch a (big) list expression over multiple lines (only) in a seperate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '[' is after the "=" part of the value definition and the first element must be in the same line.
- The list can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '[' character was in the first line.
- The tuple must be ended by a line that only contains the ']' character and is also indented so that the ']' is in same column where the '[' character was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.
- Grammar

```
\langle list \rangle ::= `[' [ \langle comma-sep-line-exprs \rangle ] `]'
\langle big-list \rangle ::= `[' \langle comma-sep-line-exprs \rangle ( `\n' \langle indent \rangle `, ' \langle comma-sep-line-exprs \rangle ) * `\n' \langle indent \rangle `]'
```

## 3.1.3 Parenthesis Function Application

• Examples

```
f(x)
f(x, y, z)
(x)to_string
apply(f)to_all
apply(f)to_all(1)
```

#### • Description

Function application in leases can be done in many different ways in an attempt to maximize readability. In this section, we discuss the ways function application can be done with parenthesis.

In the first two examples, we have the usual mathematical function application which is also used in most programming languages and should be familiar to the reader, i.e. function application is done with the arguments of the function in parenthesis separated by commas and **appended** to the function identifier.

We extend this idea by allowing the arguments to be **prepended** to the function identifier (third example). Finally, it is also possible to to have the arguments **inside** the function identifier provided the function has been **defined with parentheses inside the identifier**. For example, below is the definition of "apply()to\_all":

```
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
  = (f, cases) =>
    empty_l => empty_l
    non_empty_l:l => non_empty_l:(f <- l.head, apply(f)to_all <- l.tail)</pre>
```

The actual definition doesn't matter at this point, what matters is that the identifier is "apply()to\_all" with the parentheses **included**. This is very useful for defining functions where the argument in the middle makes the function application look and sound more like natural language.

It is possible to have many parentheses pairs in a single function application (last example). The arguments are always inserted to the function from **left to right**. Therefore, when multiple parentheses pairs are present the arguments of the leftmost parentheses are inserted first then the next ones to the right and so on.

#### • Grammar

#### 3.1.4 Prefix and Postfix Functions

#### **Prefix Functions**

• Examples

```
the_value:1
non_empty_1:1
error:e
result:r
apply(the_value:)to_all
```

#### • Description

Prefix functions are automatically generated from or\_type definitions (see "Or Types" in section 4.1.2). They are functions that convert a value of a particular type to a value that is a case of an or\_type and has values of this type inside. For example in the first example above we have:

```
1
  : Int
the_value:1
  : Possibly(Int)
```

Where the function thevalue: is automatically generated from the definition of the Possibly type:

```
or_type Possibly(A)
values the_value:A | no_value
And it has the type A => Possibly(A).
```

These functions are called prefix functions because they are prepended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function the\_value: is an argument of the function apply()to\_all. Prefix functions always end with a colon.

• Grammar

```
\langle pre-func \rangle ::= \langle identifier \rangle ':'
\langle pre-func-app \rangle ::= \langle pre-func \rangle \ ( \ \langle basic-expr \rangle \ | \ \langle paren-expr \rangle \ | \ \langle pre-func-app \rangle \ )
\langle basic-expr \rangle ::= \langle literal \rangle \ | \ \langle identifier \rangle \ | \ \langle literal \rangle \ | \ \langle literal \rangle \ | \ \langle paren-func-app \rangle \ | \ \langle post-func-app \rangle \ |
```

#### **Postfix Functions**

• Examples

```
name.first_name
list.head
date.year
tuple.1st
apply(.1st)to_all
```

 $\bullet$  Description

Postfix functions are automatically generated from tuple\_type definitions (see "Tuple Types" in section 4.1.2). They are functions that take a tuple\_type value and return a particular field (i.e. projection functions). For example in the first example above we have:

```
name
  : Name
name.first_name
  : String
Where the function .first name is automatically generated from the definition of the Name type:
tuple_type Name
value (first_name, last_name) : String x String
And it has the type Name => String.
There are also the following special projection functions that work on all product and tuple types:
.1st .2nd .3rd .4th .5th. For the 4th example above, assuming:
tuple
  : Int x String
We have:
tuple.1st
  : Int
The general types of these functions are:
  : (A)Is(B)s_1st ==> B => A
.2nd
  : (A)Is(B)s_2nd ==> B => A
```

These functions are called postfix functions because they are appended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function .1st is an argument of the function apply()to\_all. Postfix functions always begin with a dot.

• Grammar

```
\langle post\text{-}func \rangle ::= \text{`.'} \langle identifier \rangle \mid \text{`.1st'} \mid \text{`.2nd'} \mid \text{`.3rd'} \mid \text{`.4th'} \mid \text{`.5th'}
\langle post\text{-}func\text{-}app \rangle ::= (\langle paren\text{-}expr \rangle \mid \langle basic\text{-}expr \rangle) \langle post\text{-}func \rangle
```

## 3.2 Operators

## 3.2.1 Function Application and Function Composition Operators

## **Function Application Operators**

Operator	Type	
->	(A, A => B) => B	
<-	(A => B, A) => B	

The function application operators "->" and "<-" are a different way to apply functions to arguments than the usual parenthesis function application. They are meant to look like arrows that point from the argument to the function. These operators are very useful for chaining many function applications without the clutter of having to open and close parentheses for each one of the functions. For example, assuming we have the following functions with the behaviour suggested by their names and types:

```
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
string_length
  : String => Int
filter_with
  : (A => Bool, ListOf(A)s) => ListOf(A)s
is_odd
  : Int => Bool
sum_ints
  : ListOf(Int)s => Int
And a list of strings:
strings
  : ListOf(String)s
```

Here is a simple way to get the total number of characters in all the strings that have odd length:

```
chars_in_odd_length_strings
: Int
= strings -> apply(string_length)to_all -> filter_with(is_odd) -> sum_ints
```

Ofcourse this can be done equivalently using the other operator:

```
chars_in_odd_length_strings
: Int
= sum_ints <- filter_with(is_odd) <- apply(string_length)to_all <- strings</pre>
```

These operators can also be used together to put a function between two arguments if that function is commonly used that way in math (or if it looks better for a certain function). For example the "mod" function can be used like so:

Which is equivalent to:

#### **Function Composition Operators**

Operator	Туре	
0>	(A => B, B => C) => (A => C)	
<0	(B => C, A => B) => (A => C)	

The function composition operators "o>" and "<o" are used to compose functions, each one in the corresponding direction. The use of the letter 'o' is meant to be similar to the mathematical function composition symbol 'o' and the symbols '>', '<' are used so that the operator points from the function which is applied first to the function which is applied second. A neat example using function composition is the following. Assuming we have the following functions with the behaviour suggested by their names and types:

```
split_to_words
  : String => ListOf(String)s
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
reverse_string
  : String => String
merge_words
  : ListOf(String)s => String
```

We can reverse the all the words in a string like so:

reverse\_words

- : String => String
- = split\_to\_words o> apply(reverse\_string)to\_all o> merge\_words

Ofcourse this can be done equivalently using the other operator:

reverse\_words

- : String => String
- = merge\_words <o apply(reverse\_string)to\_all <o split\_to\_words

## 3.2.2 Arithmetic, Comparison and Boolean Operators

#### **Arithmetic Operators**

Operator	Type
^	$(A)$ To_The $(B)$ Has_Type $(C) ==> (A, B) => C$
*	(A)And(B)Multiply_To(C) ==> (A, B) => C
/	(A)Divided_By(B)Has_Type(C) $\Longrightarrow$ (A, B) $\Longrightarrow$ C
+	$(A)And(B)Add_To(C) \Longrightarrow (A, B) \Longrightarrow C$
-	$(A)Minus(B)Has\_Type(C) \Longrightarrow (A, B) \Longrightarrow C$

The usual arithmetic operators work as they are expected, similarly to mathematics and other programming languages for the usual types. However, they are generalized. The examples below show their generality:

```
= 4.14
>> 'a' + 'b'
  : String
  = "ab"
>> 'w' + "ord"
  : String
 = "word"
>> "Hello " + "World!"
  : String
  = "Hello World!"
>> 5 * 'a'
  : String
  = "aaaaa"
>> 5 * "hi"
  : String
  = "hihihihihi"
>> "1,2,3" - ','
  : String
  = "123"
```

Let's analyze further the example of addition. The type can be read as such: the '+' operator has the type  $(A, B) \Rightarrow C$ , provided that the type proposition  $(A)And(B)Add\_To(C)$  holds. This proposition being true, means that addition has been defined for these three types (see section "Type Logic" 4.2 for more on type propositions). For example, by the examples above we can deduce that the following propositions are true (in the order of the examples):

```
(Int)And(Int)Add_To(Int)
(Int)And(Real)Add_To(Real)
(Char)And(Char)Add_To(String)
(Char)And(String)Add_To(String)
(Int)And(Char)Multiply_To(String)
(Int)And(String)Multiply_To(String)
(String)Minus(Char)Has_Type(String)
```

This allows us to use the familiar arithmetic operators in types that are not necessarily numbers but it is somewhat intuitively obvious what the should do in those other types. Furthermore, their behaviour can be defined by the user for new user defined types!

## Comparison and Boolean Operators

Operator	Type
=	$(A)And(B)Can\_Be\_Equal ==> (A, B) => Bool$
/=	$(A)And(B)Can\_Be\_Unequal ==> (A, B) => Bool$
>=	$(A)Can_Be_Gr_Or_Eq_To(B) \Longrightarrow (A, B) \Longrightarrow Bool$
<=	$(A)$ Can_Be_Le_Or_Eq_To $(B) \Longrightarrow (A, B) \Longrightarrow Bool$
>	$(A)$ Can_Be_Greater_Than $(B) \Longrightarrow (A, B) \Longrightarrow Bool$
<	$(A)Can_Be_Less_Than(B) \Longrightarrow (A, B) \Longrightarrow Bool$
&	(Bool, Bool) => Bool

Comparison operators are also generalized. The main reason for the generalization is to be able to compare numbers of different types. Consider the following example:

```
>> 1
: Int
= 1
>> 1.1
: Real
= 1.1
>> 1.1 = 1
: Bool
= false
```

In order for the example to work we need to be able to compare integers and reals. Similarly, all the comparison operators need to be able to work on arguments of different types. Other than that, the comparison and boolean operators behave similarly to most programming languages. It's also worth noting that the "equals", "and" and "or" operators have the symbol once (= & | ) instead of twice (== && | ).

#### 3.2.3 Environment Action Operators

Operator	Type	
;>	(E)Has_Use ==> (E(A), A => E(B)) => E(B)	
;	$(E)$ Has_Then $\Longrightarrow$ $(E(A), E(B)) \Longrightarrow E(B)$	

#### Simple Example

```
print_string("I'll repeat the line.") ; get_line ;> print_string
```

The example above demonstrates the use of the environment action operators with the IOAction type, which is how IO is done in leases. Some light can be shed on how this is done, if we take a look at the types (as always!):

```
print_string
  : String => (EmptyVal)IOAction
print_string("I'll repeat the line.")
        : (EmptyVal)IOAction
get_line
        : (String)IOAction
;
        : (E)Has_Then ==> (E(A), E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line
        : (String)IOAction
        where (IOAction)Has_Then is true, E = IOAction, A = EmptyVal, B = String
;>
        : (E)Has_Use ==> (E(A), A => E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line ;> print_string
        : (EmptyVal)IOAction
        where (IOAction)Has_Use is true, E = IOAction, A = String, B = EmptyVal
```

#### Example program

```
main
  : (EmptyVal)IOAction
  = print_string <- "Hello! What's your name?" ; get_line ;> name =>
    print_string("Oh hi " + name + "! What's your age?") ; get_line ;> age =>
    print_string("Oh that's crazy " + name + "! I didn't expect you to be " + age + "!");
```

In this bigger but similar example the types are:

```
print_string
  : String => (EmptyVal)IOAction
get_line
  : (String) IOAction
print_string <- "Hello! ... "</pre>
  : (EmptyVal)IOAction
print_string("Oh hi...)
  : (EmptyVal)IOAction
print_string("Oh that's crazy...)
  : (EmptyVal)IOAction
   (E)Has_Then \Longrightarrow (E(A), E(B)) \Longrightarrow E(B)
print_string("Oh hi...) ; get_line
  : (String) IOAction
  where (IOAction) Has_Then is true, E = IOAction, A = EmptyVal, B = String
age => print_string("Oh that's crazy...)
  : String => (EmptyVal)IOAction
  : (E)Has_Use \Longrightarrow (E(A), A <math>\Longrightarrow E(B)) <math>\Longrightarrow E(B)
print_string("Oh hi...) ; get_line ;> age =>
print_string("Oh that's crazy...)
  : (EmptyVal)IOAction
  where (IOAction) Has_Use is true, E = IOAction, A = String, B = EmptyVal
print_string <- "Hello..." ; get_line</pre>
  : (String) IOAction
name => print_string("Oh hi ... (till the end)
  : String => (EmptyVal)IOAction
print_string <- "Hello..." ; get_line ;> name =>
print_string("Oh hi ... (till the end)
  : (EmptyVal)IOAction
```

Therefore, "main: (EmptyVal)IOAction" checks out. The key here is to remember that function expressions extend to the end of the whole expression. Therefore, we have "name => ... (till the end)" and "age => ... (till the end)" as the second arguments of the two occurrences of the ";>" operator.

### Description

The environment action operators are used to combine values that do environment actions into values that do more complicated environment actions. Environment actions are type functions that take a type argument and produce a type (just like ListOf()s). These type functions have the "then" operator (;) and the "use" operator (;>) defined for them. A value of the type E(A) where (E)Has\_Then does an environment action of type E that produces a value of type A which can then be combined with another one with the "then" operator. Similarly, with the "use" operator the produced value of an action can be used by a function that returns another action.

The effect of the ";" operator described in words is the following: given a value of type E(A) and a value of

type E(B) (which do environment actions that produce values of type A and B respectively), create a new value the does both actions (provided the first did not result in an error). The overall effect is a value that does an environment action of type E (the combination of the "smaller" actions) which produces a value of type B (the one produced by the second action) and therefore it is of type E(B).

Note that the value of type A produced by the first action is not used anywhere. This happens mostly when A = EmptyVal and it is because values of type E(EmptyVal) are used for their environment action only (e.g. print string(...) : (EmptyVal)IOAction).

How the two environment actions of the E(A) and E(B) values are combined to produce the new environment action is specific to the environment action type E.

The effect of the ";>" operator described in words is the following: given a value of type E(A) (which does an environment action of type E(A) that produces a value of type E(A) and a value of type E(B) (which is a function that takes a value of type E(A) and returns an environment action of type E(A) that produces a value of type E(A), combine those two values by creating a value that does the following:

- Performs the first action that produces a value of type A
- Takes the value of type A produced (provided there was no error) and passes it to the function of type A => E(B) that then returns an action
- Perfoms the resulting action

The overall effect is an environment action of type E that in the end produces a value is of type B and therefore the new value is of type E(B).

#### 3.2.4 Operator Expressions

• Examples

```
1 + 2
1 + x * 3 ^ y
"Hello " + "World!"
x -> f -> g
f o> g o> h
x = y
x >= y - z & x < 2 * y
get_line; get_line; > line => print("Second line: " + line)
```

#### • Description

Operator expressions are expressions that use operators. Operators act like two-argument-functions that are placed in between their arguments. Therefore, they have function types and they act as it is described in their respective sections above this one.

An operator expression might have multiple operators. The order of operations is explained in the next section ("Complete Table, Precedence and Associativity") in Table 2.

Just like functions, the sub-expressions that act as arguments to an operator, must have types that match the types expected by the operator.

It is possible to the second argument of an operator is a function expression. This is mostly useful with the ";>" operator (see previous section: "Environment Operators"), but it is also possible with the following operators: "->", "o>", "<o".

#### • Big Operator Expressions

It is possible to stretch a (big) operator expression over multiple lines. In that case:

- The operator expression must split in a new line after an operator (not an argument).
- Every line after the first must be indented so that in begins at the column where the first line of the operator expression begun.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

#### $\bullet$ Grammar

```
 \langle op\text{-}expr\rangle ::= \langle simple\text{-}op\text{-}expr\rangle \mid \langle op\text{-}expr\text{-}func\text{-}end\rangle \mid \langle big\text{-}op\text{-}expr\rangle \mid \langle cases\text{-}op\text{-}expr\rangle } 
 \langle simple\text{-}op\text{-}expr\rangle ::= \langle op\text{-}arg\rangle ( \text{`$\sqcup$'} \langle op\rangle \text{`$\sqcup$'} \langle op\text{-}arg\rangle ) + 
 \langle op\text{-}expr\text{-}func\text{-}end\rangle ::= \langle simple\text{-}op\text{-}expr\rangle \text{`$\sqcup$'} \langle op\rangle \text{`$\sqcup$'} \langle simple\text{-}func\text{-}expr\rangle } 
 \langle big\text{-}op\text{-}expr\rangle ::= \langle op\text{-}expr\text{-}line\rangle ( \text{`$\backslash n$'} \langle indent\rangle \langle op\text{-}expr\text{-}line\rangle )^* \\ \text{`$\backslash n$'} \langle indent\rangle ( \langle op\text{-}arg\rangle \mid \langle simple\text{-}op\text{-}expr\rangle ) [ \langle op\text{-}expr\text{-}line\rangle \text{`$\sqcup '} ] ( \langle simple\text{-}func\text{-}expr\rangle \mid \langle big\text{-}func\text{-}expr\rangle ) ) } 
 \langle cases\text{-}op\text{-}expr\rangle ::= \langle op\text{-}expr\text{-}line\rangle ( \text{`$\backslash n$'} \langle indent\rangle \langle op\text{-}expr\text{-}line\rangle )^* ( \text{`$\backslash n$'} \langle indent\rangle \mid \text{`$\sqcup '} ) \langle cases\text{-}func\text{-}expr\rangle } 
 \langle op\text{-}expr\text{-}line\rangle ::= ( \langle op\text{-}arg\rangle \mid \langle simple\text{-}op\text{-}expr\rangle ) \text{`$\sqcup '} \langle op\rangle } 
 \langle op\text{-}arg\rangle ::= \langle no\text{-}paren\text{-}op\text{-}arg\rangle \mid \langle paren\text{-}expr\rangle } \langle op\text{-}expr\text{-}line\rangle \mid \langle pre\text{-}func\text{-}opp\rangle } 
 \langle op\text{-}arg\rangle ::= \langle basic\text{-}expr\rangle \mid \langle pre\text{-}func\rangle \mid \langle post\text{-}func\rangle \mid \langle pre\text{-}func\text{-}app\rangle } 
 \langle op\rangle ::= \text{`$->'} \mid \text{`$-'} \mid \text{`$o>'} \mid \text{`$o>'} \mid \text{`$-'} \mid
```

# 3.2.5 Complete Table, Precedence and Associativity

Table 1: The complete table of leases operators along with their types and their short descriptions.

Operator	Туре	Description
->	(A, A => B) => B	Right function application
<-	(A => B, A) => B	Left function application
0>	(A => B, B => C) => (A => C)	Right function composition
<0	(B => C, A => B) => (A => C)	Left function composition
^	$(A)$ To_The $(B)$ Has_Type $(C) \Longrightarrow (A, B) \Longrightarrow C$	General exponentiation
*	(A)And(B)Multiply_To(C) $\Longrightarrow$ (A, B) $\Longrightarrow$ C	General multiplication
/	(A)Divided_By(B)Has_Type(C) $\Longrightarrow$ (A, B) $\Longrightarrow$ C	General division
+	$(A)And(B)Add_To(C) \Longrightarrow (A, B) \Longrightarrow C$	General addition
_	$(A)Minus(B)Has\_Type(C) ==> (A, B) => C$	General subtraction
=	$(A)And(B)Can\_Be\_Equal ==> (A, B) => Bool$	Equality
/=	$(A)And(B)Can\_Be\_Unequal ==> (A, B) => Bool$	Inequality
>=	$(A)Can\_Be\_Gr\_Or\_Eq\_To(B) \Longrightarrow (A, B) \Longrightarrow Bool$	Greater than or equal to
<=	$(A)$ Can_Be_Le_Or_Eq_To $(B) \Longrightarrow (A, B) \Longrightarrow Bool$	Less than or equal to
>	$(A)$ Can_Be_Greater_Than $(B) \Longrightarrow (A, B) \Longrightarrow Bool$	Greater than
<	$(A)Can\_Be\_Less\_Than(B) \Longrightarrow (A, B) \Longrightarrow Bool$	Less than
& I	(Bool, Bool) => Bool	Boolean operators
;>	$(E)Has\_Use \Longrightarrow (E(A), A \Longrightarrow E(B)) \Longrightarrow E(B)$	Do, use, do
;	(E)Has_Then $\Longrightarrow$ (E(A), E(B)) $\Longrightarrow$ E(B)	Do then do

The order of operations is done from highest to lowest precedence. In the same level of precedence the order is done from left to right if the associativity is "Left" and from right to left if the associativity is "Right". For the operators that have associativity "None" it is not allowed to place them in the same operator expression. The precedence and associativity of the operators is shown in the table below.

Table 2: The table of precedence and associativity of the leases operators.

Operator	Precedence	Associativity
->	10 (highest)	Left
<-	9	Right
0> <0	8	Left
^	7	Right
* /	6	Left
+ -	5	Left
= /= > < >= <=	4	None
&	3	Left
I	2	Left
;> ;	1	Left

## 3.3 Function Expressions

```
\langle func\text{-}expr \rangle ::= \langle simple\text{-}func\text{-}expr \rangle \mid \langle big\text{-}func\text{-}expr \rangle \mid \langle cases\text{-}func\text{-}expr \rangle
```

#### 3.3.1 Regular Function Expressions

• Examples

```
a \Rightarrow 17 * a + 42
(x, y, z) \Rightarrow (x^2 + y^2 + z^2)^{(1/2)}
```

#### • Description

Regular function expressions are used to define functions or be part of bigger expressions as anonymous functions. They are comprised by their parameters and their body. A parameter has an identifier. The parameters are either only one, in which case there is no parenthesis, or they are many, in which case they are in parenthesis, seperated by commas. The parameters and the body are seperated by an arrow (" => "). The body is an operator expression.

• Big Function Expressions

It is possible to stretch a (big) function expression over multiple lines. In that case:

- The function expression must split in a new line after the "=>" arrow.
- Every line after the first must be indented so that in begins at the column where the first character of the parameters was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.
- Grammar

### 3.3.2 "cases" Function Expressions

• Examples

```
print_sentimental_bool
  : Bool => (EmptyVal)IOAction
  = cases =>
     true => print <- "It's true!! :)"
     false => print <- "It's false... :("

or_type TrafficLight
values green | amber | red

print_sentimental_traffic_light
  : TrafficLight => (EmptyVal)IOAction
  = cases =>
     green => print <- "It's green! Let's go!!! :)"</pre>
```

```
amber => print <- "Go go go, fast!"
    red => print <- "Stop right now! You're going to kill us!!"
is_not_red
  : TrafficLight => Bool
  = cases =>
    green => true
    amber => true
    red => false
is_seventeen_or_forty_two
  : Int => Bool
  = cases =>
    17 => true
    42 => true
    ... => false
traffic_lights_match
  : (TrafficLight, TrafficLight) => Bool
  = (cases, cases) =>
    (green, green) => true
    (amber, amber) => true
    (red, red) => true
    ... => false
gcd
  : (Int, Int) => Int
  = (x, cases) =>
    0 \Rightarrow x
    y \Rightarrow gcd(y, x \rightarrow mod \leftarrow y)
is_empty
  : ListOf(A)s => Bool
  = cases =>
    empty_1 => true
    non_empty_l:anything => false
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
  = (f, cases) =>
    empty_1 => empty_1
    non_empty_l:list => non_empty_l:(f <- list.head, apply(f)to_all <- list.tail)</pre>
```

"cases" is a keyword that works as a special parameter. The difference is that instead of giving the name "cases" to that parameter, it allows the programmer to pattern match on the possible values of that parameter and return a different result for each particular case.

The last case can be "... => (body of default case)" to capture all remaining cases while dismissing the value (e.g. is\_seventeen\_or\_forty\_two example), or it can be "some\_id => (body of default case)" to capture all remaining cases while being able to use the value with the name "some\_id" (e.g. "y" in gcd example).

It is possible to use the "cases" keyword in multiple parameters to match on all of them. By doing that, each case represents a particular combination of values for the parameters involved (e.g. traffic\_lights\_match example).

It is also possible to use a "where" expression below a particular case. The "where" expression must be indented two spaces more than than the line where that particular case begins.

A function expression that uses the "cases" syntax must contain the "cases" keyword in at least one parameter. The number of matching expressions in all cases must be the same as the number of parameters with the "cases" keyword.

#### • Grammar

## 3.4 Value Definitions and "where" Expressions

## 3.4.1 Value Definitions

• Examples

```
foo
    : Int
    = 42

f
    : (Int, Int, Int) => Int
    = (a, b, c) => a + b * c

val1, val2, val3
    : Int, Bool, Char
    = 42, true, 'a'

int1, int2, int3
    : all Int
    = 1, 2, 3
```

Value definitions are the main building block of leases programs. To define a new value you give it a name, a type and an expression. The name is an identifier in the first line. The second line is indented two spaces more and begins by ": " and continues with the type expression. The third line is indented as the second, begins by "=" and continues with the value expression (which extends to as many lines as needed).

A value definition is either in the first column, where it can be "seen" by all other value definitions, or it is in a "where" expression (see section below), where it can be "seen" by the expression above the "where" and all the other definitions in the same "where" expression.

A value definition can be followed by a "where" expression where intermediate values used in the value expression are defined. In that case, the "where" expression must be indented two spaces more than the "=" line of the value definition.

It is possible to group value definitions together by seperating the names, the types and the expressions with commas. This is very useful for not cluttering the program with many definitions for values with small expressions (e.g. constants). When grouping definitions together it is also possible to use the keyword "all" to give the same type to all the values.

#### • Grammar

## 3.4.2 "where" Expressions

• Examples

```
sort :
 ListOf(Int)s => ListOf(Int)s
  = cases =>
    empty_l => empty_l
    non_empty_l:l => sort(less_l) + l.head + sort(greater_l)
      less_l, greater_l
        : all ListOf(Int)s
        = filter_with(x => x < l.head, l.tail)
        , filter_with(x \Rightarrow x \Rightarrow 1.head, 1.tail)
sum_nodes
  : TreeOf(Int)s => Int
  = tree => tree.root + tree.subtrees -> apply(sum_nodes)to_all -> sum_list
    where
    sum_list
      : ListOf(Int)s => Int
      = cases =>
        empty_1 => 0
        non_empty_1:1 => 1.head + sum_list(1.tail)
```

```
big_string
: String
= s1 + s2 + s3 + s4
  where
s1, s2, s3, s4 : all String
= "Hello, my name is Struggling Programmer."
, "I have tried way too many times to fit a big chunk of text"
, "inside my program, without it hitting the half-screen mark!"
, "I am so glad I finally discovered lcases!"
```

"where" expressions allow the programmer to use values inside an expression and define them below it. They are very useful for reusing or abbreviating expressions that are specific to a particular definition or case.

A "where" expression begins by a line that only has the word "where" in it. It is indented as described in the "Value Definitions" (3.4.1) or "cases' Function Expressions" (3.3.2) sections. The definitions are placed below the "where" line and must have the same indentation.

• Grammar

```
\langle where\text{-}expr \rangle ::= \text{`\n'} \langle indent \rangle \text{ `where\n'} (\langle value\text{-}def \rangle | \langle grouped\text{-}value\text{-}defs \rangle) +
```

## 3.5 Indentation and Complete Grammar for Values

#### 3.5.1 Indentation System

The < indent> nonterminal in not a normal BNF nonterminal. It is a context sensitive construct that enforces the indentation rules of leases. It depends on a integer value that shall be named "indentation level" (il). The < indent> nonterminal corresponds to 2\*il space characters. The indentation level follows the rules below:

#### **Indentation Rules**

- At the beginning: il = 0.
- At the end of the first line of a value definition:  $il \leftarrow il + 1$ . This also applies to grouped value definitions.
- At the end of the third line ("=" line) of a (single) value definition:  $il \leftarrow il + 1$ .
- At the end of a (single) value definition:  $il \leftarrow il 2$ .
- At the end of grouped value definitions:  $il \leftarrow il 1$ .
- After the "=>" arrow in a case:  $il \leftarrow il + 1$ .
- At the end of a case body:  $il \leftarrow il 1$ .
- In a cases function expression which does not begin at the "=" line of a value definition:
  - After the arrow "=>" at the end of the paremeters:  $il \leftarrow il + 1$ .
  - At the end of the whole cases function expression:  $il \leftarrow il-1$ .

#### 3.5.2 Complete Grammar for Values

```
\langle literal \rangle ::= TODO
(identifier) ::= [a-z] [a-z]^* ( '()' [a-z] + )^* [ [0-9] ]
\langle paren-expr \rangle ::= (( \langle op-or-func-expr \rangle))
\langle op\text{-}or\text{-}func\text{-}expr \rangle ::= \langle simple\text{-}op\text{-}expr \rangle \mid \langle op\text{-}expr\text{-}func\text{-}end \rangle \mid \langle simple\text{-}func\text{-}expr \rangle
\langle tuple \rangle ::= `(` \langle line\text{-}expr \rangle `, \Box ` \langle comma\text{-}sep\text{-}line\text{-}exprs \rangle `)`
\langle comma\text{-}sep\text{-}line\text{-}expr\rangle ::= \langle line\text{-}expr\rangle \ (\text{`,}_{\sqcup}, \langle line\text{-}expr\rangle)^*
\langle line\text{-}expr \rangle ::= \langle no\text{-}paren\text{-}op\text{-}arg \rangle \mid \langle op\text{-}or\text{-}func\text{-}expr \rangle
\langle big\text{-}tuple \rangle ::=
         '(' \langle line\text{-}expr \rangle [ '\n' \langle indent \rangle ] ',\Box' \langle comma\text{-}sep\text{-}line\text{-}exprs \rangle
         ( '\n' \langle indent \rangle ', ' \langle comma-sep-line-exprs \rangle )^*
         '\n' \langle indent \rangle ')
\langle list \rangle ::= '[' [\langle comma-sep-line-exprs \rangle]']'
\langle big\text{-}list \rangle ::= \text{`['}\langle comma\text{-}sep\text{-}line\text{-}exprs \rangle ('\n' \langle indent \rangle ', '\langle comma\text{-}sep\text{-}line\text{-}exprs \rangle)^* '\n' \langle indent \rangle ']'}
\langle paren-func-app \rangle ::=
           [\langle arguments \rangle] \langle identifier\text{-}with\text{-}arguments \rangle [\langle arguments \rangle]
           \langle arguments \rangle \langle identifier \rangle [\langle arguments \rangle]
           \langle identifier \rangle \langle arguments \rangle
\langle arguments \rangle ::= '(' \langle comma-sep-line-exprs \rangle ')'
\langle identifier\text{-}with\text{-}arguments \rangle ::=
           [a-z] [a-z_]* ( '()'[a-z_]+ )* ( arguments) [a-z_]+ ( ( '()' | ( arguments) ) [a-z_]+ )* [ [0-9] ]
\langle pre\text{-}func \rangle ::= \langle identifier \rangle ':'
\langle pre-func-app \rangle ::= \langle pre-func \rangle (\langle basic-expr \rangle \mid \langle paren-expr \rangle \mid \langle pre-func-app \rangle)
\langle basic\text{-}expr \rangle ::= \langle literal \rangle \mid \langle identifier \rangle \mid \langle tuple \rangle \mid \langle list \rangle \mid \langle paren-func\text{-}app \rangle \mid \langle post\text{-}func\text{-}app \rangle
\langle post\text{-}func \rangle ::= `.` \langle identifier \rangle
\langle post\text{-}func\text{-}app \rangle ::= (\langle paren\text{-}expr \rangle \mid \langle basic\text{-}expr \rangle) \langle post\text{-}func \rangle
\langle op\text{-}expr \rangle ::= \langle simple\text{-}op\text{-}expr \rangle \mid \langle op\text{-}expr\text{-}func\text{-}end \rangle \mid \langle biq\text{-}op\text{-}expr \rangle \mid \langle cases\text{-}op\text{-}expr \rangle
```

```
\langle simple-op-expr \rangle ::= \langle op-arg \rangle \ (`` \sqcup ` \langle op \rangle ` \sqcup ` \langle op-arg \rangle \ )+
\langle op\text{-}expr\text{-}func\text{-}end \rangle ::= \langle simple\text{-}op\text{-}expr \rangle `\ \Box' \langle op \rangle `\ \Box' \langle simple\text{-}func\text{-}expr \rangle
\langle big\text{-}op\text{-}expr\rangle ::=
                       \langle op\text{-}expr\text{-}line \rangle ::= (\langle op\text{-}arg \rangle \mid \langle simple\text{-}op\text{-}expr \rangle) `\Box' \langle op \rangle
\langle op\text{-}arg \rangle ::= \langle no\text{-}paren\text{-}op\text{-}arg \rangle \mid \langle paren\text{-}expr \rangle
\langle no\text{-}paren\text{-}op\text{-}arq \rangle ::= \langle basic\text{-}expr \rangle \mid \langle pre\text{-}func \rangle \mid \langle post\text{-}func \rangle \mid \langle pre\text{-}func\text{-}app \rangle
\langle \mathit{op} \rangle ::= \text{`--'} | \text{``<-'} | \text{`o>'} | \text{``<o'} | \text{``*'} | \text{`*'} | \text{`+'} | \text{`-'} | \text{`='} | \text{`/='} | \text{`>'} | \text{`<'} | \text{`>='} | \text{`<='} | \text{`\&'} | \text{`I'} | \text{`;} \text{`;'} | \text{`;'} | \text{`;'} | \text{``,'} | | \tau,' | \tau,' | \tau,
\langle func\text{-}expr \rangle ::= \langle simple\text{-}func\text{-}expr \rangle \mid \langle big\text{-}func\text{-}expr \rangle \mid \langle cases\text{-}func\text{-}expr \rangle
\langle simple-func-expr \rangle ::= \langle parameters \rangle ` =>   ' \langle simple-func-body \rangle
\langle biq\text{-}func\text{-}expr \rangle ::= \langle parameters \rangle ' = > \ ' \langle indent \rangle \ ( \langle simple\text{-}func\text{-}body \rangle \mid \langle biq\text{-}op\text{-}expr \rangle )
\langle parameters \rangle ::= \langle identifier \rangle \mid (\langle identifier \rangle (\langle identifier \rangle ) + (\langle identifie
\langle simple-func-body \rangle ::= \langle no-paren-op-arq \rangle \mid \langle simple-op-expr \rangle \mid \langle op-expr-func-end \rangle
\langle cases\text{-}func\text{-}expr \rangle ::= \langle cases\text{-}parameters \rangle ` \Box = \rangle ' \langle case \rangle + \langle end\text{-}case \rangle
\langle cases\text{-}parameter \rangle ::= \langle cases\text{-}parameter \rangle \mid \text{`('} \langle cases\text{-}parameter \rangle \mid \text{`,}_{\square} \langle cases\text{-}parameter \rangle \mid \text{')'}
\langle cases\text{-}parameter \rangle ::= \langle parameter \rangle \mid \text{`cases'}
\langle case \rangle ::= \text{`\n'} \langle indent \rangle \langle matching \rangle \text{`}_{\sqcup} => \text{'} \langle case\text{-}body \rangle
\langle end\text{-}case \rangle ::= \text{`\n'} \langle indent \rangle \text{ (`...'} | \langle matching \rangle \text{ )'} =>' \langle case\text{-}body \rangle
\langle matching \rangle ::= \langle literal \rangle \mid \langle identifier \rangle \mid \langle pre-func \rangle \langle matching \rangle \mid \langle tuple-matching \rangle \mid \langle list-matching \rangle
\langle tuple\text{-}matching \rangle ::= '(' \langle matching \rangle (',' \langle matching \rangle )+ ')'
\langle list-matching \rangle ::= `[`[\langle matching \rangle (`,`\langle matching \rangle)^*[`, ...']]`]`]
\langle case-body \rangle ::= (``` | `` n` \langle indent \rangle) (\langle simple-func-body \rangle | \langle big-op-expr \rangle) [\langle where-expr \rangle]
```

```
 \langle value\text{-}def \rangle ::= \langle indent \rangle \ \langle identifier \rangle \ \text{``n'} \ \langle indent \rangle \ \text{`:} \ \langle \ ' \ ' \ \rangle \ \langle indent \rangle \ \text{`=} \ | \ \langle value\text{-}expr \rangle \ [ \ \langle where\text{-}expr \rangle \ ]   \langle value\text{-}expr \rangle ::= \langle no\text{-}paren\text{-}op\text{-}arg \rangle \ | \ \langle op\text{-}expr \rangle \ | \ \langle func\text{-}expr \rangle \ | \ \langle big\text{-}tuple \rangle \ | \ \langle big\text{-}list \rangle   \langle grouped\text{-}value\text{-}defs \rangle ::= \langle indent \rangle \ \langle identifier \rangle \ ) + \langle indent \rangle \ \langle identifier \rangle \ ) + \langle indent \rangle \ \langle identifier \rangle \ ) + \langle indent \rangle \ \langle identifier \rangle \ \rangle + \langle indent \rangle \ \langle identifier \rangle \ \rangle + \langle indent \rangle \ \langle identifier \rangle \ \rangle + \langle identifier \rangle \ \langle identifier \rangle \ \rangle + \langle identifier \rangle \ \langle identifier \rangle \ \langle identifier \rangle \ \rangle + \langle identifier \rangle \ \langle identifier \rangle \ \langle identifier \rangle \ \rangle + \langle identifier \rangle \ \rangle + \langle identifier \rangle \ \langle identifie
```

# 4 Language Description: Types and Type Logic

## 4.1 Types

The constructs regarding types are **type expressions** and **type definitions** and they are described in the following sections.

#### 4.1.1 Type Expressions

Type expressions are divided into the following categories:

- Type Identifiers
- Type Variables
- Function Types
- Product Types
- Type Application Types
- Conditional Types

which are described in the following paragraphs.

The grammar of a type expression is:

```
\langle type \rangle ::= [\langle condition \rangle] \langle simple-type \rangle
\langle simple-type \rangle ::= \langle type-id \rangle | \langle type-var \rangle | \langle func-type \rangle | \langle prod-type \rangle | \langle type-app \rangle
```

## Type Identifiers

• Examples

Int
Real
Char
String
FunnyType
MyDefinedType

#### ullet Description

A type identifier is either the name of a basic type (Int, Real, Char, String) or the name of some defined type that has no type parameters. It begins with a capital letter and is followed by one or more capital or lowercase letters.

• Grammar

```
\langle type-id \rangle ::= [A-Z] [A-Za-z] +
```

## Type Variables

• Examples

A B C X Y

Z

• Examples of type variables inside bigger type expressions

```
A \Rightarrow A
(A \Rightarrow B, B \Rightarrow C) \Rightarrow (A \Rightarrow C)
((A, A) \Rightarrow A, A, ListOf(A)s) \Rightarrow A
```

• Description

Type Variables are used inside larger type expressions of polymorphic types. A polymorphic type is a type where any function of that type can be used as a function of any type that corresponds to substituting every type variable of the polymorphic type with a particular type. The easiest example of a polymorphic type is the type of the identity function where we have:

```
id
  : A => A
  = x => x

id(1)
  : Int
  where A is substituted by Int and id gets the type Int => Int

id("Hello")
  : String
  where A is substituted by String and id gets the type String => String
```

A type variable is a single capital letter.

 $\bullet$  Grammar

```
\langle type\text{-}var \rangle ::= [A-Z]
```

## **Function Types**

• Examples

```
String => String
Real => Int
A => A
Int x Int => Int
(Real, Real, Real) => Real
(A => B, B => C) => (A => C)
(Int => Int) => Int
```

## • Description

A function type expression is comprised of the expressions of the types of the parameters and the expression of the type of the result, seperated by the arrow "=>". If there are more than one parameters, the expressions of their types are inside parentheses and seperated by commas. If there is only one parameter, the expression of its type is put in parentheses only if it is a function type. The same applies to the type of the result.

• Grammar

```
\langle func\text{-}type \rangle ::= \langle param\text{-}types\text{-}expr \rangle \text{ `}_{\square} \Rightarrow_{\square} \text{'} \langle one\text{-}type \rangle
\langle param\text{-}types\text{-}expr \rangle ::= \langle one\text{-}type \rangle | \text{'}(\text{'} \langle simple\text{-}type \rangle (\text{'}, \text{'} \langle simple\text{-}type \rangle )+ \text{'}) \text{'}
\langle one\text{-}type \rangle ::= \langle type\text{-}id \rangle | \langle type\text{-}var \rangle | \langle prod\text{-}type \rangle | \langle type\text{-}app \rangle | \text{'}(\text{'} \langle func\text{-}type \rangle \text{'}) \text{'}
```

## **Product Types**

• Examples

```
Int x Int
Real x Real x Real
Int x Real x String
ListOf(Int)s x (Int x ListOf(String)s)
(Int => Int) x (Int x Real) x (Real => String)
```

• Description

Product types are the types of tuples. They are comprised of the expressions of the types of the fields seperated by the string "x" (space 'x' space) because 'x' is very similar the symbol used in the cartesian product. If any of the fields has a product or a function type then the corresponding type expression must be inside parentheses.

• Grammar

```
\langle prod\text{-}type \rangle ::= \langle field\text{-}type \rangle \ (\text{`} \sqcup \mathbf{x} \sqcup \text{`} \langle field\text{-}type \rangle \ ) +
\langle field\text{-}type \rangle ::= \langle type\text{-}id \rangle \ | \ \langle type\text{-}var \rangle \ | \ \langle type\text{-}app \rangle \ | \ \text{`} (\text{`} (\ \langle func\text{-}type \rangle \ | \ \langle prod\text{-}type \rangle \ ) \ \text{`}) \text{`}
```

## Type Application Types

• Examples

```
Possibly(Int)
ListOf(Real)s
TreeOf(String)s
Error(String)OrResult(Int)
ListOf(Int => Int)s
ListOf(A)s
```

Type application types are types that are produced by passing arguments to a type function generated by a tuple\_type or an or\_type definition. For example, given the definition of ListOf(A)s:

```
or_type ListOf(A)s
values non_empty_1:NonEmptyListOf(A)s | empty_1
```

We have that ListOf()s is a type function that receives one type parameter and returns a resulting type. For example ListOf(Int)s is the result of passing the type argument Int to ListOf()s.

Type application types have the same form as the name in the tuple\_type or or\_type definition, with the difference that type parameters are substituted by the type expressions of the arguments.

#### • Grammar

## Conditional Types

 $\bullet$  Examples

```
(A) And(B) Can_Be_Equal ==> (A, B) => Bool
(A) And(B) Add_To(C) ==> (A, B) => C
(A) Is(B)s_First ==> B => A
(T) Has_String_Representation ==> T => String
(E) Has_Use ==> (E(A), A => E(B)) => E(B)
```

#### • Description

Conditional types are the types of values that are polymorphic not because of their structure but because they have been defined (seperately) for many different combinations of types (i.e. they are ad hoc polymorphic). They are comprised of a condition and a "simple" type (i.e. a type without a condition) which are seperated by the arrow "==>". The condition is a type proposition which refers to type variables inside the "simple" type and it must hold whenever the polymorphic value of that type is used. For example:

```
first
  : (A)Is(B)s_First => B => A

can be used as follows:

pair, triple, list
  : Int x String, Real x Char x Int, ListOf(String)s
  = (42, "The answer to everything"), (3.14, 'a', 1), ["Hi!", "Hello", Heeey"]

>> pair -> first
  : Int
  = 42

>> triple -> first
  : Real
```

```
= 3.14
>> list -> first
: String
= "Hi!"
```

and that is because the following propositions hold:

```
(Int)Is(Int x String)s_First
(Real)Is(Real x Char x Int)s_First
(String)Is(ListOf(String)s)s_First
```

which it turn means that the function first has been defined for these combinations of types. For more on how conditions, propositions and ad hoc polymorphism works, see the "Type Logic" section (4.2).

• Grammar

As described in the beginning of this section, the grammar of a type is:

```
\langle type \rangle ::= [\langle condition \rangle] \langle simple-type \rangle
```

And therefore here only the grammar of the condition must be written:

```
\langle condition \rangle ::= \langle prop\text{-}name \rangle ` \sqcup == > \sqcup `
```

#### 4.1.2 Type Definitions

Type definitions are divided into tuple\_type definitions and or\_type definitions which are described in the following paragraphs.

The grammar of a type definition is:

```
\langle type\text{-}def \rangle ::= \langle tuple\text{-}type\text{-}def \rangle \mid \langle or\text{-}type\text{-}def \rangle
```

## **Tuple Types**

• Definition Examples

```
tuple_type Name
value (first_name, last_name) : String x String

tuple_type Date
value (day, month, year) : Int x Int x Int

tuple_type MathematicianInfo
value (name, nationality, date_of_birth) : Name x String x Date

tuple_type TreeOf(A)s
values (root, subtrees) : A x ListOf(TreeOf(A)s)s

tuple_type Indexed(T)
value (index, val) : Int x T
```

• Usage Examples

A tuple type is equivalent to a product type with a new name and names for the fields for convinience. A tuple type generates postfix functions for all of the fields by using a '.' before the name of the field. For example the MathematicianInfo type above generates the following functions:

.name

```
: MathematicianInfo => Name
.nationality
  : MathematicianInfo => String
.date_of_birth
  : MathematicianInfo => Date
```

These functions are named "postfix functions" because they can be appended to their argument.

• Grammar

```
 \begin{split} &\langle tuple\text{-}type\text{-}def \rangle ::= \text{`tuple\_type\_'} \, \langle type\text{-}name \rangle \text{``nvalue\_'} \text{``('} \, \langle identifier \rangle \text{(`,\_'} \, \langle identifier \rangle) + `) \, ``_{\square} :_{\square} \, \langle prod\text{-}type \rangle \\ &\langle type\text{-}name \rangle ::= \left[ \, \langle params\text{-}in\text{-}paren \rangle \, \right] \, \left( \, \langle type\text{-}id \rangle \, | \, \langle type\text{-}id\text{-}with\text{-}params \rangle \, \right) \, \left[ \, \langle params\text{-}in\text{-}paren \rangle \, \right] \\ &\langle type\text{-}id\text{-}with\text{-}params \rangle ::= \, \langle type\text{-}id \rangle \, \left( \, \langle params\text{-}in\text{-}paren \rangle \, \left[ \text{A-Za-z} \right] + \, \right) + \\ &\langle params\text{-}in\text{-}paren \rangle \, ::= \, `(' \, \langle type\text{-}var \rangle \, \left( \, `, \, ` \, \langle type\text{-}var \rangle \, \right) ^* \, `) \, `\end{split}
```

#### Or Types

• Definition Examples

```
or_type Bool
values true | false

or_type Possibly(A)
values the_value:A | no_value

// needed tuple_type for ListOf(A)s
tuple_type NonEmptyListOf(A)s
value (head, tail) : A x ListOf(A)s

or_type ListOf(A)s
```

```
values non_empty_1:NonEmptyListOf(A)s | empty_1
 or_type Error(A)OrResult(B)
 values error:A | result:B
• Usage Examples
  is_empty
   : ListOf(A)s => Bool
   = cases =>
      empty_1 => true
     non_empty_l:anything => false
 get_head
   : ListOf(A)s => Possibly(A)
   = cases =>
      empty_l => no_value
     non_empty_l:list => the_value:list.head
 sum list
   : ListOf(Int)s => Int
   = cases
      empty_1 => 0
     non_empty_1:1 => 1.head + sum_list(1.tail)
 print_err_or_res
   : Error(A)OrResult(B) => (EmptyVal)IOAction
   = cases =>
      error:e => print("Error occured: " + e -> to_string)
      result:r => print("All good! The result is: " + r -> to_string)
```

The values of an or\_type are split into cases. Some cases have other values inside. The cases which have other values inside are followed by a semicolon and the type of the internal value. The same syntax can be used for matching that particular case in a function using the "cases" syntax. An or\_type definition automatically creates prefix functions for each case with an internal value (which are simply conversions from the type of the internal value to the or\_type). For example, for the case "non\_empty\_1" of a list, the function "non\_empty\_1:" is automatically created from the definition for which we can say:

```
non_empty_1:
    : NonEmptyListOf(A)s => ListOf(A)s
Similarly:
the_value:
    : A => Possibly(A)
```

These functions are called "prefix functions" because they are prepended to their argument. For example:

```
non_e_l
   : NonEmptyListOf(Int)s
   = (1, [2, 3, 4])
>> non_empty_l:non_e_l
   : ListOf(Int)s
   = [1, 2, 3, 4]
```

These functions can be used like any other function as arguments to other functions. For example:

```
non_empty_ls_to_ls
: ListOf(NonEmptyListOf(A)s)s => ListOf(ListOf(A)s)s
= apply(non_empty_l:)to_each
```

• Grammar

```
\begin{split} &\langle \textit{or-type-def} \rangle ::= \\ & \quad \text{`or\_type$$$_{\square}$' $$ $\langle \textit{type-name} \rangle$} \\ & \quad \text{``nvalues$$$_{\square}$' $$ $\langle \textit{identifier} \rangle$ [ `:' $$ $\langle \textit{simple-type} \rangle$ ] ( `_{\square}|_{\square}$' $$ $\langle \textit{identifier} \rangle$ [ `:' $$\langle \textit{simple-type} \rangle$ ] )* \end{split}
```

## 4.2 Type Logic

Type logic is the mechanism for ad hoc polymorphism in leases. The central notion of **type logic** is the **type proposition**. A type proposition is a proposition that has types as parameters and is true or false for particular type arguments.

Type propositions can either be defined or proven (for certain type arguments). Therefore, the following constructs exist and accomplish the aforementioned respectively: **type proposition definitions** and **type theorems**. These constructs are described in detail in the following sections. From this point onwards the "type" part will be omitted, i.e. propositions are always type propositions and theorems are always type theorems.

## 4.2.1 Proposition Definitions

Proposition definitions are split into definitions of **atomic propositions** and definitions of **renaming propositions** which are described in the following paragraphs.

## **Atomic Propositions**

• Examples

```
type_proposition (A)Is(B)s_First
value
  first : B => A

type_proposition (T)Has_String_Representation
value
  to_string : T => String

type_proposition (T)Has_A_Wrapper
value
  wrapper : A => T(A)

type_proposition (T)Has_Internal_App
value
  apply()internally : (A => B, T(A)) => T(B)
```

The examples above define the following (ad hoc) polymorphic functions which have the respective (conditional) types:

```
first
   : (A)Is(B)s_First ==> B => A
to_string
```

```
: (T)Has_String_Represention ==> T => String
wrapper
: (T)Has_A_Wrapper ==> A => T(A)
apply()internally
: (T)Has_Internal_App ==> (A => B, T(A)) => T(B)
```

An atomic proposition definition defines simultaneously the **atomic proposition** itself and a **polymorphic value** (usually, but not necessarily, a function), by definining the form of the type of the value given the type parameters of the proposition. The proposition is true or false when the type parameters are substituted by specific type arguments depending on whether the implementation of the value has been defined for these type arguments. The aforementioned truthvalue determines whether the value is used correctly inside the program and therefore whether the program will typecheck. In order to add more types for which the function works, i.e. define the function for these types, i.e. make the proposition true for these types, one must prove a theorem. The specifics of theorems are described in the next section. For now, we'll show the example for everything mentioned in this paragraph for the proposition (A)Is(B)s First:

- Proposition Definition:

```
type_proposition (A)Is(B)s_First
  value
    first : B => A
- Function defined and its type:
  first
    : (A)Is(B)s_First ==> B => A
- Theorems for specific types:
  type_theorem (A)Is(A x B)s_First
  proof
    first = .1st
  type_theorem (A)Is(ListOf(A)s)s_First
  proof
    first = cases =>
      empty_1 => show_err("Tried to take the first element of an empty_1 list")
      non_empty_1:1 => 1.head
- Usage of the function
  pair, list
    : Int x String, ListOf(String)s
    = (42, "The answer to everything"), ["Hi!", "Hello", Heeey"]
  >> pair -> first
    : Int
    = 42
  >> list -> first
    : String
    = "Hi!"
```

An atomic proposition definition begins with the keyword type\_proposition followed by the name of the proposition (including the type parameters) in the first line. The second line is the keyword value. The third line is indented once and has the identifier and the type expression of the value seperated by the string ":".

## **Renaming Propositions**

• Examples

```
type_proposition (T)Has_Equality
equivalent
  (T)And(T)Can_Be_Equal

type_proposition (A)And(B)Are_Comparable
equivalent
  (A)Can_Be_Less_Than(B), (A)And(B)Can_Be_Equal, (A)Can_Be_Greater_Than(B)

type_proposition (T)Has_Comparison
equivalent
  (T)And(T)Are_Comparable
```

• Description

A renaming proposition definition is used to abbreviate one or the conjunction of many propositions (i.e. AND of all of them) into one new proposition.

A renaming proposition definition begins with the keyword type\_proposition followed by the name of the proposition (including the type parameters) in the first line. The second line is the keyword equivalent. The third line is indented once and has either one proposition or (if it is a conjunction) many propositions separated by commas (where the commas essentially mean "and").

## **Grammar for Proposition Definitions**

```
 \langle type\text{-}prop\text{-}def \rangle ::= \langle atom\text{-}prop\text{-}def \rangle \mid \langle renaming\text{-}prop\text{-}def \rangle 
 \langle atom\text{-}prop\text{-}def \rangle ::= \langle prop\text{-}name\text{-}line \rangle \text{ '} \backslash \text{nvalue} \backslash \text{n}_{\square\square} \text{'} \langle identifier \rangle \text{ '}_{\square} :_{\square} \text{'} \langle simple\text{-}type \rangle 
 \langle renaming\text{-}prop\text{-}def \rangle ::= \langle prop\text{-}name\text{-}line \rangle \text{ '} \backslash \text{nequivalent} \backslash \text{n}_{\square\square} \text{'} \langle prop\text{-}name \rangle \text{ (', ____'' \langle prop\text{-}name \rangle)} \text{*} 
 \langle prop\text{-}name\text{-}line \rangle ::= \text{ '} \backslash \text{type\_proposition}_{\square} \text{'} \langle prop\text{-}name \rangle 
 \langle prop\text{-}name \rangle ::= \text{ ('} \backslash \text{name-part} \rangle \langle params\text{-}in\text{-}paren \rangle \text{)} + \text{ ['} \backslash \text{name-part} \rangle \text{]} 
 | (\langle params\text{-}in\text{-}paren \rangle \langle name\text{-}part \rangle ) + \text{ ['} \langle params\text{-}in\text{-}paren \rangle \text{]} 
 \langle name\text{-}part \rangle ::= \text{ ([A-Za-z] | '\_'[A-Z] )} +
```

#### 4.2.2 Theorems

Theorems are split into theorems of **atomic propositions** and theorems of **implication propositions** which are described in the following paragraphs.

#### **Atomic Propositions**

• Examples

```
type_theorem (Possibly())Has_A_Wrapper
proof
  wrapper = the_value:

type_theorem (ListOf()s)Has_A_Wrapper
```

```
proof
    wrapper = x \Rightarrow [x]
  type_theorem (Possibly())Has_Internal_App
  proof
    apply()internally = (f, cases) =>
      no_value => no_value
      the_value:x => the_value:f(x)
  type_theorem (ListOf()s)Has_Internal_App
  proof
    apply()internally = (f, cases) =>
      empty_l => empty_l
      non_empty_l:1 => non_empty_l:(f(l.head), l.tail -> apply(f)internally)
• Usage
  a, b
    : all Possibly(Int)
    = wrapper(1), no_value
  11, 12, 13
    : all ListOf(Int)s
    = wrapper(1), empty_1, [1, 2, 3]
  >> a
    : Possibly(Int)
    = the_value:1
  >> b
    : Possibly(Int)
    = no_value
  >> 11
    : ListOf(Int)s
    = [1]
  >> 12
    : ListOf(Int)s
    = []
  >> a \rightarrow apply(x \Rightarrow x + 1)internally
    : Possibly(Int)
    = the_value:2
  >> b -> apply(x => x + 1)internally
    : Possibly(Int)
    = no_value
  >> 11 -> apply(x => x + 1)internally
    : ListOf(Int)s
    = [2]
  >> 12 -> apply(x => x + 1)internally
    : ListOf(Int)s
    = []
  >> 13 \rightarrow apply(x \Rightarrow x + 1)internally
    : ListOf(Int)s
    = [2, 3, 4]
```

A theorem of an atomic proposition proves the proposition for specific type arguments, by implementing the value associated to the proposition for these type arguments. Therefore, the value associated with the proposition can be used with all the combinations of type arguments for which the proposition is true, i.e. the combinations of type arguments for which the value has been implemented.

A proof of a theorem of an atomic proposition is correct when the implementation of the value associated with the proposition follows the form of the type given to the value by the definition of the proposition, i.e. the only difference between the type of the value in the theorem and the type of the value in the definition is that the type parameters of the proposition are substituted by the type arguments of the theorem.

A theorem of an atomic proposition begins with the keyword type\_theorem followed by the name of the proposition with the type parameters substituted by the specific types for which the proposition will be proven. The second line is the keyword proof. The third line is indented once and it is the line in which the proof begins. The proof begins with the identifier of the value associated with the proposition and is followed by the string " = " and the value expression which implements the value.

#### Implication Propositions

#### • Examples

```
type_theorem (A)And(B)Can_Be_Equal => (A)And(B)Can_Be_Unequal
proof
    a \= b = not(a = b)

type_theorem (A)Can_Be_Greater_Than(B) => (A)Can_Be_Le_Or_Eq_To(B)
proof
    a <= b = not(a > b)

type_proposition (A)And(B)Have_Eq_And_Gr
equivalent
    (A)And(B)Can_Be_Equal, (A)Can_Be_Greater_Than(B)

type_theorem (A)And(B)Have_Eq_And_Gr => (A)Can_Be_Gr_Or_Eq_To(B)
proof
    a >= b = a = b | a > b
```

#### • Description

A theorem of an implication proposition is very similar to a theorem of an atomic proposition in the sense that it also implements a value in the proof. The difference is that the implementation uses another ad hoc polymorphic value (or many). Therefore, the implementation does not prove the proposition associated to the value it implements, because it assumes that the polymorphic value(s) used in the implementation is(are) already defined. In other words it proves the following: "if this(these) ad hoc polymorphic value(s) is(are) defined then we can also define this other one". This can be translated into the following implication proposition: "if the proposition associated to the value(s) we are using is true then the proposition associated to the value we are defining is true", which can be condensed to the notation with the "=>" arrow used in the examples.

The proof of an implication proposition allows the compiler to automatically create the definition for an ad hoc polymorphic value for a particular combination of types given the definitions of the ad hoc polymorphic values used in the implementation for this same combination of types. This mechanism essentially gives definitions for free, that is in the sense that when you define a set of ad hoc polymorphic values for a particular set of types you get for free all the ad hoc polymorphic values that can be defined using a subset of the defined

ones.

A theorem of an implication proposition is grammatically the same as a theorem of an atomic proposition with the only difference being that an implication proposition is comprised by two atomic propositions separated by the "=>" arrow.

#### Grammar for Theorems

```
 \langle type\text{-}theo\rangle ::= \text{`type\_theorem}_{\square}\text{'}(\langle atomic\text{-}prop\rangle \mid \langle implication\text{-}prop\rangle) \text{ ``nproof}_{\square}\text{'}(\langle identifier\rangle `\square=\square' \langle value\text{-}expr\rangle } \\ \langle atomic\text{-}prop\rangle ::= \langle prop\text{-}name\text{-}sub\rangle \\ \langle implication\text{-}prop\rangle ::= \langle prop\text{-}name\text{-}sub\rangle `\square=>\square' \langle prop\text{-}name\text{-}sub\rangle } \\ \langle prop\text{-}name\text{-}sub\rangle ::= \\ (\langle name\text{-}part\rangle \langle types\text{-}in\text{-}paren\rangle) + [\langle name\text{-}part\rangle] \\ | (\langle types\text{-}in\text{-}paren\rangle \langle name\text{-}part\rangle) + [\langle types\text{-}in\text{-}paren\rangle]
```

## 4.3 Complete Grammar for Types and Type Logic

```
\langle type \rangle ::= [\langle condition \rangle] \langle simple-type \rangle
\langle simple-type \rangle ::= \langle type-id \rangle \mid \langle type-var \rangle \mid \langle func-type \rangle \mid \langle prod-type \rangle \mid \langle type-app \rangle
\langle type - id \rangle ::= [A-Z] [A-Za-z] +
\langle type\text{-}var \rangle ::= [A-Z]
\langle func\text{-}type \rangle ::= \langle param\text{-}types\text{-}expr \rangle ` \Box = > \Box ` \langle one\text{-}type \rangle
\langle param-types-expr \rangle ::= \langle one-type \rangle \mid (\langle simple-type \rangle (\langle simple-type \rangle ) + (\langle simple-typ
\langle one\text{-type} \rangle ::= \langle type\text{-}id \rangle \mid \langle type\text{-}var \rangle \mid \langle prod\text{-}type \rangle \mid \langle type\text{-}app \rangle \mid \text{`('} \langle func\text{-}type \rangle ')'
\langle prod\text{-}type \rangle ::= \langle field\text{-}type \rangle \ (` \sqcup x \sqcup ' \langle field\text{-}type \rangle \ ) +
\langle \text{field-type} \rangle ::= \langle \text{type-id} \rangle \mid \langle \text{type-var} \rangle \mid \langle \text{type-app} \rangle \mid ((\langle \text{func-type} \rangle \mid \langle \text{prod-type} \rangle)))
\langle type-app \rangle ::=
                              [\langle types-in-paren \rangle \mid \langle type-id-with-args \rangle \mid \langle types-in-paren \rangle ]
                               \langle types-in-paren \rangle \langle type-id \rangle [\langle types-in-paren \rangle]
                              \langle type-id \rangle \langle types-in-paren \rangle
\langle type\text{-}id\text{-}with\text{-}args \rangle ::= \langle type\text{-}id \rangle \ (\langle types\text{-}in\text{-}paren \rangle \ [A-Za-z]+) +
\langle types-in-paren \rangle ::= `(` \langle simple-type \rangle (`, ` \langle simple-type \rangle )* `)`
\langle condition \rangle ::= \langle prop\text{-}name \rangle ` ==> '
```

```
\langle type\text{-}def \rangle ::= \langle tuple\text{-}type\text{-}def \rangle \mid \langle or\text{-}type\text{-}def \rangle
\langle tuple\_type\_def \rangle ::= \text{`tuple\_type}_{\square}' \langle type\_name \rangle \text{``nvalue}_{\square}' \text{`('} \langle identifier \rangle \text{(',}_{\square}' \langle identifier \rangle) + ')' `_{\square}:_{\square}' \langle prod\_type \rangle = (1 - 1) \cdot (1 - 1
\langle type-name \rangle ::= \lceil \langle params-in-paren \rangle \rceil (\langle type-id \rangle \mid \langle type-id-with-params \rangle) \lceil \langle params-in-paren \rangle \rceil
\langle type-id-with-params \rangle ::= \langle type-id \rangle (\langle params-in-paren \rangle [A-Za-z]+)+
\langle params-in-paren \rangle ::= (\langle type-var \rangle (\langle type-var \rangle)^*)
\langle or\text{-}type\text{-}def \rangle ::=
                 "or\_type_{\sqcup}" \langle type-name \rangle "\nvalues_{\sqcup}" \langle identifier \rangle [":" \langle simple-type \rangle ] ("_{\sqcup}" \langle identifier \rangle [":" \langle simple-type \rangle ]) "
\langle type\text{-}prop\text{-}def \rangle ::= \langle atom\text{-}prop\text{-}def \rangle \mid \langle renaming\text{-}prop\text{-}def \rangle
\langle atom\text{-}prop\text{-}def \rangle ::= \langle prop\text{-}name\text{-}line \rangle \text{ ``nvalue`n}_{\sqcup \sqcup} \text{'} \langle identifier \rangle \text{ `}_{\sqcup} \text{:}_{\sqcup} \text{'} \langle simple\text{-}type \rangle
\langle renaming-prop-def \rangle ::= \langle prop-name-line \rangle \text{ '\nequivalent\n}_{\sqcup\sqcup}' \langle prop-name \rangle \text{ (',}_{\sqcup}' \langle prop-name \rangle)^*
\langle prop\text{-}name\text{-}line \rangle ::= \text{'type\_proposition}_{\sqcup}, \langle prop\text{-}name \rangle
\langle prop\text{-}name \rangle ::=
                  (\langle name-part \rangle \langle params-in-paren \rangle) + [\langle name-part \rangle]
                (\langle params-in-paren \rangle \langle name-part \rangle) + [\langle params-in-paren \rangle]
\langle name-part \rangle ::= ( [A-Za-z] | `\_'[A-Z] ) +
\langle type\text{-}theo \rangle ::= \text{`type\_theorem_{\square}'} (\langle atomic\text{-}prop \rangle | \langle implication\text{-}prop \rangle) \text{`} \text{`}nproof\text{`}n_{\square\square}' \langle identifier \rangle `} :_{\square=\square}' \langle value\text{-}expr \rangle
\langle atomic\text{-}prop \rangle ::= \langle prop\text{-}name\text{-}sub \rangle
\langle implication\text{-}prop \rangle ::= \langle prop\text{-}name\text{-}sub \rangle ` \sqcup = > \sqcup ` \langle prop\text{-}name\text{-}sub \rangle
\langle prop\text{-}name\text{-}sub \rangle ::=
                (\langle name-part \rangle \langle types-in-paren \rangle) + [\langle name-part \rangle]
                (\langle types-in-paren \rangle \langle name-part \rangle) + [\langle types-in-paren \rangle]
```

# 5 Language Description: Predefined

## 5.1 Values

Constants

undefined, pi

- 5.2 Types
- 5.3 Type Propositions
- 6 Parser implementation

The parser was implemented using the parsec library.

- 6.1 AST Types
- 6.2 Parsers
- 7 Translation to Haskell
- 8 Running Examples
- 9 Conclusion