Lambda Cases (lcases)

Dimitris Saridakis

Contents

1 Introduction

Haskell is a delightful language. Yet, for some reason, it doesn't seem to have its rightful place in terms of popularity in industry. Why is it so? Is it inherently hard to learn and therefore only the brave dare to use it, or could it be that the syntax is perplexing to the amateur eye? It is my belief that with some syntax changes that give a greater familiarity to the new user, there would be no language more compelling than (the new) Haskell. In an attempt to achieve that familiarity, I present some (hopefully useful) new syntax, of which some is closer to the imperative/OOP style (to attract more already experienced programmers from these languages), some is closer to mathematics (in which most programmers should be experienced) and some is closer to natural language (in which we are all already experienced).

2 Language Description: General

2.1 Program Structure

An lcases program consists of a set of definitions and theorems. Definitions are split into value definitions, type definitions and type proposition definitions. Theorems are proven type propositions. Functions as well as "Environment Actions" (see section 3.2.3) are also considered values. The definition of the "main" value determines the program's behaviour.

Program example: Extended Euclidean Alogirthm

```
// type definitions

tuple_type GcdAndCoeffs
value (gcd, a, b) : Int x Int x Int

tuple_type Coeffs
value (previous, current) : Int x Int

// algorithm

ext_euc
  : (Int, Int) => GcdAndCoeffs
  = ext_euc_rec(init_a_coeffs, init_b_coeffs)
    where
    init_a_coeffs, init_b_coeffs
      : all Coeffs
      = (1, 0), (0, 1)
    ext_euc_rec
      : (Coeffs, Coeffs, Int, Int) => GcdAndCoeffs
      = (a_coeffs, b_coeffs, x, cases) =>
        0 => (x, a_coeffs.previous, b_coeffs.previous)
        y => ext_euc_rec(next <- a_coeffs, next <- b_coeffs, y, x -> mod <- y)
          where
          next
            : Coeffs => Coeffs
            = cs => (cs.current, cs.previous - x -> div <- y * cs.current)

// reading, printing and main

read_two_ints
  : (Int x Int)IOAction
  = print <- "Please give me 2 ints";
```

```
    get_line ;> split_words o> apply(from_string)to_all o> cases =>
      [x, y] => (x, y) -> io_action
      ... => io_error <- "You didn't give me 2 ints"

print_gcd_equation
  : (Int, Int, GcdAndCoeffs) => (EmptyVal)IOAction
  = (x, y, gcd_cs) => print("gcd = " + gcd_cs.gcd + " = " + linear_comb_string)
    where
    linear_comb_string
      : String
      = gcd_cs.a + " * " + x + " + " + gcd_cs.b + " * " + y

main
  : (EmptyVal)IOAction
  = read_two_ints ;> ints =>
    print_gcd_equation(ints.1st, ints.2nd, ext_euc(ints.1st, ints.2nd))
```

Program grammar

$\langle$*program*$\rangle$ ::= ($\langle$*value-def*$\rangle$ | $\langle$*type-def*$\rangle$ | $\langle$*type-prop-def*$\rangle$ | $\langle$*type-theo*$\rangle$)+

2.2 Keywords

The lcases keywords are the following:

```
cases where all tuple_type value or_type values
type_proposition equivalent type_theorem proof
```

Each keyword's functionality is described in the respective section shown in the table below:

Keyword	Section
`cases`	3.3 Function Expressions
`where all`	3.4 Value Definitions
`tuple_type value or_type values`	4.1.2 Type Definitions
`type_proposition value equivalent type_theorem proof`	4.2 Type Logic

The "`cases`" and "`where`" keywords are also reserved words. Therefore, even though they can be generated by the "identifiers" grammar, they cannot be used as identifiers (see "Literals and Identifiers" section 3.1.1).

3 Language Description: Values

3.1 Basic Expressions

3.1.1 Literals and Identifiers

Literals

- *Examples*

```
1  2  17  42  -100
1.61  2.71  3.14  -1234.567
'a'  'b'  'c'  'x'  'y'  'z'  '.'  ','  '\n'
"Hello World!"  "What's up, doc?"  "Alrighty then!"
```

- *Description*

  There are literals for the four basic types: Int, Real, Char, String. These are the usual integers, real numbes, characters and strings. The exact specification of literals is the same as in the Haskell report.

- *Grammar*

  ⟨*literal*⟩ ::= ⟨*literal*⟩

  TODO add the grammar from the haskell report

Identifiers

- *Examples*

```
x y z
a1 a2 a3
funny_identifier
unnecessarily_long_identifier
apply()to_all
```

- *Description*

  An identifier is the name of a value or a parameter. It is used in the definition of a value and in expressions that use that value, or in the parameters of a function and in the body of that function.

  An identifier starts with a lower case letter and is followed by lower case letters or underscores. It is also possible to have pairs of parentheses in the middle of an identifier (see "Parenthesis Function Application" section 3.1.3 for why this can be useful). Finally, an identifier can be ended with a digit.

- *Grammar*

  ⟨*identifier*⟩ ::= [a-z] [a-z_]* ('`()`' [a-z_]+)* [[0-9]]

  Even though the "`cases`" and "`where`" keywords can be generated by this grammar, they cannot be used as identifiers.

3.1.2 Parenthesis, Tuples and Lists

Parenthesis

- *Examples*

```
(1 + 2)
(((1 + 2) * 3) ^ 4)
(val -> (x => f(x) + 1) -> to_string -> (s => "f(val) + 1 is: " + s))
(do(3)times <- (get_line ;> line => print("Line is: " + line)))
```

- *Description*

  An expression is put in parenthesis to prioritize it or isolate it in a bigger (operator) expression. The expressions inside parenthesis are operator or function expressions.

- *Grammar*

  $\langle$*paren-expr*$\rangle$ ::= '(' $\langle$*op-or-func-expr*$\rangle$ ')'

  $\langle$*op-or-func-expr*$\rangle$ ::= $\langle$*simple-op-expr*$\rangle$ | $\langle$*op-expr-func-end*$\rangle$ | $\langle$*simple-func-expr*$\rangle$

Tuples

- *Examples*

```
(1, "What's up, doc?")
(2, "Alrighty then!", 3.14)
(x, y, z, w)
(1, my_function, (x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1/2))
```

- *Description*

  Tuples are used to group many values (of possibly different types) into one. The type of a tuple can be either the product of the types of the fields or a defined tuple_type which is equivalent to the afformentioned product type (see "Tuple Types" in section 4.1.2). For example, the type of the second tuple above could be:

```
Int x String x Real
```

  or:

```
MyType
```

  assuming "MyType" has been defined in a similar way to the following:

```
tuple_type MyType
value (my_int, my_string, my_real) : Int x String x Real
```

- *Big Tuples*

  Example

```
my_big_tuple
  : String x Int x Real x String x String x (String x Real x Real)
  = ( "Hey, I'm the first field and I'm also a relatively big string."
    , 42, 3.14, "Hey, I'm the first small string", "Hey, I'm the second small string"
    , ("Hey, I'm a string inside the nested tuple", 2.71, 1.61)
    )
```

Description

It is possible to stretch a (big) tuple expression over multiple lines (only) in a seperate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '(' is after the "= " part of the value definition and the first field must be in the same line.
- The tuple can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '(' character was in the first line.
- The tuple must be ended by a line that only contains the ')' character and is also indented so that the ')' is in same column where the '(' character was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

  ⟨*tuple*⟩ ::= '(' ⟨*line-expr*⟩ ',␣' ⟨*comma-sep-line-exprs*⟩ ')'

  ⟨*comma-sep-line-exprs*⟩ ::= ⟨*line-expr*⟩ (',␣' ⟨*line-expr*⟩)*

  ⟨*line-expr*⟩ ::= ⟨*no-paren-op-arg*⟩ | ⟨*op-or-func-expr*⟩

  ⟨*big-tuple*⟩ ::=
  '(' ⟨*line-expr*⟩ ['\n' ⟨*indent*⟩] ',␣' ⟨*comma-sep-line-exprs*⟩
  ('\n' ⟨*indent*⟩ ',' ⟨*comma-sep-line-exprs*⟩)*
  '\n' ⟨*indent*⟩ ')'

Lists

- *Examples*

```
[1, 2, 17, 42, -100]
[1.61, 2.71, 3.14, -1234.567]
["Hello World!", "What's up, doc?", "Alrighty then!"]
[x => x + 1, x => x + 2, x => x + 3]
[x, y, z, w]
```

- *Description*

  Lists are used to group many values of the same type into one. The type of the list is ListOf(A)s where A is the type of every value inside. Therefore, the types of the first four examples are:

```
ListOf(Int)s
ListOf(Real)s
ListOf(String)s
(A)And(Int)Add_To(B) ==> ListOf(A => B)s
```

  And the last list is only legal if x, y, z and w all have the same type. Assuming they do and it's the type T, the type of the list is:

```
ListOf(T)s
```

- *Big Lists*

  Example

```
my_big_list
  : ListOf(Int => (EmptyVal)IOAction)s
  = [ x => print("I'm the first function and x + 1 is: " + (x + 1))
    , x => print("I'm the second function and x + 2 is: " + (x + 2))
    , x => print("I'm the third function and x + 3 is: " + (x + 3))
    ]
```

Description

It is possible to stretch a (big) list expression over multiple lines (only) in a seperate value definition (see "Value Definitions" section 3.4.1). In that case:

- The character '[' is after the "= " part of the value definition and the first element must be in the same line.
- The list can split in a new line only at a ',' character. Every such line must be indented so that the ',' is in same column where the '[' character was in the first line.
- The tuple must be ended by a line that only contains the ']' character and is also indented so that the ']' is in same column where the '[' character was in the first line.
- The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

  ⟨*list*⟩ ::= '`[`' [⟨*comma-sep-line-exprs*⟩] '`]`'

  ⟨*big-list*⟩ ::= '`[`' ⟨*comma-sep-line-exprs*⟩ ('`\n`' ⟨*indent*⟩ '`,`' ⟨*comma-sep-line-exprs*⟩)* '`\n`' ⟨*indent*⟩ '`]`'

3.1.3 Parenthesis Function Application

- *Examples*

```
f(x)
f(x, y, z)
(x)to_string
apply(f)to_all
apply(f)to_all(l)
```

- *Description*

  Function application in lcases can be done in many different ways in an attempt to maximize readability. In this section, we discuss the ways function application can be done with parenthesis.

  In the first two examples, we have the usual mathematical function application which is also used in most programming languages and should be familiar to the reader, i.e. function application is done with the arguments of the function in parenthesis seperated by commas and **appended** to the function identifier.

  We extend this idea by allowing the arguments to be **prepended** to the function identifier (third example). Finally, it is also possible to to have the arguments **inside** the function identifier provided the function has been **defined with parentheses inside the identifier**. For example, below is the definition of "apply()to_all":

```
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
  = (f, cases) =>
    empty_l => empty_l
    non_empty_l:l => non_empty_l:(f <- l.head, apply(f)to_all <- l.tail)
```

The actual definition doesn't matter at this point, what matters is that the identifier is "apply()to_all" with the parentheses **included**. This is very useful for defining functions where the argument in the middle makes the function application look and sound more like natural language.

It is possible to have many parentheses pairs in a single function application (last example). The arguments are always inserted to the function from **left to right**. Therefore, when multiple parentheses pairs are present the arguments of the leftmost parentheses are inserted first then the next ones to the right and so on.

- *Grammar*

⟨*paren-func-app*⟩ ::=
 [⟨*arguments*⟩] ⟨*identifier-with-arguments*⟩ [⟨*arguments*⟩]
 | ⟨*arguments*⟩ ⟨*identifier*⟩ [⟨*arguments*⟩]
 | ⟨*identifier*⟩ ⟨*arguments*⟩

⟨*arguments*⟩ ::= '`(`' ⟨*comma-sep-line-exprs*⟩ '`)`'

⟨*identifier-with-arguments*⟩ ::=
 [a-z] [a-z_]* ('`()`'[a-z_]+)* ⟨*arguments*⟩ [a-z_]+ (('`()`' | ⟨*arguments*⟩) [a-z_]+)* [[0-9]]

3.1.4 Prefix and Postfix Functions

Prefix Functions

- *Examples*

```
the_value:1
non_empty_l:l
error:e
result:r
apply(the_value:)to_all
```

- *Description*

Prefix functions are automatically generated from `or_type` definitions (see "Or Types" in section 4.1.2). They are functions that convert a value of a particular type to a value that is a case of an `or_type` and has values of this type inside. For example in the first example above we have:

```
1
  : Int
the_value:1
  : Possibly(Int)
```

Where the function `thevalue:` is automatically generated from the definition of the `Possibly` type:

```
or_type Possibly(A)
values the_value:A | no_value
```

And it has the type `A => Possibly(A)`.

These functions are called prefix functions because they are prepended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function `the_value:` is an argument of the function `apply()to_all`. Prefix functions always end with a colon.

- *Grammar*

⟨*pre-func*⟩ ::= ⟨*identifier*⟩ '`:`'

⟨*pre-func-app*⟩ ::= ⟨*pre-func*⟩ (⟨*basic-expr*⟩ | ⟨*paren-expr*⟩ | ⟨*pre-func-app*⟩)

⟨*basic-expr*⟩ ::= ⟨*literal*⟩ | ⟨*identifier*⟩ | ⟨*tuple*⟩ | ⟨*list*⟩ | ⟨*paren-func-app*⟩ | ⟨*post-func-app*⟩

Postfix Functions

- *Examples*

```
name.first_name
list.head
date.year
tuple.1st
apply(.1st)to_all
```

- *Description*

  Postfix functions are automatically generated from `tuple_type` definitions (see "Tuple Types" in section 4.1.2). They are functions that take a `tuple_type` value and return a particular field (i.e. projection functions). For example in the first example above we have:

```
name
  : Name
name.first_name
  : String
```

  Where the function `.first_name` is automatically generated from the definition of the `Name` type:

```
tuple_type Name
value (first_name, last_name) : String x String
```

  And it has the type `Name => String`.

  There are also the following special projection functions that work on **all product and tuple types**: `.1st .2nd .3rd .4th .5th`. For the 4th example above, assuming:

```
tuple
  : Int x String
```

  We have:

```
tuple.1st
  : Int
```

  The general types of these functions are:

```
.1st
  : (A)Is(B)s_1st ==> B => A
.2nd
  : (A)Is(B)s_2nd ==> B => A
...
```

  These functions are called postfix functions because they are appended to their argument. However, they can also be used as any other function. An illustration of the aforementioned is the last example, where the function `.1st` is an argument of the function `apply()to_all`. Postfix functions always begin with a dot.

- *Grammar*

  ⟨*post-func*⟩ ::= '`.`' ⟨*identifier*⟩ | '`.1st`' | '`.2nd`' | '`.3rd`' | '`.4th`' | '`.5th`'

  ⟨*post-func-app*⟩ ::= (⟨*paren-expr*⟩ | ⟨*basic-expr*⟩) ⟨*post-func*⟩

3.2 Operators

3.2.1 Function Application and Function Composition Operators

Function Application Operators

Operator	Type
`->`	(A, A => B) => B
`<-`	(A => B, A) => B

The function application operators "`->`" and "`<-`" are a different way to apply functions to arguments than the usual parenthesis function application. They are meant to look like arrows that point from the argument to the function. These operators are very useful for chaining many function applications without the clutter of having to open and close parentheses for each one of the functions. For example, assuming we have the following functions with the behaviour suggested by their names and types:

```
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
string_length
  : String => Int
filter_with
  : (A => Bool, ListOf(A)s) => ListOf(A)s
is_odd
  : Int => Bool
sum_ints
  : ListOf(Int)s => Int
```

And a list of strings:

```
strings
  : ListOf(String)s
```

Here is a simple way to get the total number of characters in all the strings that have odd length:

```
chars_in_odd_length_strings
  : Int
  = strings -> apply(string_length)to_all -> filter_with(is_odd) -> sum_ints
```

Ofcourse this can be done equivalently using the other operator:

```
chars_in_odd_length_strings
  : Int
  = sum_ints <- filter_with(is_odd) <- apply(string_length)to_all <- strings
```

These operators can also be used together to put a function between two arguments if that function is commonly used that way in math (or if it looks better for a certain function). For example the "mod" function can be used like so:

```
x -> mod <- y
```

Which is equivalent to:

```
mod(x, y)
```

Function Composition Operators

Operator	Type
`o>`	(A => B, B => C) => (A => C)
`<o`	(B => C, A => B) => (A => C)

The function composition operators ”`o>`” and ”`<o`” are used to compose functions, each one in the corresponding direction. The use of the letter ’o’ is meant to be similar to the mathematical function composition symbol ’$\circ$’ and the symbols ’`>`’, ’`<`’ are used so that the operator points from the function which is applied first to the function which is applied second. A neat example using function composition is the following. Assuming we have the following functions with the behaviour suggested by their names and types:

```
split_words
  : String => ListOf(String)s
apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
reverse_string
  : String => String
merge_words
  : ListOf(String)s => String
```

We can reverse the all the words in a string like so:

```
reverse_words
  : String => String
  = split_words o> apply(reverse_string)to_all o> merge_words
```

Ofcourse this can be done equivalently using the other operator:

```
reverse_words
  : String => String
  = merge_words <o apply(reverse_string)to_all <o split_words
```

3.2.2 Arithmetic, Comparison and Boolean Operators

Arithmetic Operators

Operator	Type
`^`	(A)To_The(B)Has_Type(C) ==> (A, B) => C
`*`	(A)And(B)Multiply_To(C) ==> (A, B) => C
`/`	(A)Divided_By(B)Has_Type(C) ==> (A, B) => C
`+`	(A)And(B)Add_To(C) ==> (A, B) => C
`-`	(A)Minus(B)Has_Type(C) ==> (A, B) => C

The usual arithmetic operators work as they are expected, similarly to mathematics and other programming languages for the usual types. However, they are generalized. The examples below show their generality:

```
>> 1 + 1
 : Int
 = 2
>> 1 + 3.14
 : Real
```

```
 = 4.14
>> 'a' + 'b'
 : String
 = "ab"
>> 'w' + "ord"
 : String
 = "word"
>> "Hello " + "World!"
 : String
 = "Hello World!"
>> 5 * 'a'
 : String
 = "aaaaa"
>> 5 * "hi"
 : String
 = "hihihihihi"
>> "1,2,3" - ','
 : String
 = "123"
```

Let's analyze further the example of addition. The type can be read as such: the '+' operator has the type (A, B) => C, provided that the type proposition (A)And(B)Add_To(C) holds. This proposition being true, means that addition has been defined for these three types (see section "Type Logic" 4.2 for more on type propositions). For example, by the examples above we can deduce that the following propositions are true (in the order of the examples):

```
(Int)And(Int)Add_To(Int)
(Int)And(Real)Add_To(Real)
(Char)And(Char)Add_To(String)
(Char)And(String)Add_To(String)
(Int)And(Char)Multiply_To(String)
(Int)And(String)Multiply_To(String)
(String)Minus(Char)Has_Type(String)
```

This allows us to use the familiar arithmetic operators in types that are not necessarily numbers but it is somewhat intuitively obvious what the should do in those other types. Furthermore, their behaviour can be defined by the user for new user defined types!

Comparison and Boolean Operators

Operator	Type
`=`	(A)Has_Equality ==> (A, A) => Bool
`/=`	(A)Has_Inequality ==> (A, A) => Bool
`>=`	(A)Has_Greater_Or_Equal ==> (A, A) => Bool
`<=`	(A)Has_Less_Or_Equal ==> (A, A) => Bool
`>`	(A)Can_Be_Greater_Than(B) ==> (A, B) => Bool
`<`	(A)Can_Be_Less_Than(B) ==> (A, B) => Bool
`& |`	(Bool, Bool) => Bool

The comparison and boolean operators behave the same as in Haskell and very similarly to most programming languages. The main difference is that in lcases the "equals", "and" and "or" operators have the symbol once

(`= & |`) rather than twice (`== && ||`).

TODO `>` `<` actually work differently ... allow for Real > Int ...

3.2.3 Environment Action Operators

Operator	Type
`;>`	(E)Is_An_Environemnt_Action `==>` (E(A), A `=>` E(B)) `=>` E(B)
`;`	(E)Is_An_Environemnt_Action `==>` (E(A), E(B)) `=>` E(B)

Simple Example

```
print_string("I'll repeat the line.") ; get_line ;> print_string
```

The example above demonstrates the use of the environment action operators with the `IOAction` type, which is how IO is done in lcases. Some light can be shed on how this is done, if we take a look at the types (as always!):

```
print_string
  : String => (EmptyVal)IOAction
print_string("I'll repeat the line.")
  : (EmptyVal)IOAction
get_line
  : (String)IOAction
;
  : (E)Is_An_Environemnt_Action ==> (E(A), E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line
  : (String)IOAction
  where (IOAction)Is_An_Environemnt_Action is true, E = IOAction, A = EmptyVal, B = String
;>
  : (E)Is_An_Environemnt_Action ==> (E(A), A => E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line ;> print_string
  : (EmptyVal)IOAction
  where (IOAction)Is_An_Environemnt_Action is true, E = IOAction, A = String, B = EmptyVal
```

Example program

```
main
  : (EmptyVal)IOAction
  = print_string <- "Hello! What's your name?" ; get_line ;> name =>
    print_string("Oh hi " + name + "! What's your age?") ; get_line ;> age =>
    print_string("Oh that's crazy " + name + "! I didn't expect you to be " + age + "!");
```

In this bigger but similar example the types are:

```
print_string
  : String => (EmptyVal)IOAction
get_line
  : (String)IOAction

print_string <- "Hello! ... "
  : (EmptyVal)IOAction
print_string("Oh hi...)
  : (EmptyVal)IOAction
print_string("Oh that's crazy...)
```

```
  : (EmptyVal)IOAction

;
  : (E)Is_An_Environemnt_Action ==> (E(A), E(B)) => E(B)

print_string("Oh hi...) ; get_line
  : (String)IOAction
  where (IOAction)Is_An_Environemnt_Action is true, E = IOAction, A = EmptyVal, B = String

age => print_string("Oh that's crazy...)
  : String => (EmptyVal)IOAction

;>
  : (E)Is_An_Environemnt_Action ==> (E(A), A => E(B)) => E(B)

print_string("Oh hi...) ; get_line ;> age =>
print_string("Oh that's crazy...)
  : (EmptyVal)IOAction
  where (IOAction)Is_An_Environemnt_Action is true, E = IOAction, A = String, B = EmptyVal

print_string <- "Hello..." ; get_line
  : (String)IOAction

name => print_string("Oh hi ... (till the end)
  : String => (EmptyVal)IOAction

print_string <- "Hello..." ; get_line ;> name =>
print_string("Oh hi ... (till the end)
  : (EmptyVal)IOAction
```

Therefore, "`main : (EmptyVal)IOAction`" checks out. The key here is to remember that function expressions extend to the end of the whole expression. Therefore, we have "`name => ... (till the end)`" and "`age => ... (till the end)`" as the second arguments of the two occurences of the "`;>`" operator.

Description

The environment action operators are used to combine values that do environment actions into values that do more complicated environment actions. Environment actions are also represented by types. More acurately, type functions that take a type as an argument and produce a new type (just like ListOf()s). A value of the type `E(A)` where `(E)Is_An_Environemnt_Action` does an environment action of type `E` that produces a value of type `A`.

The effect of the "`;`" operator described in words is the following: given a value of type `E(A)` and a value of type `E(B)` (which do environment actions that produce values of type `A` and `B` respectively), create a new value the does both actions (provided the first did not result in an error). The overall effect is a value that does an environment action of type `E` (the combination of the "smaller" actions) which produces a value of type `B` (the one produced by the second action) and therefore it is of type `E(B)`.

Note that the value of type `A` produced by the first action is not used anywhere. This happens mostly when `A = EmptyVal` and it is because values of type `E(EmptyVal)` are used for their environment action only (e.g. `print_string(...) : (EmptyVal)IOAction`).

How the two environment actions of the `E(A)` and `E(B)` values are combined to produce the new environment action is specific to the environment action type `E`.

The effect of the ”`;>`” operator described in words is the following: given a value of type `E(A)` (which does an environment action of type `E` that produces a value of type `A`) and a value of type `A => E(B)` (which is a function that takes a value of type `A` and returns an environment action of type `E` that produces a value of type `B`), combine those two values by creating a value that does the following:

- Performs the first action that produces a value of type `A`
- Takes the value of type `A` produced (provided there was no error) and passes it to the function of type `A => E(B)` that then returns an action
- Perfoms the resulting action

The overall effect is an environment action of type `E` that in the end produces a value is of type `B` and therefore the new value is of type `E(B)`.

3.2.4 Operator Expressions

- *Examples*

  ```
  1 + 2
  1 + x * 3 ^ y
  "Hello " + "World!"
  x -> f -> g
  f o> g o> h
  x = y
  x >= y - z & x < 2 * y
  get_line ; get_line ;> line => print("Second line: " + line)
  ```

- *Description*

  Operator expressions are expressions that use operators. Operators act like two-argument-functions that are placed in between their arguments. Therefore, they have function types and they act as it is described in their respective sections above this one.

  An operator expression might have multiple operators. The order of operations is explained in the next section (”Complete Table, Precedence and Associativity”) in Table 2.

  Just like functions, the sub-expressions that act as arguments to an operator, must have types that match the types expected by the operator.

  It is possible to the second argument of an operator is a function expression. This is mostly useful with the ”`;>`” operator (see previous section: ”Environment Operators”), but it is also possible with the following operators: ”`->`”, ”`o>`”, ”`<o`”.

- *Big Operator Expressions*

  It is possible to stretch a (big) operator expression over multiple lines. In that case:

  – The operator expression must split in a new line after an operator (not an argument).
  – Every line after the first must be indented so that in begins at the column where the first line of the operator expression begun.
  – The precise indentation rules are described in the section ”Indentation System” 3.5.1.

- *Grammar*

  ⟨*op-expr*⟩ ::= ⟨*simple-op-expr*⟩ | ⟨*op-expr-func-end*⟩ | ⟨*big-op-expr*⟩ | ⟨*cases-op-expr*⟩

⟨*simple-op-expr*⟩ ::= ⟨*op-arg*⟩ (‘␣’ ⟨*op*⟩ ‘␣’ ⟨*op-arg*⟩)+

⟨*op-expr-func-end*⟩ ::= ⟨*simple-op-expr*⟩ ‘␣’ ⟨*op*⟩ ‘␣’ ⟨*simple-func-expr*⟩

⟨*big-op-expr*⟩ ::=
    ⟨*op-expr-line*⟩ (‘`\n`’ ⟨*indent*⟩ ⟨*op-expr-line*⟩)*
    ‘`\n`’ ⟨*indent*⟩ (⟨*op-arg*⟩ | ⟨*simple-op-expr*⟩ | [⟨*op-expr-line*⟩ ‘␣’] (⟨*simple-func-expr*⟩ | ⟨*big-func-expr*⟩))

⟨*cases-op-expr*⟩ ::= ⟨*op-expr-line*⟩ (‘`\n`’ ⟨*indent*⟩ ⟨*op-expr-line*⟩)* (‘`\n`’ ⟨*indent*⟩ | ‘␣’) ⟨*cases-func-expr*⟩

⟨*op-expr-line*⟩ ::= (⟨*op-arg*⟩ | ⟨*simple-op-expr*⟩) ‘␣’ ⟨*op*⟩

⟨*op-arg*⟩ ::= ⟨*no-paren-op-arg*⟩ | ⟨*paren-expr*⟩

⟨*no-paren-op-arg*⟩ ::= ⟨*basic-expr*⟩ | ⟨*pre-func*⟩ | ⟨*post-func*⟩ | ⟨*pre-func-app*⟩

⟨*op*⟩ ::= ‘->’ | ‘<-’ | ‘o>’ | ‘<o’ | ‘^’ | ‘*’ | ‘/’ | ‘+’ | ‘-’ | ‘=’ | ‘/=’ | ‘>’ | ‘<’ | ‘>=’ | ‘<=’ | ‘&’ | ‘|’ | ‘;>’ | ‘;’

3.2.5 Complete Table, Precedence and Associativity

Table 1: The complete table of lcases operators along with their types and their short descriptions.

Operator	Type	Description
`->`	(A, A => B) => B	Right function application
`<-`	(A => B, A) => B	Left function application
`o>`	(A => B, B => C) => (A => C)	Right function composition
`<o`	(B => C, A => B) => (A => C)	Left function composition
`^`	(A)To_The(B)Has_Type(C) ==> (A, B) => C	General exponentiation
`*`	(A)And(B)Multiply_To(C) ==> (A, B) => C	General multiplication
`/`	(A)Divided_By(B)Has_Type(C) ==> (A, B) => C	General division
`+`	(A)And(B)Add_To(C) ==> (A, B) => C	General addition
`-`	(A)Minus(B)Has_Type(C) ==> (A, B) => C	General subtraction
`=`	(A)Has_Equality ==> (A, A) => Bool	Equality
`/=`	(A)Has_Inequality ==> (A, A) => Bool	Inequality
`>=`	(A)Has_Greater_Or_Equal ==> (A, A) => Bool	Greater than or equal to
`<=`	(A)Has_Less_Or_Equal ==> (A, A) => Bool	Less than or equal to
`>`	(A)Can_Be_Greater_Than(B) ==> (A, B) => Bool	Greater than
`<`	(A)Can_Be_Less_Than(B) ==> (A, B) => Bool	Less than
`&` `|`	(Bool, Bool) => Bool	Boolean operators
`;>`	(E)Is_An_Environemnt_Action ==> (E(A), A => E(B)) => E(B)	Do, unwrap, apply, do
`;`	(E)Is_An_Environemnt_Action ==> (E(A), E(B)) => E(B)	Do then do

The order of operations is done from highest to lowest precedence. In the same level of precedence the order is done from left to right if the associativity is "Left" and from right to left if the associativity is "Right". For the operators that have associativity "None" it is not allowed to place them in the same operator expression. The precedence and assosiativity of the operators is shown in the table below.

Table 2: The table of precedence and associativity of the lcases operators.

Operator	Precedence	Associativity
`->`	10 (highest)	Left
`<-`	9	Right
`o> <o`	8	Left
`^`	7	Right
`* /`	6	Left
`+ -`	5	Left
`= /= > < >= <=`	4	None
`&`	3	Left
`|`	2	Left
`;> ;`	1	Left

3.3 Function Expressions

⟨*func-expr*⟩ ::= ⟨*simple-func-expr*⟩ | ⟨*big-func-expr*⟩ | ⟨*cases-func-expr*⟩

3.3.1 Regular Function Expressions

- *Examples*

```
a => 17 * a + 42

(x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1 / 2)
```

- *Description*

  Regular function expressions are used to define functions or be part of bigger expressions as anonymous functions. They are comprised by their parameters and their body. A parameter has an identifier. The parameters are either only one, in which case there is no parenthesis, or they are many, in which case they are in parenthesis, seperated by commas. The parameters and the body are seperated by an arrow (" `=>` "). The body is an operator expression.

- *Big Function Expressions*

  It is possible to stretch a (big) function expression over multiple lines. In that case:

  – The function expression must split in a new line after the "`=>`" arrow.

  – Every line after the first must be indented so that in begins at the column where the first character of the parameters was in the first line.

  – The precise indentation rules are described in the section "Indentation System" 3.5.1.

- *Grammar*

  ⟨*simple-func-expr*⟩ ::= ⟨*parameters*⟩ '`␣=>␣`' ⟨*simple-func-body*⟩

  ⟨*big-func-expr*⟩ ::= ⟨*parameters*⟩ '`␣=>\n`' ⟨*indent*⟩ (⟨*simple-func-body*⟩ | ⟨*big-op-expr*⟩)

  ⟨*parameters*⟩ ::= ⟨*identifier*⟩ | '`(`' ⟨*identifier*⟩ ('`,␣`' ⟨*identifier*⟩)+ '`)`'

  ⟨*simple-func-body*⟩ ::= ⟨*no-paren-op-arg*⟩ | ⟨*simple-op-expr*⟩ | ⟨*op-expr-func-end*⟩

3.3.2 "cases" Function Expressions

- *Examples*

```
print_sentimental_bool
  : Bool => (EmptyVal)IOAction
  = cases =>
    true => print <- "It's true!! :)"
    false => print <- "It's false... :("

or_type TrafficLight
values green | amber | red

print_sentimental_traffic_light
  : TrafficLight => (EmptyVal)IOAction
  = cases =>
    green => print <- "It's green! Let's go!!! :)"
```

```
    amber => print <- "Go go go, fast!"
    red => print <- "Stop right now! You're going to kill us!!"

is_not_red
  : TrafficLight => Bool
  = cases =>
    green => true
    amber => true
    red => false

is_seventeen_or_forty_two
  : Int => Bool
  = cases =>
    17 => true
    42 => true
    ... => false

traffic_lights_match
  : (TrafficLight, TrafficLight) => Bool
  = (cases, cases) =>
    (green, green) => true
    (amber, amber) => true
    (red, red) => true
    ... => false

gcd
  : (Int, Int) => Int
  = (x, cases) =>
    0 => x
    y => gcd(y, x -> mod <- y)

is_empty
  : ListOf(A)s => Bool
  = cases =>
    empty_l => true
    non_empty_l:anything => false

apply()to_all
  : (A => B, ListOf(A)s) => ListOf(B)s
  = (f, cases) =>
    empty_l => empty_l
    non_empty_l:list => non_empty_l:(f <- list.head, apply(f)to_all <- list.tail)
```

- *Description*

  "cases" is a keyword that works as a special parameter. The difference is that instead of giving the name "cases" to that parameter, it allows the programmer to pattern match on the possible values of that parameter and return a different result for each particular case.

  The last case can be "`... => (body of default case)`" to capture all remaining cases while dismissing the value (e.g. `is_seventeen_or_forty_two` example), or it can be "`some_id => (body of default case)`" to capture all remaining cases while being able to use the value with the name "`some_id`" (e.g. "y" in `gcd` example).

It is possible to use the ”cases” keyword in multiple parameters to match on all of them. By doing that, each case represents a particular combination of values for the parameters involved
(e.g. `traffic_lights_match` example).

It is also possible to use a ”where” expression below a particular case. The ”where” expression must be indented two spaces more than than the line where that particular case begins.

A function expression that uses the ”cases” syntax must contain the ”cases” keyword in at least one parameter. The number of matching expressions in all cases must be the same as the number of parameters with the ”cases” keyword.

- *Grammar*

⟨*cases-func-expr*⟩ ::= ⟨*cases-parameters*⟩ ‘`␣=>`’ ⟨*case*⟩+ ⟨*end-case*⟩

⟨*cases-parameters*⟩ ::= ⟨*cases-parameter*⟩ | ‘`(`’ ⟨*cases-parameter*⟩ (‘`,␣`’ ⟨*cases-parameter*⟩)+ ‘`)`’

⟨*cases-parameter*⟩ ::= ⟨*parameter*⟩ | ‘`cases`’

⟨*case*⟩ ::= ‘`\n`’ ⟨*indent*⟩ ⟨*matching*⟩ ‘`␣=>`’ ⟨*case-body*⟩

⟨*end-case*⟩ ::= ‘`\n`’ ⟨*indent*⟩ (‘`...`’ | ⟨*matching*⟩) ‘`␣=>`’ ⟨*case-body*⟩

⟨*matching*⟩ ::= ⟨*literal*⟩ | ⟨*identifier*⟩ | ⟨*pre-func*⟩ ⟨*matching*⟩ | ⟨*tuple-matching*⟩ | ⟨*list-matching*⟩

⟨*tuple-matching*⟩ ::= ‘`(`’ ⟨*matching*⟩ (‘`,`’ ⟨*matching*⟩)+ ‘`)`’

⟨*list-matching*⟩ ::= ‘`[`’ [⟨*matching*⟩ (‘`,`’ ⟨*matching*⟩)*] ‘`]`’

⟨*case-body*⟩ ::= (⟨*simple-func-body*⟩ | ⟨*big-op-expr*⟩) [⟨*where-expr*⟩]

3.4 Value Definitions and ”where” Expressions

3.4.1 Value Definitions

- *Examples*

```
foo
  : Int
  = 42

f
  : (Int, Int, Int) => Int
  = (a, b, c) => a + b * c

val1, val2, val3
  : Int, Bool, Char
  = 42, true, 'a'

int1, int2, int3
  : all Int
  = 1, 2, 3
```

- *Description*

  Value definitions are the main building block of lcases programs. To define a new value you give it a name, a type and an expression. The name is an identifier in the first line. The second line is indented two spaces more and begins by ": " and continues with the type expression. The third line is indented as the second, begins by "= " and continues with the value expression (which extends to as many lines as needed).

  A value definition is either in the first column, where it can be "seen" by all other value definitions, or it is in a "where" expression (see section below), where it can be "seen" by the expression above the "where" and all the other definitions in the same "where" expression.

  A value definition can be followed by a "where" expression where intermediate values used in the value expression are defined. In that case, the "where" expression must be indented two spaces more than the "=" line of the value definition.

  It is possible to group value definitions together by seperating the names, the types and the expressions with commas. This is very useful for not cluttering the program with many definitions for values with small expressions (e.g. constants). When grouping definitions together it is also possible to use the keyword "all" to give the same type to all the values.

- *Grammar*

  ⟨*value-def*⟩ ::= ⟨*indent*⟩ ⟨*identifier*⟩ '`\n`' ⟨*indent*⟩ '`:␣`' ⟨*type*⟩ '`\n`' ⟨*indent*⟩ '`=␣`' ⟨*value-expr*⟩ [⟨*where-expr*⟩]

  ⟨*value-expr*⟩ ::= ⟨*no-paren-op-arg*⟩ | ⟨*op-expr*⟩ | ⟨*func-expr*⟩ | ⟨*big-tuple*⟩ | ⟨*big-list*⟩

  ⟨*grouped-value-defs*⟩ ::=
  ⟨*indent*⟩ ⟨*identifier*⟩ ('`,␣`' ⟨*identifier*⟩)+
  '`\n`' ⟨*indent*⟩ '`:␣`' (⟨*type*⟩ ('`,␣`' ⟨*type*⟩)+ | '`all`' ⟨*type*⟩)
  '`\n`' ⟨*indent*⟩ '`=␣`' ⟨*comma-sep-line-exprs*⟩ ('`\n`' ⟨*indent*⟩ '`,`' ⟨*comma-sep-line-exprs*⟩)*

3.4.2 "where" Expressions

- *Examples*

```
TODO find good consistent def of Has_Order
sort :
  (A)Has_Order ==> ListOf(A)s => ListOf(A)s
  = cases =>
    empty_l => empty_l
    non_empty_l:l => sort(less_l) + l.head + sort(greater_l)
      where
      less_l, greater_l
        : all ListOf(A)s
        = filter_with(x => x < l.head, l.tail)
        , filter_with(x => x >= l.head, l.tail)

tuple_type Coeffs
value (previous, current) : Int x Int

tuple_type GcdAndCoeffs
value (gcd, a, b) : Int x Int x Int

ext_euc
  : (Int, Int) => GcdAndCoeffs
  = ext_euc_rec((1, 0), (0, 1))
```

```
ext_euc_rec
  : (Coeffs, Coeffs, Int, Int) => GcdAndCoeffs
  = (a_coeffs, b_coeffs, x, cases) =>
    0 => (x, a_coeffs.previous, b_coeffs.previous)
    y => ext_euc_rec(next <- a_coeffs, next <- b_coeffs, y, x -> mod <- y)
      where
      next: Coeffs => Coeffs
        = cs => (cs.current, cs.previous - x / y * cs.current)

big_string
  : String
  = s1 + s2 + s3 + s4
    where
    s1, s2, s3, s4 : all String
      = "Hello, my name is Struggling Programmer."
      , "I have tried way too many times to fit a big chunk of text"
      , "inside my program, without it hitting the half-screen mark!"
      , "I am so glad I finally discovered lcases!!!"
```

- *Description*

  ”where” expressions allow the programmer to use values inside an expression and define them below it. They are very useful for reusing or abbreviating expressions that are specific to a particular definition or case.

  A ”where” expression begins by a line that only has word ”where” in it. It is indented as described in the ”Value Definitions” (3.4.1) or ”’cases’ Function Expressions” (3.3.2) sections. The definitions are placed below the ”where” line and must have the same indentation.

- *Grammar*

  ⟨*where-expr*⟩ ::= ‘`\n`’ ⟨*indent*⟩ ‘`where\n`’ (⟨*value-def*⟩ | ⟨*grouped-value-defs*⟩)+

3.5 Indentation and Complete Grammar for Values

3.5.1 Indentation System

The <*indent*> nonterminal in not a normal BNF nonterminal. It is a context sensitive construct that enforces the indentation rules of lcases. It depends on a integer value that shall be named ”indentation level” (il). The <*indent*> nonterminal corresponds to $2 * il$ space characters. The indentation level follows the rules below:

Indentation Rules

- At the beginnng: $il = 0$.
- At the end of the first line of a value definition: $il \leftarrow il + 1$. This also applies to grouped value definitions.
- At the end of the third line (”=” line) of a (single) value definition: $il \leftarrow il + 1$.
- At the end of a (single) value definition: $il \leftarrow il - 2$.
- At the end of grouped value definitions: $il \leftarrow il - 1$.
- After the ”`=>`” arrow in a case: $il \leftarrow il + 1$.
- At the end of a case body: $il \leftarrow il - 1$.
- In a cases function expression which does not begin at the ”=” line of a value definition:
  - After the arrow ”`=>`” at the end of the paremeters: $il \leftarrow il + 1$.
  - At the end of the whole cases function expression: $il \leftarrow il - 1$.

3.5.2 Complete Grammar for Values

⟨*literal*⟩ ::= TODO

⟨*identifier*⟩ ::= [a-z] [a-z_]* (‘`()`’ [a-z_]+)* [[0-9]]

⟨*paren-expr*⟩ ::= ‘`(`’ ⟨*op-or-func-expr*⟩ ‘`)`’

⟨*op-or-func-expr*⟩ ::= ⟨*simple-op-expr*⟩ | ⟨*op-expr-func-end*⟩ | ⟨*simple-func-expr*⟩

⟨*tuple*⟩ ::= ‘`(`’ ⟨*line-expr*⟩ ‘`,␣`’ ⟨*comma-sep-line-exprs*⟩ ‘`)`’

⟨*comma-sep-line-exprs*⟩ ::= ⟨*line-expr*⟩ (‘`,␣`’ ⟨*line-expr*⟩)*

⟨*line-expr*⟩ ::= ⟨*no-paren-op-arg*⟩ | ⟨*op-or-func-expr*⟩

⟨*big-tuple*⟩ ::=
‘`(`’ ⟨*line-expr*⟩ [‘`\n`’ ⟨*indent*⟩] ‘`,␣`’ ⟨*comma-sep-line-exprs*⟩
(‘`\n`’ ⟨*indent*⟩ ‘`,`’ ⟨*comma-sep-line-exprs*⟩)*
‘`\n`’ ⟨*indent*⟩ ‘`)`’

⟨*list*⟩ ::= ‘`[`’ [⟨*comma-sep-line-exprs*⟩] ‘`]`’

⟨*big-list*⟩ ::= ‘`[`’ ⟨*comma-sep-line-exprs*⟩ (‘`\n`’ ⟨*indent*⟩ ‘`,`’ ⟨*comma-sep-line-exprs*⟩)* ‘`\n`’ ⟨*indent*⟩ ‘`]`’

⟨*paren-func-app*⟩ ::=
[⟨*arguments*⟩] ⟨*identifier-with-arguments*⟩ [⟨*arguments*⟩]
| ⟨*arguments*⟩ ⟨*identifier*⟩ [⟨*arguments*⟩]
| ⟨*identifier*⟩ ⟨*arguments*⟩

⟨*arguments*⟩ ::= ‘`(`’ ⟨*comma-sep-line-exprs*⟩ ‘`)`’

⟨*identifier-with-arguments*⟩ ::=
[a-z] [a-z_]* (‘`()`’[a-z_]+)* ⟨*arguments*⟩ [a-z_]+ ((‘`()`’ | ⟨*arguments*⟩) [a-z_]+)* [[0-9]]

⟨*pre-func*⟩ ::= ⟨*identifier*⟩ ‘`:`’

⟨*pre-func-app*⟩ ::= ⟨*pre-func*⟩ (⟨*basic-expr*⟩ | ⟨*paren-expr*⟩ | ⟨*pre-func-app*⟩)

⟨*basic-expr*⟩ ::= ⟨*literal*⟩ | ⟨*identifier*⟩ | ⟨*tuple*⟩ | ⟨*list*⟩ | ⟨*paren-func-app*⟩ | ⟨*post-func-app*⟩

⟨*post-func*⟩ ::= ‘`.`’ ⟨*identifier*⟩

⟨*post-func-app*⟩ ::= (⟨*paren-expr*⟩ | ⟨*basic-expr*⟩) ⟨*post-func*⟩

⟨*op-expr*⟩ ::= ⟨*simple-op-expr*⟩ | ⟨*op-expr-func-end*⟩ | ⟨*big-op-expr*⟩ | ⟨*cases-op-expr*⟩

⟨simple-op-expr⟩ ::= ⟨op-arg⟩ ('␣' ⟨op⟩ '␣' ⟨op-arg⟩)+

⟨op-expr-func-end⟩ ::= ⟨simple-op-expr⟩ '␣' ⟨op⟩ '␣' ⟨simple-func-expr⟩

⟨big-op-expr⟩ ::=
  ⟨op-expr-line⟩ ('\n' ⟨indent⟩ ⟨op-expr-line⟩)*
  '\n' ⟨indent⟩ (⟨op-arg⟩ | ⟨simple-op-expr⟩ | [⟨op-expr-line⟩ '␣'] (⟨simple-func-expr⟩ | ⟨big-func-expr⟩))

⟨cases-op-expr⟩ ::= ⟨op-expr-line⟩ ('\n' ⟨indent⟩ ⟨op-expr-line⟩)* ('\n' ⟨indent⟩ | '␣') ⟨cases-func-expr⟩

⟨op-expr-line⟩ ::= (⟨op-arg⟩ | ⟨simple-op-expr⟩) '␣' ⟨op⟩

⟨op-arg⟩ ::= ⟨no-paren-op-arg⟩ | ⟨paren-expr⟩

⟨no-paren-op-arg⟩ ::= ⟨basic-expr⟩ | ⟨pre-func⟩ | ⟨post-func⟩ | ⟨pre-func-app⟩

⟨op⟩ ::= '->' | '<-' | 'o>' | '<o' | '^' | '*' | '/' | '+' | '-' | '=' | '/=' | '>' | '<' | '>=' | '<=' | '&' | '|' | ';>' | ';'

⟨func-expr⟩ ::= ⟨simple-func-expr⟩ | ⟨big-func-expr⟩ | ⟨cases-func-expr⟩

⟨simple-func-expr⟩ ::= ⟨parameters⟩ '␣=>␣' ⟨simple-func-body⟩

⟨big-func-expr⟩ ::= ⟨parameters⟩ '␣=>\n' ⟨indent⟩ (⟨simple-func-body⟩ | ⟨big-op-expr⟩)

⟨parameters⟩ ::= ⟨identifier⟩ | '(' ⟨identifier⟩ (',␣' ⟨identifier⟩)+ ')'

⟨simple-func-body⟩ ::= ⟨no-paren-op-arg⟩ | ⟨simple-op-expr⟩ | ⟨op-expr-func-end⟩

⟨cases-func-expr⟩ ::= ⟨cases-parameters⟩ '␣=>' ⟨case⟩+ ⟨end-case⟩

⟨cases-parameters⟩ ::= ⟨cases-parameter⟩ | '(' ⟨cases-parameter⟩ (',␣' ⟨cases-parameter⟩)+ ')'

⟨cases-parameter⟩ ::= ⟨parameter⟩ | 'cases'

⟨case⟩ ::= '\n' ⟨indent⟩ ⟨matching⟩ '␣=>' ⟨case-body⟩

⟨end-case⟩ ::= '\n' ⟨indent⟩ ('...' | ⟨matching⟩) '␣=>' ⟨case-body⟩

⟨matching⟩ ::= ⟨literal⟩ | ⟨identifier⟩ | ⟨pre-func⟩ ⟨matching⟩ | ⟨tuple-matching⟩ | ⟨list-matching⟩

⟨tuple-matching⟩ ::= '(' ⟨matching⟩ (',' ⟨matching⟩)+ ')'

⟨list-matching⟩ ::= '[' [⟨matching⟩ (',' ⟨matching⟩)* [', ...']] ']'

⟨case-body⟩ ::= ('␣' | '\n' ⟨indent⟩) (⟨simple-func-body⟩ | ⟨big-op-expr⟩) [⟨where-expr⟩]

⟨*value-def*⟩ ::= ⟨*indent*⟩ ⟨*identifier*⟩ '`\n`' ⟨*indent*⟩ '`:␣`' ⟨*type*⟩ '`\n`' ⟨*indent*⟩ '`=␣`' ⟨*value-expr*⟩ [⟨*where-expr*⟩]

⟨*value-expr*⟩ ::= ⟨*no-paren-op-arg*⟩ | ⟨*op-expr*⟩ | ⟨*func-expr*⟩ | ⟨*big-tuple*⟩ | ⟨*big-list*⟩

⟨*grouped-value-defs*⟩ ::=
    ⟨*indent*⟩ ⟨*identifier*⟩ ('`,␣`' ⟨*identifier*⟩)+
    '`\n`' ⟨*indent*⟩ '`:␣`' (⟨*type*⟩ ('`,␣`' ⟨*type*⟩)+ | '`all`' ⟨*type*⟩)
    '`\n`' ⟨*indent*⟩ '`=␣`' ⟨*comma-sep-line-exprs*⟩ ('`\n`' ⟨*indent*⟩ '`,`' ⟨*comma-sep-line-exprs*⟩)*

⟨*where-expr*⟩ ::= '`\n`' ⟨*indent*⟩ '`where\n`' (⟨*value-def*⟩ | ⟨*grouped-value-defs*⟩)+

4 Language Description: Types and Type Logic

4.1 Types

The constructs regarding types are two: **type expressions** and **type definitions**.

Type expressions are divided into the following categories:

- Type Identifiers
- Type Variables
- Function Types
- Product Types
- Type Application Types
- Conditional Types

which are described in the following section.

The grammar of a type expression is:

⟨*type*⟩ ::= [⟨*condition*⟩] ⟨*simple-type*⟩

⟨*simple-type*⟩ ::= ⟨*type-id*⟩ | ⟨*type-var*⟩ | ⟨*func-type*⟩ | ⟨*prod-type*⟩ | ⟨*type-app*⟩

Type definitions are divided into `tuple_type` definitions and `or_type` definitions and are described in section 4.1.2.

The grammar of a type definition is:

⟨*type-def*⟩ ::= ⟨*tuple-type-def*⟩ | ⟨*or-type-def*⟩

4.1.1 Type Expressions

Type Identifiers

- *Examples*

```
Int
Real
Char
String
FunnyType
MyDefinedType
```

- *Description*

  A type identifier is either the name of a basic type (Int, Real, Char, String) or the name of some defined type that has no type parameters. It begins with a capital letter and is followed by one or more capital or lowercase letters.

- *Grammar*

  ⟨*type-id*⟩ ::= [A-Z] [A-Za-z]+

Type Variables

- *Examples*

```
A
B
C
X
Y
Z
```

- *Examples of type variables inside bigger type expressions*

```
A => A
(A => B, B => C) => (A => C)
((A, A) => A, A, ListOf(A)s) => A
```

- *Description*

  Type Variables are used inside larger type expressions of polymorphic types. A polymorphic type is a type where any function of that type can be used as a function of any type that corresponds to substituting every type variable of the polymorphic type with a particular type. The easiest example of a polymorphic type is the type of the identity function where we have:

```
id
  : A => A
  = x => x

id(1)
  : Int
  where A is substituted by Int and id gets the type Int => Int

id("Hello")
  : String
  where A is substituted by String and id gets the type String => String
```

  A type variable is a single capital letter.

- *Grammar*

  ⟨*type-var*⟩ ::= [A-Z]

Function Types

- *Examples*

```
String => String
Real => Int
A => A
Int x Int => Int
(Real, Real, Real) => Real
(A => B, B => C) => (A => C)
(Int => Int) => Int
```

- *Description*

  A function type expression is comprised of the expressions of the types of the parameters and the expression of the type of the result, seperated by the arrow "`=>`". If there are more than one parameters, the expressions of their types are inside parentheses and seperated by commas. If there is only one parameter, the expression of its type is put in parentheses only if it is a function type. The same applies to the type of the result.

- *Grammar*

  ⟨*func-type*⟩ ::= ⟨*param-types-expr*⟩ '␣`=>`␣' ⟨*one-type*⟩

  ⟨*param-types-expr*⟩ ::= ⟨*one-type*⟩ | '`(`' ⟨*simple-type*⟩ ('`, `' ⟨*simple-type*⟩)+ '`)`'

  ⟨*one-type*⟩ ::= ⟨*type-id*⟩ | ⟨*type-var*⟩ | ⟨*prod-type*⟩ | ⟨*type-app*⟩ | '`(`' ⟨*func-type*⟩ '`)`'

Product Types

- *Examples*

```
Int x Int
Real x Real x Real
Int x Real x String
ListOf(Int)s x (Int x ListOf(String)s)
(Int => Int) x (Int x Real) x (Real => String)
```

- *Description*

  Product types are the types of tuples. They are comprised of the expressions of the types of the fields seperated by the string " x " (space 'x' space) because 'x' is very similar the symbol used in the cartesian product. If any of the fields has a product or a function type then the corresponding type expression must be inside parentheses.

- *Grammar*

  ⟨*prod-type*⟩ ::= ⟨*field-type*⟩ ('␣`x`␣' ⟨*field-type*⟩)+

  ⟨*field-type*⟩ ::= ⟨*type-id*⟩ | ⟨*type-var*⟩ | ⟨*type-app*⟩ | '`(`' (⟨*func-type*⟩ | ⟨*prod-type*⟩) '`)`'

Type Application Types

- *Examples*

```
Possibly(Int)
ListOf(Real)s
TreeOf(String)s
Error(String)OrResult(Int)
ListOf(Int => Int)s
ListOf(A)s
```

- *Description*

  Type application types are types that are produced by passing arguments to a type function generated by a `tuple_type` or an `or_type` definition. For example, given the definition of `ListOf(A)s`:

  ```
  or_type ListOf(A)s
  values non_empty_l:NonEmptyListOf(A)s | empty_l
  ```

  We have that `ListOf()s` is a type function that receives one type parameter and returns a resulting type. For example `ListOf(Int)s` is the result of passing the type argument `Int` to `ListOf()s`.

  Type application types have the same form as the name in the `tuple_type` or `or_type` definition, with the difference that type parameters are substituted by the type expressions of the arguments.

- *Grammar*

  ⟨*type-app*⟩ ::=
  [⟨*types-in-paren*⟩] ⟨*type-id-with-args*⟩ [⟨*types-in-paren*⟩]
  | ⟨*types-in-paren*⟩ ⟨*type-id*⟩ [⟨*types-in-paren*⟩]
  | ⟨*type-id*⟩ ⟨*types-in-paren*⟩

  ⟨*type-id-with-args*⟩ ::= ⟨*type-id*⟩ (⟨*types-in-paren*⟩ [A-Za-z]+)+

  ⟨*types-in-paren*⟩ ::= '`(`' ⟨*simple-type*⟩ ('`, `' ⟨*simple-type*⟩)* '`)`'

Conditional Types

- *Examples*

  ```
  (A)Has_Equality ==> (A, A) => Bool
  (A)And(B)Add_To(C) ==> (A, B) => C
  (A)Is(B)s_First ==> B => A
  (T)Has_String_Representation ==> T => String
  (E)Is_An_Environemnt_Action ==> (E(A), A => E(B)) => E(B)
  ```

- *Description*

  Conditional types are the types of values that are polymorphic not because of their structure but because they have been defined (seperately) for many different combinations of types (i.e. they are ad hoc polymorphic). They are comprised of a condition and a "simple" type (i.e. a type without a condition) which are seperated by the arrow "`==>`". The condition is a type proposition which refers to type variables inside the "simple" type and it must hold whenever the polymorphic value of that type is used. For example:

  ```
  first
    : (A)Is(B)s_First => B => A
  ```

  can be used as follows:

  ```
  pair, triple, list
    : Int x String, Real x Char x Int, ListOf(String)s
    = (42, "The answer to everything"), (3.14, 'a', 1), ["Hi!", "Hello", Heeey"]

  >> pair -> first
    : Int
    = 42
  >> triple -> first
    : Real
  ```

```
  = 3.14
>> list -> first
  : String
  = "Hi!"
```

and that is because the following propositions hold:

```
(Int)Is(Int x String)s_First
(Real)Is(Real x Char x Int)s_First
(String)Is(ListOf(String)s)s_First
```

which it turn means that the function `first` has been defined for these combinations of types. For more on how conditions, propositions and ad hoc polymorphism works, see the "Type Logic" section (4.2).

- *Grammar*

  As described in the beginning of this section, the grammar of a type is:

  ⟨*type*⟩ ::= [⟨*condition*⟩] ⟨*simple-type*⟩

  And therefore here only the grammar of the condition must be written:

  ⟨*condition*⟩ ::= ⟨*prop-name*⟩ '␣==>␣'

4.1.2 Type Definitions

Tuple Types

- *Definition Examples*

```
tuple_type Name
value (first_name, last_name) : String x String

tuple_type Date
value (day, month, year) : Int x Int x Int

tuple_type MathematicianInfo
value (name, nationality, date_of_birth) : Name x String x Date

tuple_type TreeOf(A)s
values (root, subtrees) : A x ListOf(TreeOf(A)s)s

tuple_type Indexed(T)
value (index, val) : Int x T
```

- *Usage Examples*

```
euler_info
  : MathematicianInfo
  = (("Leonhard", "Euler"), "Swiss", (15, 4, 1707))

name_to_string
  : Name => String
  = n => "\nFirst Name: " + n.first_name + "\nLast Name: " + n.last_name
```

```
print_name_and_nationality
  : MathematicianInfo => (EmptyVal)IOAction
  = ci => print(ci.name -> name_to_string + "\nNationality: " + ci.nationality)

sum_nodes
  : TreeOf(Int)s => Int
  = tree => tree.root + tree.subtrees -> apply(sum_nodes)to_all -> sum_list
```

- *Description*

  A tuple type is equivalent to a product type with a new name and names for the fields for convinience. A tuple type generates postfix functions for all of the fields by using a '.' before the name of the field. For example the `MathematicianInfo` type above generates the following functions:

  ```
  .name
    : MathematicianInfo => Name
  .nationality
    : MathematicianInfo => String
  .date_of_birth
    : MathematicianInfo => Date
  ```

  These functions are named "postfix functions" because they can be appended to their argument.

- *Grammar*

  ⟨*tuple-type-def*⟩ ::= ‘`tuple_type␣`’ ⟨*type-name*⟩ ‘`\nvalue␣`’ ‘`(`’ ⟨*identifier*⟩ (‘`,␣`’ ⟨*identifier*⟩)+ ‘`)`’ ‘`␣:␣`’ ⟨*prod-type*⟩

  ⟨*type-name*⟩ ::= [⟨*params-in-paren*⟩] (⟨*type-id*⟩ | ⟨*type-id-with-params*⟩) [⟨*params-in-paren*⟩]

  ⟨*type-id-with-params*⟩ ::= ⟨*type-id*⟩ (⟨*params-in-paren*⟩ [A-Za-z]+)+

  ⟨*params-in-paren*⟩ ::= ‘`(`’ ⟨*type-var*⟩ (‘`, `’ ⟨*type-var*⟩)* ‘`)`’

Or Types

- *Definition Examples*

  ```
  or_type Bool
  values true | false

  or_type Possibly(A)
  values the_value:A | no_value

  // needed tuple_type for ListOf(A)s
  tuple_type NonEmptyListOf(A)s
  value (head, tail) : A x ListOf(A)s

  or_type ListOf(A)s
  values non_empty_l:NonEmptyListOf(A)s | empty_l

  or_type Error(A)OrResult(B)
  values error:A | result:B
  ```

- *Usage Examples*

```
is_empty
  : ListOf(A)s => Bool
  = cases =>
    empty_l => true
    non_empty_l:anything => false

get_head
  : ListOf(A)s => Possibly(A)
  = cases =>
    empty_l => no_value
    non_empty_l:list => the_value:list.head

sum_list
  : ListOf(Int)s => Int
  = cases
    empty_l => 0
    non_empty_l:l => l.head + sum_list(l.tail)

print_err_or_res
  : Error(A)OrResult(B) => (EmptyVal)IOAction
  = cases =>
    error:e => print("Error occured: " + e -> to_string)
    result:r => print("All good! The result is: " + r -> to_string)
```

- *Description*

  Values of an `or_type` are one of many cases. Some cases have other values inside. The cases which have other values inside are followed by a semicolon and the type of the internal value. The same syntax can be used for matching that particular case in a function using the "cases" syntax, with the difference that after the colon, we write the name given to the value inside. An `or_type` definition automatically creates prefix functions for each case with an internal value. For example, for the case "`non_empty_l`" of a list, the function "`non_empty_l:`" is automatically created from the definition for which we can say:

  ```
  non_empty_l:
    : NonEmptyListOf(A)s => ListOf(A)s
  ```

  Similarly:

  ```
  the_value:
    : A => Possibly(A)
  ```

  These functions are called "prefix functions" because they are prepended to their argument. For example:

  ```
  non_empty_l
    : NonEmptyListOf(Int)s
    = (1, [2, 3, 4])
  l
    : ListOf(Int)s
    = non_empty_l:non_empty_l
  ```

  These functions can be used like any other function as arguments to other functions. For example:

  ```
  non_empty_ls_to_ls
    : ListOf(NonEmptyListOf(A)s)s => ListOf(ListOf(A)s)s
    = apply(non_empty_l:)to_each
  ```

- *Grammar*

  ⟨*or-type-definition*⟩ ::=
  `‘or_type␣’` ⟨*type-name*⟩
  `‘\nvalues␣’` ⟨*identifier*⟩ [‘`:`’ ⟨*type*⟩] (‘`␣|␣`’ ⟨*identifier*⟩ [‘`:`’ ⟨*type*⟩])*

4.2 Type Logic

Type logic is the mechanism for ad hoc polymorphism in lcases. The central notion of **type logic** is the **type proposition**. A type proposition is a proposition about types (the proposition's type parameters) and the proposition either true or false when the proposition's type parameters are substituted by particular type arguments.

Type propositions can either be defined or proven. Therefore, the following constructs exist and accomplish the aforementioned respectively: **type proposition definitions** and **type theorems**. These constructs are described in detail in the following sections.

4.2.1 Type Proposition Definitions

Type proposition definitions are split into definitions of **atomic type propositions** and definitions of **conjunction type propositions** which are described in the following paragraphs.

Atomic Type Propositions

- *Examples*

```
type_proposition (A)Is(B)s_First
value
  first : B => A

type_proposition (T)Has_String_Representation
value
  to_string : T => String

type_proposition (T)Has_A_Wrapper
value
  wrapper : A => T(A)

type_proposition (T)Has_Internal_App
value
  apply()internally : (A => B, T(A)) => T(B)

type_proposition (T)Has_Wrapd_Intern_App
value
  apply_wrapd()intern : (T(A => B), T(A)) => T(B)

type_proposition (T)Has_Unwrap_Apply
value
  unwrap_apply : (T(A), A => T(B)) => T(B)
```

The examples above define the following (ad hoc) polymorphic functions which have the respective (conditional) types:

```
first
  : (A)Is(B)s_First ==> B => A
```

```
to_string
  : (T)Has_String_Represantion ==> T => String

wrapper
  : (T)Has_A_Wrapper ==> A => T(A)

apply()internally
  : (T)Has_Internal_App ==> (A => B, T(A)) => T(B)

apply_wrapd()intern
  : (T)Has_Wrapd_Intern_App ==> (T(A => B), T(A)) => T(B)

unwrap_apply
  : (T)Has_Unwrap_Apply ==> (T(A), A => T(B)) => T(B)
```

- *Description*

  An atomic type proposition definition defines simultaneously the **atomic type proposition** itself and a **polymorphic value** (usually, but not necessarily, a function), by definining the form of the type of the value given the type parameters of the proposition. The type proposition is true or not true when the type parameters are substituted by specific type arguments depending on whether the implementation of the value has been defined for these type arguments. The aforementioned truthvalue will determines whether the value is used correctly inside the program and therefore whether the program will typecheck. In order to add more types for which the function works, i.e. define the function for these types, i.e. make the type proposition true for these types, one must prove a type theorem. The specifics of type theorems are described in the next section. For now, we'll show the example for everything mentioned in this paragraph for the proposition `(A)Is(B)s_First`:

  – Type Proposition Definition:

  ```
  type_proposition (A)Is(B)s_First
  value
    first : B => A
  ```

  – Function defined and its type:

  ```
  first
    : (A)Is(B)s_First ==> B => A
  ```

  – Type theorems for specific types:

  ```
  type_theorem (A)Is(A x B)s_First
  proof
    first = .1st

  type_theorem (A)Is(ListOf(A)s)s_First
  proof
    first = cases =>
      empty_l => show_err("Tried to take the first element of an empty_l list")
      non_empty_l:l => l.head
  ```

  – Usage of the function

  ```
  pair, list
    : Int x String, ListOf(String)s
    = (42, "The answer to everything"), ["Hi!", "Hello", Heeey"]
  ```

```
>> pair -> first
  : Int
  = 42
>> list -> first
  : String
  = "Hi!"
```

An atomic type proposition definition begins with the keyword `type_proposition` followed by the name of the proposition (including the type parameters) in the first line. The second line is the keyword `value`. The third line is indented once and has the identifier and the type expression of the value seperated by the string " : ".

- *Grammar*

  ⟨*atom-prop-def*⟩ ::= '`type_proposition␣`' ⟨*prop-name*⟩ '`\nvalue\n␣␣`' ⟨*identifier*⟩ '`␣:␣`' ⟨*simple-type*⟩

  ⟨*prop-name*⟩ ::=
  (⟨*name-part*⟩ ⟨*params-in-paren*⟩)+ [⟨*name-part*⟩]
  | (⟨*params-in-paren*⟩ ⟨*name-part*⟩)+ [⟨*params-in-paren*⟩]

  ⟨*name-part*⟩ ::= ([A-Za-z] | '_'[A-Z])+

Conjunction Type Propositions

- *Examples*

```
type_proposition (T)Has_Order
equivalent
  (T)Has_Equality, (T)Can_Be_Greater_Than(T)

type_proposition (A)And(B)Are_Comparable
equivalent
  (A)Can_Be_Greater_Than(B), (A)Can_Be_Less_Than(B)

type_proposition (T)Is_An_App_Functor
equivalent
  (T)Has_Wrapd_Intern_App, (T)Has_A_Wrapper

type_proposition (T)Is_A_Monad
equivalent
  (T)Has_Unwrap_Apply, (T)Has_A_Wrapper
```

- *Description*

  A conjuction type proposition definition is used to abbreviate the conjuction of many type propositions (i.e. AND of all of them) into one new type proposition.

  A conjunction type proposition definition begins with the keyword `type_proposition` followed by the name of the proposition (including the type parameters) in the first line. The second line is the keyword `equivalent`. The third line is indented once and has the type propostions of the conjuction seperated by commas (where the commas essentially mean "and").

- *Grammar*

  ⟨*conjunction*⟩ ::=
  '`type_proposition␣`' ⟨*prop-name*⟩ '`\nequivalent\n␣␣`' ⟨*prop-name*⟩ ('`,␣`' ⟨*prop-name*⟩)+

4.2.2 Type Theorems

Type theorems are split into theorems of **atomic type propositions** and theorems of **implication type propositions** which are described in the following paragraphs.

Atomic Type Propositions

- *Examples*

```
type_theorem (Possibly())Has_A_Wrapper
proof
  wrapper = the_value:

type_theorem (ListOf()s)Has_A_Wrapper
proof
  wrapper = x => [x]

type_theorem (Possibly())Has_Internal_App
proof
  apply()internally = (f, cases) =>
    no_value => no_value
    the_value:x => the_value:f(x)

type_theorem (ListOf()s)Has_Internal_App
proof
  apply()internally = (f, cases) =>
    empty_l => empty_l
    non_empty_l:l => non_empty_l:(f(l.head), l.tail -> apply(f)internally)
```

- *Usage*

```
a, b
  : all Possibly(Int)
  = wrapper(1), no_value

l1, l2, l3
  : all ListOf(Int)s
  = wrapper(1), empty_l, [1, 2, 3]

>> a
  : Possibly(Int)
  = the_value:1
>> b
  : Possibly(Int)
  = no_value
>> l1
  : ListOf(Int)s
  = [1]
>> l2
  : ListOf(Int)s
  = []

>> a -> apply(x => x + 1)internally
  : Possibly(Int)
```

```
  = the_value:2
>> b -> apply(x => x + 1)internally
  : Possibly(Int)
  = no_value
>> l1 -> apply(x => x + 1)internally
  : ListOf(Int)s
  = [2]
>> l2 -> apply(x => x + 1)internally
  : ListOf(Int)s
  = []
>> l3 -> apply(x => x + 1)internally
  : ListOf(Int)s
  = [2, 3, 4]
```

- *Description*

  A theorem of an atomic type proposition proves the proposition for specific type arguments, by implementing the value associated to the proposition for these type arguments. Therefore, the value associated with the proposition can be used with all the combinations of type arguments for which the type proposition is true, i.e. the combinations of type arguments for which the value has been implemented.

  A proof of a theorem of an atomic type proposition is correct when the implementation of the value associated with the proposition follows the form of the type given to the value by the definition of the proposition, i.e. the only difference between the type of the value in the theorem and the type of the value in the definition is that the type parameters of the proposition are substituted by the type arguments of the theorem.

  A theorem of an atomic type proposition begins with the keyword `type_theorem` followed by the name of the proposition with the type parameters substituted by the specific types for which the proposition will be proven. The second line is the keyword `proof`. The third line is indented once and it is the line in which the proof begins. The proof begins with the identifier of the value associated with the proposition and is followed by the string " `=` " and the value expression which implements the value.

- *Grammar*

  ⟨*atom-prop-theo*⟩ ::= '`type_theorem␣`' ⟨*prop-name-sub*⟩ '`\nproof\n␣␣`' ⟨*identifier*⟩ '`␣=␣`' ⟨*value-expr*⟩

  ⟨*prop-name-sub*⟩ ::=
  (⟨*name-part*⟩ ⟨*types-in-paren*⟩)+ [⟨*name-part*⟩]
  | (⟨*types-in-paren*⟩ ⟨*name-part*⟩)+ [⟨*types-in-paren*⟩]

Implication Type Propositions

- *Examples*

```
type_theorem (T)Has_Equality => (T)Has_Inequality
proof
  a \= b = not(a = b)

type_theorem (T)Can_Be_Greater_Than(T) => (T)Has_Less_Or_Equal
proof
  a <= b = not(a > b)

type_theorem (T)Has_Order => (T)Has_Greater_Or_Equal
proof
  a >= b = a = b | a > b
```

```
type_theorem (T)Is_An_App_Functor => (T)Has_Internal_App
proof
  apply()internally = (f, x) => apply_wrapd(wrap(f))intern(x)

type_theorem (T)Is_A_Monad => (T)Has_Wrapd_Intern_App
proof
  apply_wrapd()intern = (f, x) =>
    unwrap_apply(f, f' => unwrap_apply(x, x' => wrap(f'(x'))))
```

- *Description*

  A theorem of an implication type proposition is very similar to a theorem of an atomic type proposition in the sense that it also has an implementation of a value (in the proof). The difference is that the implementation uses another ad hoc polymorphic value (or many). Therefore, the implementation does not prove the proposition associated to the value it implements, because it assumes that the polymorphic value(s) used in the implementation is(are) already defined. In other words it proves the following: "if this(these) ad hoc polymorphic value(s) is(are) defined then we can also define this other one". This can be translated into the following implication proposition: "if the proposition associated to the value(s) we are using is true then the proposition associated to the value we are defining is true", which can be condensed to the notation with the "`=>`" arrow used in the examples.

  The proof of an implication type proposition allows the compiler to automatically create the definition for an ad hoc polymorphic value for a particular combination of types given the definitions of the ad hoc polymorphic values used in the implementation for this same combination of types. This mechanism essentially gives definitions for free, that is in the sense that when you define a set of ad hoc polymorphic values for a particular set of types you get for free all the ad hoc polymorphic values that can be defined using a subset of the defined ones.

- *Grammar*

  $\langle$*implication*$\rangle$ ::=
  ‘`type_proposition`’ $\langle$*prop-name*$\rangle$ ‘`␣=>␣`’ $\langle$*prop-name*$\rangle$ ‘`\nproof\n␣␣`’ $\langle$*identifier*$\rangle$ ‘`␣=␣`’ $\langle$*value-expr*$\rangle$

4.3 Complete Grammar for Types and Type Logic

5 Language Description: Predefined

5.1 Constants

5.2 Functions

5.3 Types

5.4 Type Propositions

6 Parser implementation

The parser was implemented using the parsec library.

6.1 AST Types

6.2 Parsers

7 Translation to Haskell

8 Running Examples

9 Conclusion