

# Lambda Cases (lcases)

Dimitris Saridakis

## Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Language Description</b>	<b>2</b>
2.1	Keywords . . . . .	3
2.2	Basic Value Expressions . . . . .	3
2.2.1	Literals and Identifiers . . . . .	3
2.2.2	Parenthesis, Tuples and Lists . . . . .	5
2.2.3	Parenthesis Function Application . . . . .	8
2.3	Value Operators . . . . .	9
2.3.1	Function Application Operators . . . . .	9
2.3.2	Function Composition Operators . . . . .	9
2.3.3	Arithmetic Operators . . . . .	10
2.3.4	Comparison and Boolean Operators . . . . .	11
2.3.5	Environment Action Operators . . . . .	11
2.3.6	Operator Expressions . . . . .	13
2.3.7	Complete Table, Precedence and Associativity . . . . .	14
2.4	Function Expressions . . . . .	15
2.4.1	Simple Function Expressions . . . . .	15
2.4.2	"fields" Special Parameter . . . . .	16
2.4.3	"cases" Syntax . . . . .	16
2.5	Value Definitions, "where" and Indentation . . . . .	18
2.5.1	Value Definitions . . . . .	18
2.5.2	"where" Expressions . . . . .	19
2.5.3	Indentation System . . . . .	20
2.6	Types . . . . .	20
2.6.1	Type expressions . . . . .	20
2.6.2	Tuple Types . . . . .	21
2.6.3	Or Types . . . . .	22
2.7	Type Logic . . . . .	23
2.7.1	Type Predicate . . . . .	23
2.7.2	Type Proposition . . . . .	23
2.7.3	Type Theorem . . . . .	23
2.8	Predefined . . . . .	23
2.8.1	Functions . . . . .	23
2.9	Program . . . . .	23
2.9.1	"main" Value . . . . .	23
2.10	Complete Grammar . . . . .	23
<b>3</b>	<b>lcases vs Haskell: Similarities and Differences</b>	<b>23</b>

<b>4</b>	<b>Parser implementation</b>	<b>23</b>
4.1	AST Types . . . . .	24
4.2	Parsers . . . . .	24
<b>5</b>	<b>Translation to Haskell</b>	<b>24</b>
<b>6</b>	<b>Running examples</b>	<b>24</b>
<b>7</b>	<b>Conclusion</b>	<b>24</b>
<b>8</b>	<b>To be removed or incorporated</b>	<b>24</b>

## 1 Introduction

Haskell is a delightful language. Yet, for some reason, it doesn't seem to have it's rightful place in terms of popularity in industry. Why is it so? Is it inherently hard to learn and therefore only the brave dare to use it, or could it be that the syntax is perplexing to the amateur eye? It is my belief that with some syntax changes that give a greater familiarity to the new user, there would be no language more compelling than (the new) Haskell. In an attempt to achieve that familiarity, I present some (hopefully useful) new syntax, of which some is closer to the imperative/OOP style (to attract more already experienced programmers from these languages), some is closer to mathematics (in which most programmers should be experienced) and some is closer to natural language (in which we are all already experienced).

## 2 Language Description

An lcases program consists of a set of definitions and theorems. Definitions are split into value definitions, type definitions, type predicate definitions and type proposition definitions. Theorems are proven type propositions. The definition of the "main" value determines the program's behaviour. Constants and functions are all considered values and they have no real distinction other than the fact that functions have a function type and constants don't. Functions (just like "values") can be passed to other functions as arguments or can be returned as a result of other functions.

### Program example: extended euclidean alogirthm

```
// type definitions

tuple_type Coeffs
value (previous, current) : Int x Int

tuple_type GcdAndCoeffs
value (gcd, a, b) : Int x Int x Int

// algorithm

init_a_coeffs, init_b_coeffs: all Coeffs
= (1, 0), (0, 1)

ext_euc: (Int, Int) => GcdAndCoeffs
= ext_euc_rec(init_a_coeffs, init_b_coeffs)

ext_euc_rec: (Coeffs, Coeffs, Int, Int) => GcdAndCoeffs
= (a_coeffs, b_coeffs, x, cases) =>
  0 => (x, a_coeffs.previous, b_coeffs.previous)
```

```

y => ext_euc_rec(next <- a_coeffs, next <- b_coeffs, y, x -> mod <- y)
  where
  next: Coeffs => Coeffs
    = fields => (current, previous - x / y * current)

// reading, printing and main

read_two_ints : (Int x Int)IOAction
= print <- "Please give me 2 ints";
  get_line ;> split_words o> apply(from_string)to_all o> ints =>
  ints -> length -> cases =>
    2 => ints -> with_io
    ... => io_error <- "You didn't give me 2 ints"

print_gcd_equation : (Int, Int, GcdAndCoeffs) => (EmptyValue)IOAction
= (x, y, fields) => print("gcd = " + gcd + " = " + sum_string)
  where
  sum_string : String
    = a + " * " + x + " + " + b + " * " + y

main : (EmptyValue)IOAction
= read_two_ints ;> ints =>
  print_gcd_and_coeffs(ints.1st, ints.2nd, ext_euc(ints.1st, ints.2nd))

```

## 2.1 Keywords

The lcases keywords are the following:

fields cases where all tuple\_type value or\_type values  
 type\_pred function type\_prop equiv type\_theo proof

Each keyword's funtionality is described in the respective section shown in the table below:

Keyword	Section
fields cases	2.4 Function Expressions
where all	2.5 Value Definitions
tuple_type value or_type values	2.6 Types
type_pred function type_prop equiv type_theo proof	2.7 Type Logic

The "fields", "cases" and "where" keywords are also reserved words. Therefore, even though they can be generated by the "identifiers" grammar, they cannot be used as identifiers (see "Literals and Identifiers" section 2.2.1).

## 2.2 Basic Value Expressions

### 2.2.1 Literals and Identifiers

#### Literals

- *Examples*

```

1 2 17 42 -100
1.61 2.71 3.14 -1234.567

```

```
'a' 'b' 'c' 'x' 'y' 'z' '.' ',' '\n'
"Hello World!" "What's up, doc?" "Alrighty then!"
```

- *Description*

We have literals for the four basic types: Int, Real, Char, String. These are the usual integers, real numbers, characters and strings. The exact specification of literals is the same as in the Haskell report.

- *Grammar*

$\langle literal \rangle ::= \langle literal \rangle$

TODO add the grammar from the haskell report

## Identifiers

- *Examples*

```
x y z
a1 a2 a3
funny_identifier
unnecessarily_long_identifier
apply()to_all
```

- *Description*

An identifier is a string used as the name of a value. It is first used in the definition of the value (see "value definition" section 2.5) and later used in the definition of other values that use that defined value. An identifier starts with a lower case letter and is followed by lower case letters or underscores. It is also possible to have pairs of parentheses in the middle of an identifier (see "parenthesis function application" section 2.2.3 for why this can be useful). Finally, an identifier can be ended with a digit.

- *Grammar*

$\langle identifier \rangle ::= [a-z]([a-z\_]|'('[a-z\_])^*[0-9]$

Even though the "fields", "cases" and "where" keywords can be generated by this grammar, they cannot be used as identifiers.

## 2.2.2 Parenthesis, Tuples and Lists

### Parenthesis

- *Examples*

```
(1 + 2)
in
(1 + 2) * 3

(x => f(x) + 1) and (s => "f(val) + 1 is: " + s)
in
val -> (x => f(x) + 1) -> to_string -> (s => "f(val) + 1 is: " + s)

("Line is: " + line) and (get_line ;> line => print("Line is: " + line))
in
do(3)times <- (get_line ;> line => print("Line is: " + line))
```

- *Description*

An expression is put in parenthesis to prioritize it or isolate it in a bigger (operator) expression. The expressions inside parenthesis are "simple" operator or function expressions, where "simple" means that the "cases" syntax does not appear anywhere in the expression (see "cases" syntax section 2.4.3).

- *Indentation Rules*

It is possible to stretch a (big) parenthesis expression over multiple lines. In that case:

- The beginning line must be ended after the '(' character.
- All the lines that contain the internal expression must be indented more than the beginning line.
- The ')' character must be the first character of the last line.
- The last line must be indented exactly as much as the first.
- The precise indentation rules are described in detail in the section "Indentation System" 2.5.3.

Example:

```
1 + (
  2 * 3 ^ 4 -
  5 / 6
) ^ 7
```

- *Grammar*

$\langle \text{paren-expr} \rangle ::= '(\langle \text{simple-expr} \rangle)'$

$\langle \text{simple-expr} \rangle ::= \langle \text{simple-op-expr} \rangle \mid \langle \text{simple-func-expr} \rangle$

$\langle \text{stretched-paren-expr} \rangle ::=$   
 $'(\backslash \mathbf{n} \langle \text{indentation} \rangle (\langle \text{simple-expr} \rangle \mid \langle \text{stretched-simple-expr} \rangle) \backslash \mathbf{n} \langle \text{indentation} \rangle)'$

$\langle \text{stretched-simple-expr} \rangle ::= \langle \text{stretched-simple-op-expr} \rangle \mid \langle \text{stretched-simple-func-expr} \rangle$

## Tuples

- *Examples*

```
(1, "What's up, doc?")
(2, "Alrighty then!", 3.14)
(x, y, z, w)
(1, my_function, (x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1/2))
```

- *Description*

Tuples are used to group many values (of possibly different types) into one. The type of the tuple can be either the product of the types of the subvalues or a defined `tuple_type` which is equivalent to the aforementioned product type i.e. the product type is in the definition of the `tuple_type` (see "tuple\_type" section 2.6.2). For example, the type of the second example above could be:

```
Int x String x Real
```

or:

```
MyType
```

assuming "MyType" has been defined in a similar way to the following:

```
tuple_type MyType
value (my_int, my_string, my_real) : Int x String x Real
```

- *Indentation Rules*

It is possible to stretch a (big) tuple expression over multiple lines. In that case:

- The tuple must begin in a new indented line starting with the character '(' and at least the first subvalue (unless it is also a stretched expression).
- The tuple can split in a new line only at a ',' character. Every such line must be indented exactly as the first.
- The tuple must be ended by a line that only contains the ')' character and is indented exactly as the first.
- In case a stretched tuple is a subvalue of another stretched expression (tuple or list) it must be indented more.
- The precise indentation rules are described in detail in the section "Indentation System" 2.5.3.

Example:

```
( "Hey, I'm the first subvalue and I'm also a relatively big string."
, 42, 3.14, "Hey, I'm the first small string", "Hey, I'm the second small string"
,
  ( "Hey, I'm a string inside the nested tuple", 2.71, 1.61
  , x, y, z
  )
)
```

- *Grammar*

$\langle \text{tuple} \rangle ::= '(\langle \text{basic-or-simple-expr} \rangle (', \sqcup' \langle \text{basic-or-simple-expr} \rangle )+ ')$

$\langle \text{basic-or-simple-expr} \rangle ::= \langle \text{basic-expr} \rangle \mid \langle \text{simple-expr} \rangle$

$\langle \text{basic-expr} \rangle ::= \langle \text{literal} \rangle \mid \langle \text{identifier} \rangle \mid \langle \text{tuple} \rangle \mid \langle \text{list} \rangle \mid \langle \text{paren-func-app} \rangle$

$\langle \text{stretched-tuple} \rangle ::=$   
 $\quad \text{'\textbackslash n' } \langle \text{indentation} \rangle \text{'(' } \langle \text{expr-in-stretched} \rangle$   
 $\quad ( [ \text{'\textbackslash n' } \langle \text{indentation} \rangle ] \text{' , ' } \langle \text{expr-in-stretched} \rangle ) +$   
 $\quad \text{'\textbackslash n' } \langle \text{indentation} \rangle \text{'(' } \langle \text{expr-in-stretched} \rangle$

$\langle \text{expr-in-stretched} \rangle ::= \langle \text{basic-or-simple-expr} \rangle \mid \langle \text{stretched-tuple} \rangle \mid \langle \text{stretched-list} \rangle \mid \langle \text{stretched-simple-expr} \rangle$

## Lists

- *Examples*

```
[1, 2, 17, 42, -100]
[1.61, 2.71, 3.14, -1234.567]
["Hello World!", "What's up, doc?", "Alrighty then!"]
[x => x + 1, x => x + 2, x => x + 3]
[x, y, z, w]
```

- *Description*

Lists are used to group many values of the same type into one. The type of the list is `ListOf(A)s` where `A` is the type of every value inside. Therefore, the types of the first four examples are:

```
ListOf(Int)s
ListOf(Real)s
ListOf(String)s
(A)And(Int)AddTo(B) --> ListOf(A => B)s
```

And the last list is only legal if `x`, `y`, `z` and `w` all have the same type. Assuming they do and it's the type `T`, the type of the list is:

```
ListOf(T)s
```

- *Indentation Rules*

It is possible to stretch a (big) list expression over multiple lines. In that case:

- The list must begin in a new indented line starting with the character '[' and at least the first element (unless it is also a stretched expression).
- The list can split in a new line only at a ',' character. Every such line must be indented exactly as the first.
- The list must be ended by a line that only contains the ']' character and is indented exactly as the first.
- In case a stretched list is a subvalue of another stretched expression (tuple or list) it must be indented more.
- The precise indentation rules are described in detail in the section "Indentation System" 2.5.3.

Example:

```
[ x => print("I'm the first function and x + 1 is: " + (x + 1))
, x => print("I'm the second function and x + 2 is: " + (x + 2))
, x => print("I'm the third function and x + 3 is: " + (x + 3))
]
```

- *Grammar*

$\langle list \rangle ::= '[' [ \langle basic-or-simple-expr \rangle ( ',' \langle basic-or-simple-expr \rangle )^* ] ']'$

$\langle stretched-list \rangle ::=$   
 $\quad '\backslash n' \langle indentation \rangle '[' \langle expr-in-stretched \rangle$   
 $\quad ( [ '\backslash n' \langle indentation \rangle ] ', \langle expr-in-stretched \rangle )^*$   
 $\quad '\backslash n' \langle indentation \rangle ']'$

### 2.2.3 Parenthesis Function Application

- *Examples*

```
f(x)
f(x, y, z)
(x)to_string
apply(f)to_all
apply(f)to_all(1)
```

- *Description*

Function application in lcases can be done in many different ways in an attempt to maximize readability. In this section, we discuss the ways function application can be done with parenthesis.

In the first two examples, we have the usual mathematical function application which is also used in most programming languages and should be familiar to the reader. That is, function application is done with the arguments of the function in parenthesis separated by commas and **appended** to the function identifier.

We extend this idea by allowing the arguments to be **prepended** to the function identifier (third example). Finally, it is also possible to have the arguments **inside** the function identifier provided the function has been **defined with parentheses inside the identifier**. For example, below is the definition of "apply()to\_all":

```
apply()to_all: (A => B, ListOf(A)s) => ListOf(B)s
= (f, cases) =>
  empty => empty
  non_empty:l => non_empty:(f <- l.head, apply(f)to_all <- l.tail)
```

The actual definition doesn't matter at this point, what matters is that the identifier is "apply()to\_all" with the parentheses **included**. This is very useful for defining functions where the argument in the middle makes the function application look and sound more like natural language.

It is possible that many parentheses pairs are present in a single function application (last example). The arguments are always inserted to the function from **left to right**. Therefore, when multiple parentheses pairs are present the arguments of the leftmost parentheses are inserted first then the next ones to the right and so on.

- *Grammar*

$\langle paren-func-app \rangle ::=$   
 $\quad \langle arguments \rangle ( \langle identifier-with-arguments \rangle | \langle identifier \rangle ) [ \langle arguments \rangle ]$   
 $\quad | \langle identifier-with-arguments \rangle [ \langle arguments \rangle ]$   
 $\quad | \langle identifier \rangle \langle arguments \rangle$

$\langle arguments \rangle ::= '(' \langle basic-or-simple-expr \rangle ( ',' \langle basic-or-simple-expr \rangle )^* ')'$



$\langle \text{identifier-with-arguments} \rangle ::=$   
 $[a-z] \langle \text{id-char-or-paren-id-char} \rangle^* ( \langle \text{arguments} \rangle [a-z\_]\langle \text{id-char-or-paren-id-char} \rangle^* ) + [ [0-9] ]$   
 $\langle \text{id-char-or-paren-id-char} \rangle ::= [a-z\_]\mid '()' [a-z\_]$

## 2.3 Value Operators

### 2.3.1 Function Application Operators

Operator	Type
<code>-&gt;</code>	$(A, A \Rightarrow B) \Rightarrow B$
<code>&lt;-</code>	$(A \Rightarrow B, A) \Rightarrow B$

The function application operators "`->`" and "`<-`" are a different way to apply functions to arguments than the usual parenthesis function application. They are meant to look like arrows that point from the argument to the function. These operators are very useful for chaining many function applications without the clutter of having to open and close parentheses for each one of the functions. For example, assuming we have the following functions with the behaviour suggested by their names and types:

```

apply()to_all : (A => B, ListOf(A)s) => ListOf(B)s
string_length: String => Int
filter_with : (A => Bool, ListOf(A)s) => ListOf(A)s
is_odd : Int => Bool
sum_ints : ListOf(Int)s => Int

```

And a list of strings:

```
strings : ListOf(String)s
```

Here is a simple way to get the number of characters in all the strings that have odd length:

```

chars_in_odd_length_strings : Int
= strings -> apply(string_length)to_all -> filter_with(is_odd) -> sum_ints

```

Ofcourse this can be done equivalently using the other operator:

```

chars_in_odd_length_strings : Int
= sum_ints <- filter_with(is_odd) <- apply(string_length)to_all <- strings

```

These operators can also be used together to put a function between two arguments if that function is commonly used that way in math (or if it looks better for a certain function). For example the "mod" function can be used like so:

```
x -> mod <- y
```

Instead of:

```
mod(x, y)
```

### 2.3.2 Function Composition Operators

Operator	Type
<code>o&gt;</code>	$(A \Rightarrow B, B \Rightarrow C) \Rightarrow (A \Rightarrow C)$
<code>&lt;o</code>	$(B \Rightarrow C, A \Rightarrow B) \Rightarrow (A \Rightarrow C)$

The function composition operators " $\circ>$ " and " $\circ<$ " are used to compose functions, each one in the corresponding direction. The use of the letter 'o' is meant to be similar to the mathematical function composition symbol ' $\circ$ ' and the symbols ' $>$ ', ' $<$ ' are used so that the operator points from the function which is applied first to the function which is applied second. A neat example using function composition is the following. Assuming we have the following functions with the behaviour suggested by their names and types:

```
split_words : String => ListOf(String)s
apply()to_all : (A => B, ListOf(A)s) => ListOf(B)s
reverse_string: String => String
merge_words : ListOf(String)s => String
```

We can reverse the all the words in a string like so:

```
reverse_words : String => String
  = split_words o> apply(reverse_string)to_all o> merge_words
```

Ofcourse this can be done equivalently using the other operator:

```
reverse_words : String => String
  = merge_words <o apply(reverse_string)to_all <o split_words
```

### 2.3.3 Arithmetic Operators

Operator	Type
$\wedge$	(A)ToThe(B)Gives(C) $--> (A, B) => C$
$*$	(A)And(B)MultiplyTo(C) $--> (A, B) => C$
$/$	(A)Divides(B)To(C) $--> (B, A) => C$
$+$	(A)And(B)AddTo(C) $--> (A, B) => C$
$-$	(A)SubtractsFrom(B)To(C) $--> (B, A) => C$

The usual arithmetic operators work as they are expected, similarly to mathematics and other programming languages for the usual types. However, they are generalized. The examples below show their generality:

```
>> 1 + 1
2
>> 1 + 3.14
4.14
>> 'a' + 'b'
"ab"
>> 'w' + "ord"
"word"
>> "Hello " + "World!"
"Hello World!"
>> 5 * 'a'
"aaaaa"
>> 5 * "hi"
"hihihihihi"
>> "1,2,3" - ',', '
"123"
```

Let's analyze further the example of addition. The type can be read as such: the '+' operator has the type  $(A, B) => C$ , provided that the type proposition  $(A)And(B)AddTo(C)$  holds. This proposition being true, means that addition has been defined for these three types (see section "type logic" 2.7 for more on type propositions). For example, by the examples above we can deduce that the following propositions are true (in the order of the examples):

```

(Int)And(Int)AddTo(Int)
(Int)And(Real)AddTo(Real)
(Char)And(Char)AddTo(String)
(Char)And(String)AddTo(String)
(Int)And(Char)MultiplyTo(String)
(Int)And(String)MultiplyTo(String)
(Char)SubtractsFrom(String)To(String)

```

This allows us to use the familiar arithmetic operators in types that are not necessarily numbers but it is somewhat intuitively obvious what they should do in those other types. Furthermore, their behaviour can be defined by the user for new user defined types!

### 2.3.4 Comparison and Boolean Operators

Operator	Type
<code>=</code> <code>/=</code>	$(A) \text{HasEquality} \rightarrow (A, A) \Rightarrow \text{Bool}$
<code>&gt;</code> <code>&lt;</code> <code>&gt;=</code> <code>&lt;=</code>	$(A) \text{HasOrder} \rightarrow (A, A) \Rightarrow \text{Bool}$
<code>&amp;</code> <code> </code>	$(\text{Bool}, \text{Bool}) \Rightarrow \text{Bool}$

The comparison and boolean operators behave the same as in Haskell and very similarly to most programming languages. The main difference is that in `lcases` the "equals", "and" and "or" operators have the symbol once (`=` `&` `|`) rather than twice (`==` `&&` `||`).

### 2.3.5 Environment Action Operators

Operator	Type
<code>; &gt;</code>	$(E) \text{IsAnEnvAction} \rightarrow (E(A), A \Rightarrow E(B)) \Rightarrow E(B)$
<code>;</code>	$(E) \text{IsAnEnvAction} \rightarrow (E(A), E(B)) \Rightarrow E(B)$

#### Simple Example

```
print_string("I'll repeat the line.") ; get_line ;> print_string
```

The example above demonstrates the use of the environment action operators with the `IOAction` environment action, which is how IO is done in `lcases`. Some light can be shed on how this is done, if we take a look at the types (as always!):

```

print_string : String => (EmptyValue)IOAction
print_string("I'll repeat the line.") : (EmptyValue)IOAction
get_line    : (String)IOAction

; : (E)IsAnEnvAction --> (E(A), E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line : (String)IOAction
  where
    (IOAction)IsAnEnvAction, E = IOAction, A = EmptyValue, B = String

;> : (E)IsAnEnvAction --> (E(A), A => E(B)) => E(B)
print_string("I'll repeat the line.") ; get_line ;> print_string : (EmptyValue)IOAction
  where
    (IOAction)IsAnEnvAction, E = IOAction, A = String, B = EmptyValue

```

## Example program

```
main : (EmptyValue)IOAction
  = print_string <- "Hello! What's your name?" ; get_line ;> name =>
    print_string("Oh hi " + name + "! What's your age?") ; get_line ;> age =>
    print_string("Oh that's crazy " + name + "! I didn't expect you to be " + age + "!");
```

In this bigger but similar example the types are:

```
print_string : String => (EmptyValue)IOAction
get_line : (String)IOAction

print_string <- "Hello! ... " : (EmptyValue)IOAction
print_string("Oh hi...) : (EmptyValue)IOAction
print_string("Oh that's crazy...) : (EmptyValue)IOAction

; : (E)IsAnEnvAction --> (E(A), E(B)) => E(B)
print_string("Oh hi...) ; get_line : (String)IOAction
  where
    (IOAction)IsAnEnvAction, E = IOAction, A = EmptyValue, B = String

age => print_string("Oh that's crazy...) : String => (EmptyValue)IOAction

;> : (E)IsAnEnvAction --> (E(A), A => E(B)) => E(B)

print_string("Oh hi...) ; get_line ;> age =>
print_string("Oh that's crazy...) : (EmptyValue)IOAction
  where
    (IOAction)IsAnEnvAction, E = IOAction, A = String, B = EmptyValue

print_string <- "Hello..." ; get_line : (String)IOAction
name => print_string("Oh hi ... (till the end) : String => (EmptyValue)IOAction

print_string <- "Hello..." ; get_line ;> name =>
print_string("Oh hi ... (till the end)
  : (EmptyValue)IOAction
```

Therefore, "main : (EmptyValue)IOAction" checks out. The key here is to remember that function expressions extend to the end of the whole expression. Therefore, we have "name => ... (till the end)" and "age => ... (till the end)" as the second arguments of the two occurrences of the ";>" operator.

## Description

The environment action operators are used to combine values that have some environment action embedded. Environment actions are also represented by types. These types take other types as arguments and produce new types (just like ListOf(s)). The values of the type E(A) where (E)IsAnEnvAction have inside both a value of type A together with an environment action of type E.

The effect of the operator ";" described in words is the following: given a value of type E(A) (which is a value of type A with an environment action of type E) and a value of type E(B) (which is a value of type B with an environment action of type E), combine those two values by producing the value of type B with a new environment action of type E which is the combination of the two environment actions. The overall effect produces a value of type B with an environment action of type E and therefore the overall value is of type E(B).

Note that the value of type A inside the first value is not used anywhere. This is mostly used when A = EmptyValue.

This happens because values of type `E(EmptyValue)` are used for their environment action only (e.g. `print_string(...) : (EmptyValue)IOAction`).

How the two environment actions of the `E(A)` and `E(B)` values are combined to produce the new environment action is specific to the environment action type `E`.

The effect of the operator `;>` described in words is the following: given a value of type `E(A)` (which is a value of type `A` with an environment action of type `E`) and a value of type `A => E(B)` (which is a function from a value of type `A` to a value of type `B` with an environment action of type `E`), combine those two values by passing the value of type `A` inside the first value as an argument to the second value. This results in a value of type `E(B)` (which is a value of type `B` with an environment action of type `E`). Then, produce this value of type `B` with a new environment action of type `E` that is produced by combining the first environment action with the resulting environment action. The overall effect produces a value of type `B` with an environment action of type `E` and therefore the overall value is of type `E(B)`.

### 2.3.6 Operator Expressions

- *Examples*

```
1 + 2
1 + x * 3 ^ y
"Hello " + "World!"
x -> f -> g
f o> g o> h
x = y
x >= y - z & x < 2 * y
get_line ; get_line ;> line => print("Second line: " + line)
```

- *Description*

Operator expressions are expressions that use operators. Operators act like two-argument-functions that are placed in between their arguments. Therefore, they have function types and they act as it is described in their respective sections above this one.

An operator expression might have multiple operators. The order of operations is explained in the next section ("Complete Table, Precedence and Associativity") in Table 2.

Just like functions, the sub-expressions that act as arguments to an operator, must have types that match the types expected by the operator.

It is possible to end an operator expression with a function. This is mostly useful with the `;>` operator (see previous section: "Environment Operators"), but it is also possible with the following operators: `->`, `o>`, `<o`.

- *Grammar*

```
<op-expr> ::= <op-expr-start> ( <op-arg> | <func-expr> )
<simple-op-expr> ::= <op-expr-start> ( <op-arg> | <simple-func-expr> )
<op-expr-start> ::= ( <op-arg> '␣' <op> ( '␣' | '\n' <indentation> ) )+
<op-arg> ::= <basic-expr> | <paren-expr>
<op> ::= '->' | '<-' | 'o>' | '<o' | '^' | '*' | '/' | '+' | '-' | '=' | '/=' | '>' | '<' | '>=' | '<=' | '&' | '|' | ';' | ';>';'
```

### 2.3.7 Complete Table, Precedence and Associativity

Table 1: The complete table of lcases operators along with their types and their short descriptions.

Operator	Type	Description
$\rightarrow$	$(A, A \Rightarrow B) \Rightarrow B$	Right function application
$\leftarrow$	$(A \Rightarrow B, A) \Rightarrow B$	Left function application
$\circ\rightarrow$	$(A \Rightarrow B, B \Rightarrow C) \Rightarrow (A \Rightarrow C)$	Right function composition
$\leftarrow\circ$	$(B \Rightarrow C, A \Rightarrow B) \Rightarrow (A \Rightarrow C)$	Left function composition
$\sim$	$(A)\text{ToThe}(B)\text{Gives}(C) \dashrightarrow (A, B) \Rightarrow C$	General exponentiation
$*$	$(A)\text{And}(B)\text{MultiplyTo}(C) \dashrightarrow (A, B) \Rightarrow C$	General multiplication
$/$	$(A)\text{Divides}(B)\text{To}(C) \dashrightarrow (B, A) \Rightarrow C$	General division
$+$	$(A)\text{And}(B)\text{AddTo}(C) \dashrightarrow (A, B) \Rightarrow C$	General addition
$-$	$(A)\text{SubtractsFrom}(B)\text{To}(C) \dashrightarrow (B, A) \Rightarrow C$	General subtraction
$= / =$	$(A)\text{HasEquality} \dashrightarrow (A, A) \Rightarrow \text{Bool}$	Equality operators
$> < >= <=$	$(A)\text{HasOrder} \dashrightarrow (A, A) \Rightarrow \text{Bool}$	Order operators
$\& \mid$	$(\text{Bool}, \text{Bool}) \Rightarrow \text{Bool}$	Boolean operators
$; >$	$(E)\text{IsAnEnvAction} \dashrightarrow (E(A), A \Rightarrow E(B)) \Rightarrow E(B)$	Monad bind
$;$	$(E)\text{IsAnEnvAction} \dashrightarrow (E(A), E(B)) \Rightarrow E(B)$	Monad then

The order of operations is done from highest to lowest precedence. In the same level of precedence the order is done from left to right if the associativity is "Left" and from right to left if the associativity is "Right". For the operators that have associativity "None" it is not allowed to place them in the same operator expression. The precedence and associativity of the operators is shown in the table below.

Table 2: The table of precedence and associativity of the lcases operators.

Operator	Precedence	Associativity
->	10 (highest)	Left
<-	9	Right
o> <o	8	Left
^	7	Right
* /	6	Left
+ -	5	Left
= /= > < >= <=	4	None
&	3	Left
	2	Left
; > ;	1	Left

## 2.4 Function Expressions

Function expressions are either simple function expressions or function expressions that use the "cases" syntax. For more details and examples of both see the following sections.

### Grammar

$\langle func\text{-}expr \rangle ::= \langle simple\text{-}func\text{-}expr \rangle \mid \langle cases\text{-}func\text{-}expr \rangle$

#### 2.4.1 Simple Function Expressions

- *Examples*

`a => 17 * a + 42`

`(x, y, z) => (x ^ 2 + y ^ 2 + z ^ 2) ^ (1 / 2)`

- *Description*

Simple function expressions are used to define functions or be part of bigger expressions as anonymous functions. They are comprised by their parameters and their body. A parameter is either an identifier or the keyword "fields" (see next section: "fields" Special Parameter). The parameters are either only one, in which case there is no parenthesis, or they are many, in which case they are in parenthesis, separated by commas. The parameters and the body are separated by an arrow ("=>"). The body is a basic expression or a "simple" operator expression meaning that there is no "cases" syntax inside the operator expression.

- *Grammar*

$\langle simple\text{-}func\text{-}expr \rangle ::= \langle parameters \rangle \text{'>'} ( \langle basic\text{-}expr \rangle \mid \langle simple\text{-}op\text{-}expr \rangle )$

$\langle parameters \rangle ::= \langle parameter \rangle \mid \text{'('} \langle parameter \rangle ( \text{'>' } \langle parameter \rangle ) + \text{'>'}$

$\langle parameter \rangle ::= \langle identifier \rangle \mid \text{'fields'}$

$$\langle \textit{stretched-simple-func-expr} \rangle ::=$$

$$\langle \textit{parameters} \rangle \text{ '␣' } \Rightarrow \text{ '\n' } \langle \textit{indentation} \rangle$$

$$( \langle \textit{basic-expr} \rangle \mid \langle \textit{simple-op-expr} \rangle \mid \langle \textit{stretched-tuple} \rangle \mid \langle \textit{stretched-list} \rangle \mid \langle \textit{stretched-simple-op-expr} \rangle )$$

### 2.4.2 "fields" Special Parameter

- *Examples*

```
add_ints_and_print_with_string: Int x Int x String
  = fields => print("Ints: " + (1st + 2nd) + "\nString: " + 3rd)

tuple_type GcdAndCoeffs
value (gcd, a, b) : Int x Int x Int

print_gcd_equation : (Int, Int, GcdAndCoeffs) => (EmptyValue)IOAction
  = (x, y, fields) => print("gcd = " + gcd + " = " + sum_string)
  where
    sum_string : String
      = a + " * " + x + " + " + b + " * " + y
```

- *Description*

The **"fields"** special parameter can be used when the parameter is of a **product type** (e.g. `Int x Int x String`) or of a **tuple type** (see "Tuple Types" section 2.6.2). It allows the direct use of the **subvalues** (**fields**) of that parameter in the function body without the need to name the parameter itself. This is very useful when the whole parameter is not used but (some or all of) it's subvalues are.

In the case where the parameter is of a **product type** the special identifiers "1st, 2nd, 3rd, 4th and 5th" can be used for the respective subvalues. Similarly, when the parameter is of a **tuple type** the identifiers used in the type's definition can be used for the subvalues.

The use of the **"fields"** keyword in multiple parameters is possible as long as it is not used in two or more parameters of the same type, to avoid ambiguities.

### 2.4.3 "cases" Syntax

- *Examples*

```
print_sentimental_bool : Bool => (EmptyValue)IOAction
  = cases =>
    true => print <- "It's true!! :)"
    false => print <- "It's false... :("

or_type TrafficLight
values green | amber | red

print_sentimental_traffic_light : Bool => (EmptyValue)IOAction
  = cases =>
    green => print <- "It's green! Let's go!!! :)"
    amber => print <- "Go go go, fast!"
    red => print <- "Stop right now! You're going to kill us!!"

is_not_red : TrafficLight => Bool
  = cases =>
```



```

    green => true
    amber => true
    red => false

is_seventeen_or_forty_two : Int => Bool
= cases =>
    17 => true
    42 => true
    ... => false

traffic_lights_match : (TrafficLight, TrafficLight) => Bool
= (cases, cases) =>
    green, green => true
    amber, amber => true
    red, red => true
    ... => false

gcd : (Int, Int) => Int
= (x, cases) =>
    0 => x
    y => gcd(y, x -> mod <- y)

is_empty : ListOf(A)s => Bool
= cases =>
    empty => true
    non_empty:anything => false

apply()to_all: (A => B, ListOf(A)s) => ListOf(B)s
= (f, cases) =>
    empty => empty
    non_empty:list => non_empty:(f <- list.head, apply(f)to_all <- list.tail)

```

- *Description*

"cases" is a keyword that works as a special parameter. The difference is that instead of giving the name "cases" to that parameter, it let's you pattern match on the possible values of that parameter and return a different result for each particular case.

The "cases" keyword can only be used on parameters that have either one of the basic types (Int, Real, Char, String) or an `or_type` (e.g. Bool, ListOf(A)s).

The last case can be "`... => (body of default case)`" to capture all remaining cases while dismissing the value (e.g. `is_seventeen_or_forty_two` example), or it can be "`some_identifier => (body of default case)`" to capture all remaining cases while being able to use the value with the name "`some_identifier`" (e.g. "`y`" in `gcd` example).

It is possible to use the "cases" keyword in multiple parameters to match on all of them. By doing that, each case represents a particular combination of values for the parameters involved (e.g. `traffic_lights_match` example).

A function expression that uses the "cases" syntax must contain the "cases" keyword in at least one parameter. The number of matching expressions in all cases must be the same as the number of parameters with the "cases" keyword.

- *Grammar*

$$\langle \text{cases-func-expr} \rangle ::= \langle \text{cases-parameters} \rangle \text{'\_=>'} \langle \text{cases} \rangle$$

$$\langle \text{cases-parameters} \rangle ::= \langle \text{cases-parameter} \rangle \mid \text{'('} \langle \text{cases-parameter} \rangle \text{' ,\_ ' } \langle \text{cases-parameter} \rangle \text{' )+ '}$$

$$\langle \text{cases-parameter} \rangle ::= \langle \text{parameter} \rangle \mid \text{'cases'}$$

$$\langle \text{cases} \rangle ::= \langle \text{case} \rangle + [ \langle \text{default-case} \rangle ]$$

$$\langle \text{case} \rangle ::= \text{'\n'} \langle \text{indentation} \rangle \langle \text{matching} \rangle \text{' ,\_ ' } \langle \text{matching} \rangle \text{' )* ' } \text{'\_=>\_'} \langle \text{simple-op-expr} \rangle [ \langle \text{where-expr} \rangle ]$$

$$\langle \text{default-case} \rangle ::= \text{'\n'} \langle \text{indentation} \rangle \text{' \dots\_=>\_'} \langle \text{simple-op-expr} \rangle$$

$$\langle \text{matching} \rangle ::= \langle \text{literal} \rangle \mid \langle \text{identifier} \rangle [ \text{' : ' } \langle \text{identifier} \rangle ]$$

## 2.5 Value Definitions, "where" and Indentation

### 2.5.1 Value Definitions

- *Examples*

```
foo : Int
    = 42
```

```
val1, val2, val3 : Int, Bool, Char
    = 42, true, 'a'
```

```
int1, int2, int3 : all Int
    = 1, 2, 3
```

```
f : (Int, Int, Int) => Int
    = (a, b, c) => a + b * c
```

- *Description*

Value definitions are the main building block of lcases programs. To define a new value you give it a name, a type and an expression. It is possible to group value definitions together by separating the names, the types and the expressions with commas. This is very useful for not cluttering the program with many definitions for values with small expressions (e.g. constants). When grouping definitions together it is also possible to use the keyword "all" to give the same type to all the values.

The name and the type are separated by the string " : ". The line below is indented two spaces more than the "name-type" line, begins with "=" and continues with the expression. The value definitions are either beginning in the first column, in which case they can be "seen" by all other definitions, or they are in a "where" expression (see section below), in which case they can be "seen" by the expression specified by the indentation rules (see "Indentation System" section 2.5.3).

- *Grammar*

$$\begin{aligned} \langle \text{value-def} \rangle ::= & \\ & \langle \text{indentation} \rangle \langle \text{identifier} \rangle \text{'\_ : \_'} \langle \text{type} \rangle \\ & \text{'\n'} \langle \text{indentation} \rangle \text{'\_ = \_'} \langle \text{value-expr} \rangle [ \langle \text{where-expr} \rangle ] \end{aligned}$$

$$\langle value\text{-}expr \rangle ::= \langle basic\text{-}expr \rangle \mid \langle op\text{-}expr \rangle \mid \langle func\text{-}expr \rangle$$
$$\langle \textit{grouped-value-defs} \rangle ::=$$

$$\langle \textit{indentation} \rangle \langle \textit{identifier} \rangle ( \langle \textit{' , ' } \rangle \langle \textit{identifier} \rangle ) + \langle \textit{' : ' } \rangle ( \langle \textit{type} \rangle ( \langle \textit{' , ' } \rangle \langle \textit{type} \rangle ) + | \langle \textit{' all' } \rangle \langle \textit{type} \rangle )$$

$$\langle \textit{' \backslash n' } \rangle \langle \textit{indentation} \rangle \langle \textit{' = ' } \rangle \langle \textit{basic-or-simple-expr} \rangle ( [ \langle \textit{' \backslash n' } \rangle \langle \textit{indentation} \rangle ] \langle \textit{' , ' } \rangle \langle \textit{basic-or-simple-expr} \rangle ) +$$

### 2.5.2 "where" Expressions

- *Examples*

```

sort : (A)HasOrder--> ListOf(A)s => ListOf(A)s
= cases =>
  empty => empty
  non_empty:l => sort(less_l) + l.head + sort(greater_l)
    where
      less_l : ListOf(A)s
        = filter_with(x => x < l.head, l.tail)
      greater_l : ListOf(A)s
        = filter_with(x => x >= l.head, l.tail)

tuple_type Coeffs
value (previous, current) : Int x Int

tuple_type GcdAndCoeffs
value (gcd, a, b) : Int x Int x Int

ext_euc: (Int, Int) => GcdAndCoeffs
= ext_euc_rec((1, 0), (0, 1))

ext_euc_rec: (Coeffs, Coeffs, Int, Int) => GcdAndCoeffs
= (a_coeffs, b_coeffs, x, cases) =>
  0 => (x, a_coeffs.previous, b_coeffs.previous)
  y => ext_euc_rec(next <- a_coeffs, next <- b_coeffs, y, x -> mod <- y)
    where
      next: Coeffs => Coeffs
        = fields => (current, previous - x / y * current)

big_string : String
= s1 + s2 + s3 + s4
  where
    s1, s2, s3, s4 : all String
      = "Hello, my name is Struggling Programmer."
      , "I have tried way too many times to fit a big chunk of text"
      , "inside my program, without it hitting the half-screen mark!"
      , "I am so glad I finally discovered lcases!!!"

```

- *Description*

"where" expressions allow the programmer to use values inside an expression and define them below it. They are very useful for reusing or abbreviating expressions that are specific to a particular definition (i.e. they are not used anywhere outside that definition). "where" expressions begin by a line that only has word "where" in it and is indented so that it follows the indentation rules (see "Indentation System" section below). The definitions are placed below the "where" line and must have the same indentation.

- *Grammar*

$\langle where\text{-}expr \rangle ::= \text{'\n'} \langle indentation \rangle \text{'where\n'} ( \langle value\text{-}def \rangle \mid \langle grouped\text{-}value\text{-}defs \rangle ) +$

### 2.5.3 Indentation System

The  $\langle indentation \rangle$  nonterminal is not a normal BNF nonterminal. It is a context sensitive construct that enforces the indentation rules of lcases.

#### Informal indentation rules

- An expression inside parentheses (which extends over multiple lines) is indented more than the lines with
- Whenever the  $\langle indentation \rangle$  nonterminal appears in the grammar there must be 2\*indetation level

## 2.6 Types

### 2.6.1 Type expressions

#### Examples

Int

String => String

Int x Int

Int x Int => Real

A => A

(A => B, B => C) => (A => C)

((A, A) => A, A, ListOf(A)s) => A

((B, A) => B, B, ListOf(A)s) => B

(T)HasStringRepresantion --> T => String

#### Description

Examples	Description
Int Char String	Base types
A => A (A => B, B => C) => (A => C)	Polymorphic types. A, B, C ... are type variables

#### Differences from Haskell

lcases	haskell	difference description
A => A	a => a	Type variables for polymorphic types are

## Grammar

$\langle type \rangle ::= \langle type-application \rangle \mid \langle product-type \rangle \mid \langle function-type \rangle$

$\langle type-application \rangle ::= [ \langle types-in-paren \rangle ] \langle type-identifier \rangle ( \langle types-in-paren \rangle ( [A-Za-z] )^* )^* [ \langle types-in-paren \rangle ]$

$\langle types-in-paren \rangle ::= ' ( ' \langle type \rangle ( ' , ' \langle type \rangle )^* ' ) '$

$\langle type-identifier \rangle ::= [A-Z] ( [A-Za-z] )^*$

$\langle product-type \rangle ::= \langle product-subtype \rangle ( ' \times ' \langle product-subtype \rangle )^+$

$\langle product-subtype \rangle ::= ' ( ' ( \langle function-type \rangle \mid \langle product-type \rangle ) ' ) ' \mid \langle type-application \rangle$

$\langle function-type \rangle ::= \langle input-types-expression \rangle ' \Rightarrow ' \langle one-type \rangle$

$\langle input-types-expression \rangle ::= \langle one-type \rangle \mid \langle two-or-more-types-in-paren \rangle$

$\langle one-type \rangle ::= \langle type-application \rangle \mid \langle product-type \rangle \mid ' ( ' \langle function-type \rangle ' ) '$

$\langle two-or-more-types-in-paren \rangle ::= ' ( ' \langle type \rangle ( ' , ' \langle type \rangle )^+ ' ) '$

### 2.6.2 Tuple Types

#### Definition Examples

```
tuple_type Name
value (first_name, last_name) : String x String

tuple_type ClientInfo
value (name, age, nationality) : Name x Int x String

tuple_type Date
value (day, month, year) : Int x Int x Int

tuple_type (A)And(B)
value (a_value, b_value) : A x B

tuple_type (ExprT)WithPosition
value (expr, line, column) : ExprT x Int x Int
```

#### Usage Examples

```
giorgos_info : ClientInfo
  = ("Giorgos", "Papadopoulos"), 42, "Greek")

john_info : ClientInfo
  = ("John", "Doe"), 42, "American")

name_to_string : Name => String
  = fields => "First Name: " + first_name + "\nLast Name: " + last_name

print_name_and_nationality : ClientInfo => (EmptyValue)IOAction
```

```

= fields => print(name -> name_to_string + "\nNationality: " + nationality)

print_error_in_expr : (SomeDefinedExprT)WithPosition => (EmptyValue)IOAction
= ewp =>
  print("Error in the expression:" + es + "\nAt the position: (" + ls + ", " + cs + ")")
  where
    es, ls, cs : all String
    = ewp.expr->to_string, ewp.line->to_string, ewp.column->to_string

```

## Description

Tuple types group many values into a single value. They are specified by their name, the names of their sub-values and the types of their subvalues. They generate projection functions for all of their subvalues by using a `'.'` before the name of the subvalue. For example the `ClientInfo` type above generates the following functions:

```

.name : ClientInfo => String
.age : ClientInfo => Int
.nationality : ClientInfo => String

```

These functions shall be named "postfix functions" as they can just be appended to their argument.

## Definition Grammar

```

<tuple-type-definition> ::=
  'tuple_type' <type-application>
  '\nvalue' ' (' <identifier> (',' <identifier>)* ')' ' : ' <product-type>

```

### 2.6.3 Or Types

#### Examples

```

or_type Bool
values true | false

```

```

or_type Possibly(A)
values the_value:A | no_value

```

```

or_type ListOf(A)s
values non_empty:NonEmptyListOf(A)s | empty

```

```

tuple_type NonEmptyListOf(A)s
value (head, tail) : A x ListOf(A)s

```

```

is_empty : ListOf(A)s => Bool
= cases =>
  empty => true
  non_empty:anything => false

```

```

get_head : ListOf(A)s => Possibly(A)
= cases =>
  empty => no_value
  non_empty:list => the_value:list.head

```

## Description

Values of an `or_type` are one of many cases that possibly have other values inside. The cases which have other values inside are followed by a semicolon and the type of the internal value. The same syntax can be used for matching that particular case in a function using the "cases" syntax, with the difference that after the colon, we write the name given to the value inside. `Or_types` and basic types are the only types on which the "cases" syntax can be used. The cases of an `or_type` which have a value inside create functions. For example, the case "non\_empty" of a list creates the function "non\_empty:" for which we can say:

```
non_empty: : NonEmptyListOf(A)s => ListOf(A)s
```

Similarly:

```
the_value: : A => Possibly(A)
```

These functions shall be named "prefix functions" as they are prepended to their argument. For example:

```
head_and_tail : NonEmptyListOf(Int)s  
= (1, [2, 3, 4])
```

```
list : ListOf(Int)s  
= non_empty:head_and_tail
```

These functions can be used like any other function as arguments to other functions. For example:

```
heads_and_tails_to_lists : ListOf(NonEmptyListOf(A)s)s => ListOf(ListOf(A)s)s  
= apply(non_empty:)to_each
```

## Definition Grammar

$\langle or\_type\_definition \rangle ::=$   
     $\langle 'or\_type\_ \square' \langle type\_application \rangle$   
     $\langle '\backslash nvalues\_ \square' \langle identifier \rangle [ ':' \langle type \rangle ] ( '\square | \square' \langle identifier \rangle [ ':' \langle type \rangle ] )^*$

## 2.7 Type Logic

### 2.7.1 Type Predicate

### 2.7.2 Type Proposition

### 2.7.3 Type Theorem

## 2.8 Predefined

### 2.8.1 Functions

## 2.9 Program

### 2.9.1 "main" Value

## 2.10 Complete Grammar

### Program

$\langle program \rangle ::= (\langle value\_definitions \rangle | \langle type\_def \rangle)^+$

## 3 lcases vs Haskell: Similarities and Differences

## 4 Parser implimentation

The parser was implemented using the parsec library.

## 4.1 AST Types

## 4.2 Parsers

## 5 Translation to Haskell

## 6 Running examples

## 7 Conclusion

## 8 To be removed or incorporated

Functor  $f \Rightarrow (A) \text{HasInternalFuncApp}$   
Applicative  $f \Rightarrow (A) \text{CanApplyWrappedFuncToWrappedArg}$

### Examples in Haskell

```
data ClientInfo =  
    ClientInfoC String Int String
```

```
data WithPosition a =  
    WithPositionC a Int Int
```

```
data Pair a b =  
    PairC a b
```

### Examples in Haskell

```
{-# language LambdaCase #-}
```

```
data Bool =  
    Ctrue | Cfalse
```

```
data Possibly a =  
    Cwrapper a | Cnothing
```

```
data ListOf_s a =  
    Cnon_empty (NonEmptyValueListOf_s a) | Cempty
```

```
data NonEmptyValueListOf_s a =  
    CNonEmptyValueListOf_s a (ListOf_s a)
```

```
is_empty :: ListOf_s a => Bool  
is_empty = \case  
    Cempty => Ctrue  
    Cnon_empty (CNonEmptyValueListOf_s head tail) => Cfalse
```

```
get_head :: ListOf_s a => Possibly a  
get_head = \case  
    Cempty => Cnothing  
    Cnon_empty (CNonEmptyValueListOf_s head tail) => Cwrapper head
```



## Examples in Haskell

```
foo :: Int
foo = 42
```

```
val1 :: Int
val1 = 42
val2 :: Bool
val2 = true
val3 :: Char
val3 = 'a'
```

```
int1 :: Int
int1 = 1
int2 :: Int
int2 = 2
int3 :: Int
int3 = 3
```

```
succ :: Int => Int
succ = \x => x + 1
```

```
f :: Int => Int => Int => Int
f = \a b c => a + b * c
```

Or Types the following have automatically generated functions:

is\_case:



## Hi

- *Examples*

- *Description*

hi

- *Grammar*

$\langle identifier \rangle ::= [a-z]([a-z\_]|'(' [a-z\_])^* [0-9])$